
'   TABLE OF CONTENTS 

'   including a few short macro descriptions 

'

' Purpose

' Warranty notice

' Copyright notice

' Installation

' MacroBundle update

'   Service macros:

'

' InsertMBToolbar 


' RemoveMTToolbar 

' InsertMBMenu 


' RemoveMTMenu 

'

'   Least squares macros:

'

' LS
a general least squares fitting routine for linear, 

' 
polynomial, and multivariate fitting, assuming one 

' 
dependent variable. LS0 forces the fit through the 

'
origin, LS1 does not. The output provides the 

'
parameter values, their standard deviations, the 

' 
standard deviation of the  fit to the function, and 

'
the covariance matrix; display of the matrix of 

'
linear correlation coefficients is optional

' LSPoly  
applies LS to polynomial fitting to a 

'
polynomial of gradually increasing order

' LSMulti 
applies LS to an increasing number of terms

' LSPermute 
computes the standard deviation of the fit  

' 
for all possible permutations of multivariate 

' 
parameters of up to six terms

' Ortho

a Gram-Schmidt orthogonalization alternative 

' 
to traditional least squares analysis

' ELS
least squares smoothing and differentiation for an 

'   equidistant (in the independent variable) but otherwise 

'   arbitrary function using a 'Savitzky-Golay' moving 

'   polynomial fit. ELSfixed uses a fixed-order polynomial,

'   ELSauto self-optimizes the order as it moves along the  

'   function.  

' WLS
the equivalent of LS but including user-assignable 

'   weights

' SolverAid
provides uncertainty estimates (standard 

'   deviations and the covariance matrix) for Solver-derived 

'   parameter values
 

' Propagation
computes the propagation of uncertainty for  

'   a single function, for various independent input para-

'   meters with known standard deviations, or for mutually    

'   dependent parameters with a known covariance matrix 

'   

'   Fourier transform and (de)convolution macros:

' 

' FT
a general-purpose Fourier transform macro for forward 

'   or inverse Fourier transformation of 2^n data where n is 

'   an integer larger than 2

' (De)convolve

general convolution and deconvolution. 

'   The convolution macro is generally applicable, the

'   deconvolution macro is not 

' (De)ConvolveFT
convolution and deconvolution based on 

'   Fourier transformation

' DeconvolveIt

iterative (van Cittert) deconvolution

'   DeconvolveIt0 has no constraints, DeconvolveIt1 assumes

'   that the function is non-negative everywhere

' Gabor

a macro providing time-frequency analysis

'   

'   Miscellaneous macros:

' 

' Semi-integrate & semi-differentiate
   two small macros 

'   for cyclic voltammetry, assuming planar diffusion 

' MovieDemos

' SolverScan
lets Solver scan a two-dimensional array

'   of parameter values

' Mapper
generates colored or gray-scale 2-D maps

' RootFinder
finds a single root by bisection

' ColumnSolver
applies Solver to column-organized data 

' 

' PURPOSE
'

' The macros in this MacroBundle are primarily offered 

' as examples of macro writing. They can also be used as  

' such in scientific data analysis. Moreover, they can 

' be modified by the user (under his responsibility,  

' at his own risk, and subject to the conditions 

' specified below) for more specialized applications

' 

' WARRANTY NOTICE
'

' Because of the inherent complexity of this software,

' and the consequent likelihood of undetected errors, the

' programs in this MacroBundle are made available without

' any warranty whatsoever. There is no guarantee that these

' programs will work, or will be error-free. Always verify

' your answers independently if decisions of possible

' consequence will be based on them. Neither implicit nor

' explicit warranty is made regarding this software,

' including (but not limited to) implied warranties of

' fitness for any particular purpose or application.

'

' COPYRIGHT NOTICE
'

' The material in this MacroBundle is copyrighted.

' Commercial use without specific, written permission from

' the copyright holder is strictly prohibited. However,

' permission is hereby granted for private or educational

' use, provided that such use is strictly non-commercial

' and/or is neither tied to nor combined with any sale,

' rental, advertisement, promotion, or other commercial 

' activity.

'

' It will be appreciated if, where appropriate, reference

' is made to this software by listing the web site http://

' www.oup-usa.org/advancedexcel where it can be found, and

' /or the book "Advanced Excel for scientific data analy-

' sis" where these macros and functions are explained and

' illustrated.

'

' INSTALLATION
'

' The macros in this MacroBundle are written in VBA (Visual

' BASIC for Applications) and will not work in versions of

' Excel preceding Excel 5, because those early versions did 

' not use VBA as their macro language.

'

' The installation procedure is slightly different for

' Excel 5 and Excel 95 on the one hand, and for more

' recent versions on the other. However, all such instal-

' lation procedures start with selecting this text (e.g.,

' with Edit -> SelectAll), copying it to the clipboard,

' and subsequently opening Excel. Thereafter follow the

' instructions under either (A) or (B) as appropriate.

'

' (A) For installation in Excel 97 or more recent versions,

' use the function key F11 or Tools -> Macro -> Visual

' Basic Editor, followed by (in the VBEditor menu) Insert

' -> Module. Then paste the MacroBundle into that module.

'

' (B) For installation in Excel 5 or Excel 95, after Excel

' has been opened, use Insert -> Macro -> Module to open a

' module, then paste the MacroBundle text from the clip-

' board into the module. Note: most but not all of these

' programs have been tested to run properly in Excel 5 and

' Excel 95. Therefore, some may occasionally use instruc-

' tions that were not included in these versions of Excel.

'

' With the above you will have access to all macros of this

' MacroBundle via Alt+F8 (Mac: Opt+F8) or Tools -> Macro ->

' Macros. For more convenient access, first install the

' MacroBundle Toolbar or the MacroBundle Menu. To install,

' e.g., the Toolbar, select Tools, Macro, Macros, then

' double-click on InstallMBToolbar. For the Menu, instead

' double-click on InstallMBMenu. There is no benefit in

' installling both. The Toolbar is the more convenient, 

' the Menu takes up less space but requires an additional 

' keystroke anytime a macro is selected.

'

' The MacroBundle Toolbar or MacroBundle Menu can be saved 

' with your spreadsheet, and they can be removed with 

' RemoveMBToolbar and RemoveMBMenu respectively. You can

' also add your own custom macros to that Toolbar or Menu.

'

' If you want these macros to be available every time you

' open a spreadsheet, you can incorporate them in the

' Personal.xls file, which is automatically opened whenever

' you open Excel. However, only place well-tested macros in

' Personal.xls. A poor instruction during macro develop-

' ment, if done in Personal.xls, may get it to 'hang up',

' in which case you may need expert help to extricate you.

'

' On any given line, all text to the right of an apostrophe

' (such as in this instroduction) is considered a comment,

' and is therefore ignored by the VBEditor. Consequently

' it is not necessary to remove them.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^      MACROBUNDLE UPDATE      ^'''''''''''''

''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

'''''''''''''''''''''''''''''''' version 4.4, April 1, 2004

'
' The following updates have been made in the MacroBundle 

' as of March 1, 2004:

' *  A new macro, ColumnSolver, has been added. It applies 

' Solver sequentially to all cells in a given column rather 

' than to just one cell. It takes one target cell, one or 

' more parameter cells, and (for EqualTo) one parameter 

' value at a time, and is convenient for, e.g., inverse 

' interpolation of algebraic functions. Like SolverScan, it 

' requires that Solver.xla has been activated, so that the 

' macro can call Solver.

' *  RootFinder now has two input boxes, allowing input 

' from two independent columns, one containing the function 

' F(x), the other the parameter x. 

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''^     INSERT MACROBUNDLE TOOLBAR    ^'''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''v.4 March 1, 2004

' PURPOSE:

'

' This subroutine places an extra toolbar in the Excel

' spreadsheet in order to facilitate access to the custom

' macros of this MacroBundle.

'

' SUBROUTINES:

'

' All the custom macros of the MacroBundle are callable

' with this toolbar. It also establishes the shortcut hot-

' keys to call these macros directly from the keyboard with

' Alt or / followed by the underlined character(s).

'

' LIMITATIONS:

'

' This subroutine works in Excel 97 and in more recent

' versions of Excel; it will not work in Excel 95 or

' earlier versions, which handled toolbars differently.

'

' NOTE:

'

' It is possible to insert a pull-down menu in the standard

' toolbar also, using the macro InsertMBMenu. Although it

' does not lead to any malfunction, there is no advantage

' to installing both the extra menu in the menu bar, and

' the extra toolbar.

Sub InsertMBToolbar()

Dim TBar As CommandBar

Dim Button1 As CommandBarButton

Dim Button2 As CommandBarPopup

Dim Button20 As CommandBarButton

Dim Button21 As CommandBarButton

Dim Button3 As CommandBarPopup

Dim Button30 As CommandBarButton

Dim Button31 As CommandBarButton

Dim Button4 As CommandBarPopup

Dim Button40 As CommandBarButton

Dim Button41 As CommandBarButton

Dim Button5 As CommandBarPopup

Dim Button50 As CommandBarButton

Dim Button51 As CommandBarButton

Dim Button6 As CommandBarPopup

Dim Button60 As CommandBarButton

Dim Button61 As CommandBarButton

Dim Button7 As CommandBarPopup

Dim Button70 As CommandBarButton

Dim Button71 As CommandBarButton

Dim Button8 As CommandBarButton

Dim Button9 As CommandBarPopup

Dim Button90 As CommandBarButton

Dim Button91 As CommandBarButton

Dim Button92 As CommandBarButton

Dim Button10 As CommandBarButton

Dim Button11 As CommandBarPopup

Dim Button110 As CommandBarButton

Dim Button111 As CommandBarButton

Dim Button12 As CommandBarPopup

Dim Button120 As CommandBarButton

Dim Button121 As CommandBarButton

Dim Button13 As CommandBarPopup

Dim Button130 As CommandBarButton

Dim Button131 As CommandBarButton

Dim Button14 As CommandBarPopup

Dim Button140 As CommandBarButton

Dim Button141 As CommandBarButton

Dim Button15 As CommandBarButton

Dim Button16 As CommandBarPopup

Dim Button160 As CommandBarButton

Dim Button161 As CommandBarButton

Dim Button162 As CommandBarButton

Dim Button163 As CommandBarButton

Dim Button17 As CommandBarButton

' Delete earlier version of MacroBundle,

' if existing, to prevent conflicts

On Error Resume Next

CommandBars("MacroBundle").Delete

' Create a commandbar

Set TBar = CommandBars.Add

With TBar

  .Name = "MacroBundle"

  .Position = msoBarTop

  .Visible = True

End With

' Create Button1 for &Propagation

Set Button1 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlButton)

With Button1

  .Caption = "&Propagation"

  .Style = msoButtonCaption

  .OnAction = "Propagation"

End With

' Create Button2 for &LS

Set Button2 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

With Button2

  .Caption = "&LS"

  .TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & "LS0 or LS1"

  .BeginGroup = True

End With

' Create submenus for LS&0 and LS&1 respectively

Set Button20 = Button2.Controls.Add(Type:=msoControlButton)

With Button20

  .Caption = "LS&0"

  .Style = msoButtonCaption

  .OnAction = "LS0"

End With

Set Button21 = Button2.Controls.Add(Type:=msoControlButton)

With Button21

  .Caption = "LS&1"

  .Style = msoButtonCaption

  .OnAction = "LS1"

End With

' Create Button3 for &WLS

Set Button3 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button3.Caption = "&WLS"

Button3.TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & "WLS0 or WLS1"

' Create submenus for WLS&0 and WLS&1 respectively

Set Button30 = Button3.Controls.Add(Type:=msoControlButton)

Button30.Caption = "WLS&0"

Button30.OnAction = "WLS0"

Set Button31 = Button3.Controls.Add(Type:=msoControlButton)

Button31.Caption = "WLS&1"

Button31.OnAction = "WLS1"

' Create Button4 for &ELS

Set Button4 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button4.Caption = "&ELS"

' Create submenus for ELS&auto and ELS&fixed respectively

Set Button40 = Button4.Controls.Add(Type:=msoControlButton)

Button40.Caption = "ELS&auto"

Button40.OnAction = "ELSauto"

Set Button41 = Button4.Controls.Add(Type:=msoControlButton)

Button41.Caption = "ELS&fixed"

Button41.OnAction = "ELSfixed"

' Create Button5 for LSMul&ti

Set Button5 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

With Button5

  .Caption = "LSMul&ti"

  .BeginGroup = True

  .TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & _

    "LSMulti0 or LSMulti1"

End With

' Create submenus for LSMulti&0 and LSMulti&1 respectively

Set Button50 = Button5.Controls _

  .Add(Type:=msoControlButton)

Button50.Caption = "LSMulti&0"

Button50.OnAction = "LSMulti0"

Set Button51 = Button5.Controls.Add(Type:=msoControlButton)

Button51.Caption = "LSMulti&1"

Button51.OnAction = "LSMulti1"

' Create Button6 for LSPol&y

Set Button6 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button6.Caption = "LSPol&y"

Button6.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "LSPoly0 or LSPoly1"

' Create submenus for LSPoly&0 and LSPoly&1 respectively

Set Button60 = Button6.Controls.Add(Type:=msoControlButton)

Button60.Caption = "LSPoly&0"

Button60.OnAction = "LSPoly0"

Set Button61 = Button6.Controls.Add(Type:=msoControlButton)

Button61.Caption = "LSPoly&1"

Button61.OnAction = "LSPoly1"

' Create Button7 for &Ortho

Set Button7 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button7.Caption = "&Ortho"

Button7.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "Ortho0 or Ortho1"

' Create submenus for Ortho&0 and Ortho&1 respectively

Set Button70 = Button7.Controls.Add(Type:=msoControlButton)

Button70.Caption = "Ortho&0"

Button70.OnAction = "Ortho0"

Set Button71 = Button7.Controls.Add(Type:=msoControlButton)

Button71.Caption = "Ortho&1"

Button71.OnAction = "Ortho1"

' Create Button8 for Solver&Aid

Set Button8 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlButton)

With Button8

  .Caption = "Solver&Aid"

  .Style = msoButtonCaption

  .BeginGroup = True

  .OnAction = "SolverAid"

End With

' Create Button9 for ColSolver

Set Button9 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button9.Caption = "Col&umnSolver"

' Create submenus for ColSolverM&ax, ColSolverM&in,

' and ColSolver&Value respectively

Set Button90 = Button9.Controls.Add(Type:=msoControlButton)

Button90.Caption = "ColumnSolverM&ax"

Button90.OnAction = "ColumnSolverMax"

Set Button91 = Button9.Controls.Add(Type:=msoControlButton)

Button91.Caption = "ColumnSolverM&in"

Button91.OnAction = "ColumnSolverMin"

Set Button92 = Button9.Controls.Add(Type:=msoControlButton)

Button92.Caption = "ColumnSolver&Value"

Button92.OnAction = "ColumnSolverValue"

' Create Button10 for Solver&Scan

Set Button10 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlButton)

With Button10

  .Caption = "Solver&Scan"

  .Style = msoButtonCaption

  .OnAction = "SolverScan"

End With

' Create Button11 for &FT

Set Button11 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

With Button11

  .Caption = "&FT"

  .BeginGroup = True

  .TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & "ForwardFT or" & _

    Chr(13) & "InverseFT"

End With

' Create submenus for &ForwardFT

' and &InverseFT respectively

Set Button110 = _

  Button11.Controls.Add(Type:=msoControlButton)

With Button110

  .Caption = "&ForwardFT"

  .Style = msoButtonCaption

  .OnAction = "ForwardFT"

End With

Set Button111 = _

  Button11.Controls.Add(Type:=msoControlButton)

With Button111

  .Caption = "&InverseFT"

  .Style = msoButtonCaption

  .OnAction = "InverseFT"

End With

' Create Button11 for (De)&Convolve

Set Button12 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

With Button12

  .Caption = "(De)&Convolve"

  .BeginGroup = True

  .TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & "Convolve or" & _

    Chr(13) & "Deconvolve"

End With

' Create submenus for &Convolve

' and &Deconvolve respectively

Set Button120 = _

  Button12.Controls.Add(Type:=msoControlButton)

With Button120

  .Caption = "&Convolve"

  .Style = msoButtonCaption

  .OnAction = "Convolve"

End With

Set Button121 = _

  Button12.Controls.Add(Type:=msoControlButton)

With Button121

  .Caption = "&Deconvolve"

  .Style = msoButtonCaption

  .OnAction = "Deconvolve"

End With

' Create Button13 for (De)Con&volveFT

Set Button13 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button13.Caption = "(De)Con&volveFT"

Button13.TooltipText = "Highlight array" & Chr(13) & _

    "before pressing" & Chr(13) & "ConvolveFT or" & _

    Chr(13) & "DeconvolveFT"

' Create submenus for &ConvolveFT

' and &DeconvolveFT respectively

Set Button130 = Button13.Controls _

  .Add(Type:=msoControlButton)

Button130.Caption = "&ConvolveFT"

Button130.OnAction = "ConvolveFT"

Set Button131 = Button13.Controls _

  .Add(Type:=msoControlButton)

With Button131

  .Caption = "&DeconvolveFT"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveFT"

End With

' Create Button14 for Deconvolve&It

Set Button14 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button14.Caption = "Deconvolve&It"

Button14.TooltipText = "Highlight array" & Chr(13) _

  & "before pressing" & Chr(13) & "DeconvolveIt0," _

  & Chr(13) & "or DeconvolveIt1."

' Create submenus for DeconvolveIt&0 and DeconvolveIt&1

Set Button140 = Button14.Controls _

  .Add(Type:=msoControlButton)

With Button140

  .Caption = "DeconvolveIt&0"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveIt0"

End With

Set Button141 = Button14.Controls _

  .Add(Type:=msoControlButton)

With Button141

  .Caption = "DeconvolveIt&1"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveIt1"

End With

' Create Button15 for &Gabor

Set Button15 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlButton)

With Button15

  .Caption = "&Gabor"

  .BeginGroup = True

  .Style = msoButtonCaption

  .OnAction = "Gabor"

End With

' Create Button16 for &Mapper

Set Button16 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlPopup)

Button16.TooltipText = "Select Mapper0 for grayscale" _

  & Chr(13) & "or either Mapper1, Mapper2" & Chr(13) & _

  "     or Mapper3 for color"

Button16.Caption = "&Mapper"

Button16.BeginGroup = True

' Create submenus for Mapper&0 through Mapper&3

Set Button160 = Button16.Controls _

  .Add(Type:=msoControlButton)

With Button160

  .Caption = "Mapper&0"

  .Style = msoButtonCaption

  .OnAction = "Mapper0"

End With

Set Button161 = Button16.Controls _

  .Add(Type:=msoControlButton)

With Button161

  .Caption = "Mapper&1"

  .Style = msoButtonCaption

  .OnAction = "Mapper1"

End With

Set Button162 = Button16.Controls _

  .Add(Type:=msoControlButton)

With Button162

  .Caption = "Mapper&2"

  .Style = msoButtonCaption

  .OnAction = "Mapper2"

End With

Set Button163 = Button16.Controls _

  .Add(Type:=msoControlButton)

With Button163

  .Caption = "Mapper&3"

  .Style = msoButtonCaption

  .OnAction = "Mapper3"

End With

' Create Button17 for &RootFinder

Set Button17 = CommandBars("MacroBundle").Controls _

  .Add(Type:=msoControlButton)

With Button17

  .Caption = "&RootFinder"

  .BeginGroup = True

  .Style = msoButtonCaption

  .OnAction = "RootFinder"

End With

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub RemoveMBToolbar()

On Error Resume Next

CommandBars("MacroBundle").Delete

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^    INSERT MACROBUNDLE MENU    ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''v.4 March 1, 2004

' PURPOSE:

'

' This subroutine places an extra pull-down menu in the

' Excel menu bars, between Tools and Data (or Chart), in

' order to make it easier to access the custom macros of

' the MacroBundle.

'

' SUBROUTINES:

'

' Many custom macros of the MacroBundle are callable with

' this extra menu. It also establishes the shortcut hotkeys

' to call these macros directly from the keyboard with Alt

' or / followed by the underlined character(s).

'

' LIMITATIONS:

'

' This subroutine works for Excel 97 and more recent ver-

' sions; it will not work in Excel 5 or Excel 95, which

' handle menus quite differently. For example, in Excel 95

' you can insert menu items using the menu editor, but not

' with VBA.

'

' This macro inserts the extra CustomMacros menu in the

' standard toolbar, CommandBars(1), between Tools and

' Data, but NOT in the chart toolbar, CommandBars(2).

'

' NOTE:

'

' The InsertMBMenu macro is offered as an alternative to

' the InsertMBToolbar macro. Although it does not lead to

' any malfunction, there is no advantage to having both

' the extra menu in the menu bar, and the extra toolbar,

' available simultaneously in the same workbook.

Sub InsertMBMenu()

Dim Menu As CommandBarPopup

Dim Menu0 As CommandBarControl

Dim Menu1 As CommandBarControl

Dim Menu2 As CommandBarPopup

Dim Menu20 As CommandBarControl

Dim Menu21 As CommandBarControl

Dim Menu3 As CommandBarPopup

Dim Menu30 As CommandBarControl

Dim Menu31 As CommandBarControl

Dim Menu4 As CommandBarPopup

Dim Menu40 As CommandBarControl

Dim Menu41 As CommandBarControl

Dim Menu5 As CommandBarPopup

Dim Menu50 As CommandBarControl

Dim Menu51 As CommandBarControl

Dim Menu6 As CommandBarPopup

Dim Menu60 As CommandBarControl

Dim Menu61 As CommandBarControl

Dim Menu7 As CommandBarPopup

Dim Menu70 As CommandBarControl

Dim Menu71 As CommandBarControl

Dim Menu8 As CommandBarControl

Dim Menu9 As CommandBarPopup

Dim Menu90 As CommandBarControl

Dim Menu91 As CommandBarControl

Dim Menu92 As CommandBarControl

Dim Menu10 As CommandBarControl

Dim Menu11 As CommandBarPopup

Dim Menu110 As CommandBarControl

Dim Menu111 As CommandBarControl

Dim Menu12 As CommandBarPopup

Dim Menu120 As CommandBarControl

Dim Menu121 As CommandBarControl

Dim Menu13 As CommandBarPopup

Dim Menu130 As CommandBarControl

Dim Menu131 As CommandBarControl

Dim Menu14 As CommandBarPopup

Dim Menu140 As CommandBarControl

Dim Menu141 As CommandBarControl

Dim Menu15 As CommandBarControl

Dim Menu16 As CommandBarPopup

Dim Menu160 As CommandBarControl

Dim Menu161 As CommandBarControl

Dim Menu162 As CommandBarControl

Dim Menu163 As CommandBarControl

Dim Menu17 As CommandBarControl

' Delete possible earlier menu insertions to prevent

' conflicts and position the new menu on the Menu bar

' just before Data

On Error Resume Next

CommandBars(1).Controls("Custom&Macros").Delete

On Error Resume Next

Set Menu0 = CommandBars(1).FindControl(ID:=30011)

Set Menu = CommandBars(1).Controls _

  .Add(Type:=msoControlPopup, _

  Before:=Menu0.Index, Temporary:=True)

Menu.Caption = "Custom&Macros"

' Create a menu item for Propagation

Set Menu1 = Menu.Controls.Add(Type:=msoControlButton)

Menu1.Caption = "&Propagation "

Menu1.OnAction = "Propagation"

' Create a menu item for LS

Set Menu2 = Menu.Controls.Add(Type:=msoControlPopup)

Menu2.Caption = "&LS "

' Create submenus for LS0 and LS1 respectively

Set Menu20 = Menu2.Controls.Add(Type:=msoControlButton)

Menu20.Caption = "LS&0"

Menu20.OnAction = "LS0"

Set Menu21 = Menu2.Controls.Add(Type:=msoControlButton)

Menu21.Caption = "LS&1"

Menu21.OnAction = "LS1"

' Create a menu item for WLS

Set Menu3 = Menu.Controls.Add(Type:=msoControlPopup)

Menu3.Caption = "&WLS"

' Create submenus for WLS0 and WLS1 respectively

Set Menu30 = Menu3.Controls.Add(Type:=msoControlButton)

Menu30.Caption = "WLS&0"

Menu30.OnAction = "WLS0"

Set Menu31 = Menu3.Controls.Add(Type:=msoControlButton)

Menu31.Caption = "WLS&1"

Menu31.OnAction = "WLS1"

' Create Button4 for ELS

Set Menu4 = Menu.Controls.Add(Type:=msoControlPopup)

Menu4.Caption = "&ELS"

' Create submenus for ELSauto and ELSfixed respectively

Set Menu40 = Menu4.Controls.Add(Type:=msoControlButton)

Menu40.Caption = "ELS&auto"

Menu40.OnAction = "ELSauto"

Set Menu41 = Menu4.Controls.Add(Type:=msoControlButton)

Menu41.Caption = "ELS&fixed"

Menu41.OnAction = "ELSfixed"

' Create Menu5 for LSMulti

Set Menu5 = Menu.Controls.Add(Type:=msoControlPopup)

Menu5.Caption = "LSMul&ti"

Menu5.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "LSMulti0 or LSMulti1"

' Create submenus for LSMulti0 and LSMulti1 respectively

Set Menu50 = Menu5.Controls.Add(Type:=msoControlButton)

Menu50.Caption = "LSMulti&0"

Menu50.OnAction = "LSMulti0"

Set Menu51 = Menu5.Controls.Add(Type:=msoControlButton)

Menu51.Caption = "LSMulti&1"

Menu51.OnAction = "LSMulti1"

' Create Menu6 for LSPoly

Set Menu6 = Menu.Controls.Add(Type:=msoControlPopup)

Menu6.Caption = "LSPol&y"

Menu6.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "LSPoly0 or LSPoly1"

' Create submenus for LSPoly0 and LSPoly1 respectively

Set Menu60 = Menu6.Controls.Add(Type:=msoControlButton)

Menu60.Caption = "LSPoly&0"

Menu60.OnAction = "LSPoly0"

Set Menu61 = Menu6.Controls.Add(Type:=msoControlButton)

Menu61.Caption = "LSPoly&1"

Menu61.OnAction = "LSPoly1"

' Create Menu7 for Ortho

Set Menu7 = Menu.Controls.Add(Type:=msoControlPopup)

Menu7.Caption = "&Ortho"

Menu7.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "Ortho0 or Ortho1"

' Create submenus for Ortho0 and Ortho1 respectively

Set Menu70 = Menu7.Controls.Add(Type:=msoControlButton)

Menu70.Caption = "Ortho&0"

Menu70.OnAction = "Ortho0"

Set Menu71 = Menu7.Controls.Add(Type:=msoControlButton)

Menu71.Caption = "Ortho&1"

Menu71.OnAction = "Ortho1"

' Create Menu8 for SolverAid

Set Menu8 = Menu.Controls.Add(Type:=msoControlButton)

Menu8.Caption = "Solver&Aid"

Menu8.OnAction = "SolverAid"

' Create Menu9 for ColSolver

Set Menu9 = Menu.Controls.Add(Type:=msoControlPopup)

Menu9.Caption = "Col&umnSolver"

' Create submenus for ColSolverMax, ColSolverMin, and

' ColSolverValue respectively

Set Menu90 = Menu9.Controls.Add(Type:=msoControlButton)

Menu90.Caption = "ColumnSolverM&ax"

Menu90.OnAction = "ColumnSolverMax"

Set Menu91 = Menu9.Controls.Add(Type:=msoControlButton)

Menu91.Caption = "ColumnSolverM&in"

Menu91.OnAction = "ColumnSolverMin"

Set Menu92 = Menu9.Controls.Add(Type:=msoControlButton)

Menu92.Caption = "ColumnSolver&Value"

Menu92.OnAction = "ColumnSolverValue"

' Create Menu10 for SolverScan

Set Menu10 = Menu.Controls.Add(Type:=msoControlButton)

Menu10.Caption = "Solver&Scan"

Menu10.OnAction = "SolverScan"

' Create Menu11 for FT

Set Menu11 = Menu.Controls.Add(Type:=msoControlPopup)

Menu11.Caption = "&FT"

Menu11.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "ForwardFT or" & _

  Chr(13) & "InverseFT"

' Create submenus for ForwardFT and InverseFT respectively

Set Menu110 = Menu11.Controls.Add(Type:=msoControlButton)

With Menu110

  .Caption = "&ForwardFT"

  .Style = msoButtonCaption

  .OnAction = "ForwardFT"

End With

Set Menu111 = Menu11.Controls.Add(Type:=msoControlButton)

With Menu111

  .Caption = "&InverseFT"

  .Style = msoButtonCaption

  .OnAction = "InverseFT"

End With

' Create Menu12 for (De)Convolve

Set Menu12 = Menu.Controls.Add(Type:=msoControlPopup)

Menu12.Caption = "(De)&Convolve"

Menu12.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "Convolve or" & _

  Chr(13) & "Deconvolve"

' Create submenus for Convolve and Deconvolve respectively

Set Menu120 = Menu12.Controls.Add(Type:=msoControlButton)

With Menu120

  .Caption = "&Convolve"

  .Style = msoButtonCaption

  .OnAction = "Convolve"

End With

Set Menu121 = Menu12.Controls.Add(Type:=msoControlButton)

With Menu121

  .Caption = "&Deconvolve"

  .Style = msoButtonCaption

  .OnAction = "Deconvolve"

End With

' Create Menu13 for ConvolveFT

Set Menu13 = Menu.Controls.Add(Type:=msoControlPopup)

Menu13.Caption = "Con&volveFT"

Menu13.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "ConvolveFT or" & _

  Chr(13) & "DeconvolveFT"

' Create submenus for ConvolveFT

' and DeconvolveFT respectively

Set Menu130 = Menu13.Controls.Add(Type:=msoControlButton)

With Menu130

  .Caption = "&ConvolveFT"

  .Style = msoButtonCaption

  .OnAction = "ConvolveFT"

End With

Set Menu131 = Menu13.Controls.Add(Type:=msoControlButton)

With Menu131

  .Caption = "&DeconvolveFT"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveFT"

End With

' Create Menu14 for DeConvolveIt

Set Menu14 = Menu.Controls.Add(Type:=msoControlPopup)

Menu14.Caption = "Deconvolve&It"

Menu14.TooltipText = "Highlight array" & Chr(13) & _

  "before pressing" & Chr(13) & "DeconvolveIt0" _

  & Chr(13) & "or DeconvolveIt1."

' Create submenus for DeconvolveIt0

' and DeconvolveIt1 respectively

Set Menu140 = Menu14.Controls.Add(Type:=msoControlButton)

With Menu140

  .Caption = "DeconvolveIt&0"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveIt0"

End With

Set Menu141 = Menu14.Controls.Add(Type:=msoControlButton)

With Menu141

  .Caption = "&DeconvolveIt&1"

  .Style = msoButtonCaption

  .OnAction = "DeconvolveIt1"

End With

' Create Menu15 for Gabor

Set Menu15 = Menu.Controls.Add(Type:=msoControlButton)

With Menu15

  .Caption = "&Gabor"

  .Style = msoButtonCaption

  .OnAction = "Gabor"

End With

' Create Menu16 for Mapper

Set Menu16 = Menu.Controls.Add(Type:=msoControlPopup)

Menu16.Caption = "&Mapper"

' Create submenus for Mapper

Set Menu160 = Menu16.Controls.Add(Type:=msoControlButton)

With Menu160

  .Caption = "Mapper&0"

  .Style = msoButtonCaption

  .OnAction = "Mapper0"

End With

Set Menu161 = Menu16.Controls.Add(Type:=msoControlButton)

With Menu161

  .Caption = "Mapper&1"

  .Style = msoButtonCaption

  .OnAction = "Mapper1"

End With

Set Menu162 = Menu16.Controls.Add(Type:=msoControlButton)

With Menu162

  .Caption = "Mapper&2"

  .Style = msoButtonCaption

  .OnAction = "Mapper2"

End With

Set Menu163 = Menu16.Controls.Add(Type:=msoControlButton)

With Menu163

  .Caption = "Mapper&3"

  .Style = msoButtonCaption

  .OnAction = "Mapper3"

End With

' Create Menu17 for RootFinder

Set Menu17 = Menu.Controls.Add(Type:=msoControlButton)

With Menu17

  .Caption = "&RootFinder"

  .Style = msoButtonCaption

  .OnAction = "RootFinder"

End With

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub RemoveMBMenu()

On Error Resume Next

CommandBars(1).Controls("Custom&Macros").Delete

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''^                            ^'''''''''''''''

'''''''''''''^     LINEAR LEAST SQUARES     ^''''''''''''''

''''''''''''''^                            ^'''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

'

' The macros LS1 and LS0 compute the parameters and their

' standard deviations for an unweighted least squares fit

' to data in 2 or more columns. The macros set the input

' parameter p for the subroutine LeastSquares: p = 1 causes

' a general unweighted least squares fit to the data, while

' p = 0 forces the fit to pass through the origin, i.e., it

' assumes that y = 0 for x = 0.

'

' SUBROUTINES:

'

' This macro requires the subroutines Multiply, Invert, and

' Transpose

'

' INPUT:

'

' The input data must be organized in columns, arranged

' as follows. The first column must contain the dependent

' variable y. The second (and any subsequent) column(s)

' must contain the independent variable(s) x.

'

' OUTPUT:

'

' The macro labels the coefficients and standard devi-

' ations, except when there is no space available for

' labels because the first data column is at the left

' edge of the spreadsheet.

'

' PROCEDURE:

'

' Before calling the macro, make sure that the output area

' (two lines below the input data block) does not contain

' valuable data.

'

' In order to start the process, highlight the entire input

' data block, and call LS1 or LS0.

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

Sub LS0()           ' for an unweighted least squares

                    ' fit through the origin

Dim p As Double

p = 0

Call LeastSquares(p)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LS1()           ' for a general unweighted

                    ' least squares fit

Dim p As Double

p = 1

Call LeastSquares(p)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LeastSquares(p)

Dim cMax As Integer, i As Integer, j As Integer

Dim jj As Integer, m As Integer, n As Integer

Dim rMax As Integer, Root As Double, SSR As Double

Dim StDevY As Double, u As Double, varY As Double

Dim DataArray As Variant, outputArray As Variant

Dim lccArray As Variant, vArray As Variant

Dim v0Array As Variant

Dim myRange As Range

Dim aa, AC, Answer, hAnswer, jAnswer

Dim bArray, btArray, btqArray

Dim M1, M2, pArray, piArray, qArray

Dim XArray, xtArray, YArray, ytArray, ytyArray

' Determination of the array size:

Begin:

rMax = Selection.Rows.Count

cMax = Selection.Columns.Count

u = 1

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Least Squares Fit")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least 2:

If cMax < 2 Then

  MsgBox "There must be at least two columns," & _

    Chr(13) & "one for Y, and one or more for X.", _

    , "Least Squares Fit"

  End

End If

' Check that there are more data than coefficients :

If rMax - cMax - p + 1 <= 0 Then

  MsgBox "With " & rMax & " data, LS" & p & " can only deter-" & _

    Chr(13) & "mine " & rMax - 1 & " least squares coefficients." & _

    Chr(13) & Chr(13) & "Add more data, or reduce the re-" & _

    Chr(13) & "quested number of coefficients."

  End

End If

' Dimension the arrays:

ReDim YArray(1 To rMax, 1 To 1) As Double

ReDim XArray(1 To rMax, 1 To cMax) As Double

ReDim ytArray(1 To 1, 1 To rMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To cMax, 1 To rMax) As Double

ReDim pArray(1 To cMax, 1 To cMax) As Double

ReDim piArray(1 To cMax, 1 To cMax) As Double

ReDim qArray(1 To cMax, 1 To 1) As Double

ReDim bArray(1 To cMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To cMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim M1(1 To rMax, 1 To rMax) As Double

ReDim M2(1 To rMax, 1 To rMax) As Double

ReDim vArray(1 To cMax, 1 To cMax) As Double

ReDim v0Array(1 To cMax - 1 + p, 1 To cMax - 1 + p) _

  As Double

ReDim lccArray(1 To cMax, 1 To cMax) As Double

' Read the dataArray, then fill yArray and xArray.

DataArray = Selection.Value

For i = 1 To rMax

  YArray(i, 1) = DataArray(i, 1)

Next i

For i = 1 To rMax

  If IsEmpty(DataArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "Least Squares Fit"

    End

  End If

Next i

For j = 2 To cMax

  For i = 1 To rMax

    If IsEmpty(DataArray(i, j)) Then

      MsgBox "X-value(s) missing", , "Least Squares Fit"

      End

    End If

  Next i

Next j

' Fill the first column of xArray with zeroes (for p = 0)

' or ones (for p = 1), the rest with the data in the

' x-column(s)

For i = 1 To rMax

  XArray(i, 1) = CDbl(p)

Next i

For j = 2 To cMax

  For i = 1 To rMax

    XArray(i, j) = DataArray(i, j)

  Next i

Next j

' Compute b = (X' X)" X' Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows,

' columns) are:

'   Y          = yArray        (  rmax,     1)

'   X          = xArray        (  rmax,  cmax)

'   X'         = xtArray       (  cmax,  rmax)

'   X' X       = pArray        (  cmax,  cmax)

'  (X' X)"     = piArray       (  cmax,  cmax)

'   X' Y       = qArray        (  cmax,     1)

'   b          = bArray        (  cmax,     1)

Call Transpose(XArray, rMax, cMax, xtArray)

Call Multiply(xtArray, cMax, rMax, XArray, cMax, pArray)

Call Invert(pArray, cMax, piArray)

Call Multiply(xtArray, cMax, rMax, YArray, 1, qArray)

Call Multiply(piArray, cMax, cMax, qArray, 1, bArray)

' Check against overwriting spreadsheet data

m = 0

If (p = 0 And cMax = 2) Then

  For i = 1 To 3

    Selection.Offset(1, 0).Select

    outputArray = Selection.Value

    For j = 1 To cMax

      If IsEmpty(outputArray(rMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-3, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & "three lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

Else

  For i = 1 To 2 + p + cMax

    Selection.Offset(1, 0).Select

    outputArray = Selection.Value

    For j = 1 To cMax

      If IsEmpty(outputArray(rMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-2 - p - cMax, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & 2 + p + cMax & " lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

End If

' The additional arrays and their dimensions (rows,

' columns) are:

'   Y'         = ytArray       (     1,  rmax)

'   Y' Y       = ytyArray      (     1,     1)

'   b'         = btArray       (     1,  cmax)

'   b' X' Y    = btqArray      (     1,     1)

Call Transpose(YArray, rMax, 1, ytArray)

Call Transpose(bArray, cMax, 1, btArray)

Call Multiply(ytArray, 1, rMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, cMax, qArray, 1, btqArray)

' Calculate SSR = Y'Y - b'X'Y; then the variance

' of y as varY = SSR/(rMax-cMax-p+1); and vArray,

' the covariance matrix, as V = (X'X)" times varY.

SSR = ytyArray(1, 1) - btqArray(1, 1)

varY = SSR / (rMax - cMax - p + 1)

StDevY = Sqr(Abs(varY))

For i = 1 To cMax

  For j = 1 To cMax

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(rMax, 0).Select

' Prepare the output format

For j = 1 To cMax

  ActiveCell.Font.Bold = True

  ActiveCell.Font.Italic = True

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -cMax).Select

If (p = 0 And cMax = 2) Then

  For i = 1 To 2

    For j = 1 To cMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -cMax).Select

  Next i

  ActiveCell.Offset(-3, 0).Select

Else

  For i = 1 To 1 + p + cMax

    For j = 1 To cMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -cMax).Select

  Next i

  ActiveCell.Offset(-2 - p - cMax, 0).Select

End If

' Prepare the output labels, suppressing them when space

' for them is unavailable or data will be overwritten

aa = ActiveCell.Address

AC = Mid(aa, 2, 1)

If (AC = "A" And p = 1) Then GoTo NoLabel

ActiveCell.Offset(0, -p).Select

If p = 1 Then

  If (IsEmpty(ActiveCell) Or ActiveCell.Value = "Coeff:") _

  Then

    GoTo Step1

  Else

    ActiveCell.Offset(0, p).Select

    GoTo NoLabel

  End If

End If

Step1:

With ActiveCell

  .Value = "Coeff:"

  .Font.Bold = True

  .Font.Italic = True

  .HorizontalAlignment = xlRight

End With

ActiveCell.Offset(1, 0).Select

If p = 1 Then

  If (IsEmpty(ActiveCell) Or ActiveCell.Value = "StDev:") _

  Then

    GoTo Step2

  Else

    ActiveCell.Offset(-1, p).Select

    GoTo NoLabel

  End If

End If

Step2:

With ActiveCell

  .Value = "StDev:"

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

End With

ActiveCell.Offset(-1, p).Select

If p = 0 And cMax = 2 Then GoTo NoLabel

ActiveCell.Offset(3, -p).Select

If p = 1 Then

  If (IsEmpty(ActiveCell) Or ActiveCell.Value = "CM:") Then

    GoTo Step3

  Else

    ActiveCell.Offset(-3, p).Select

    GoTo NoLabel

  End If

End If

Step3:

With ActiveCell

  .Value = "CM:"

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 11

  .HorizontalAlignment = xlRight

End With

ActiveCell.Offset(-3, p).Select

NoLabel:

ActiveCell.Offset(0, 1 - p).Select

For j = 2 - p To cMax

  ActiveCell.Value = bArray(j, 1)

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, 1 - p - cMax).Select

For j = 2 - p To cMax

  If vArray(j, j) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = Sqr(vArray(j, j))

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, 1 - p - cMax).Select

Selection.Value = StDevY

If p = 0 And cMax = 2 Then GoTo LastLine

ActiveCell.Offset(1, 0).Select

For i = 2 - p To cMax

  For j = 2 - p To cMax

    ActiveCell.Value = vArray(i, j)

    ActiveCell.Font.ColorIndex = 11

    ActiveCell.Offset(0, 1).Select

  Next j

  ActiveCell.Offset(1, 1 - p - cMax).Select

Next i

Application.ScreenUpdating = True

' Provide as optional output the array of linear

' correlation coefficients. The user specifies

' the cell block in which to write this array

If p = 0 And cMax = 2 Then GoTo LastLine

jAnswer = MsgBox("Do you want to see the " _

  & "matrix of linear correlation" _

  & Chr(13) & "coefficients? It will need a " _

  & "block of " & cMax + p - 1 _

  & " by " & cMax + p - 1 & " cells.", vbYesNo, _

  "Least Squares Fit")

OutlineMatrix:

If jAnswer = vbYes Then

  Set myRange = Application.InputBox(Prompt:= _

    "The array should be located in:", Type:=8)

  myRange.Select

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> cMax + p - 1 Then

    MsgBox "The selected range does not have " _

      & cMax + p - 1 & " rows. Please correct.", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

  If Selection.Columns.Count <> cMax + p - 1 Then

    MsgBox "The selected range does not have " _

      & cMax + p - 1 & " columns. Please correct", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

' Draw a box around the reserved area

    Selection.Borders(xlEdgeTop).LineStyle = xlDashDot

    Selection.Borders(xlEdgeTop).Weight = xlThin

    Selection.Borders(xlEdgeRight).LineStyle = xlDashDot

    Selection.Borders(xlEdgeRight).Weight = xlThin

    Selection.Borders(xlEdgeBottom).LineStyle = xlDashDot

    Selection.Borders(xlEdgeBottom).Weight = xlThin

    Selection.Borders(xlEdgeLeft).LineStyle = xlDashDot

    Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the array of linear correlation coefficients

  If p = 0 Then

    For i = 1 To cMax - 1

      For j = 1 To cMax - 1

        v0Array(i, j) = vArray(i + 1, j + 1)

      Next j

    Next i

    For i = 1 To cMax - 1

      For j = 1 To cMax - 1

        Root = Sqr(v0Array(i, i) * v0Array(j, j))

        lccArray(i, j) = v0Array(i, j) / Root

    Next j

    Next i

    Selection.Value = lccArray

  End If

  If p = 1 Then

    For i = 1 To cMax

      For j = 1 To cMax

        v0Array(i, j) = vArray(i, j)

      Next j

    Next i

    For i = 1 To cMax

      For j = 1 To cMax

        Root = Sqr(vArray(i, i) * vArray(j, j))

        lccArray(i, j) = vArray(i, j) / Root

      Next j

    Next i

    Selection.Value = lccArray

  End If

End If

LastLine:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

'''''''''''''''''^                         ^'''''''''''''''

''''''''''''''''^    LEAST SQUARES POLY     ^''''''''''''''

'''''''''''''''''^                         ^'''''''''''''''

''''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

'

' The macros LSPoly0 and LSPoly1 compute the parameters and

' their standard deviations for an unweighted least squares

' fit of a dependent variable y to a power series in x of

' increasing (positive integer) order m. The user specifies

' the highest order to be displayed.

'

' Use the custom macros Ortho1 or Ortho0 in order to decide

' what polynomial order might be optimal.

'

' SUBROUTINES:

'

' The macros LSPoly0 and LSPoly1 call the subroutine

' LeastSquaresPoly which, in turn, calls the subroutines

' Invert, Multiply, and Transpose.

'

' INPUT:

'

' The input data must be arranged in a block of two adja-

' cent columns, with the dependent variable y in the left-

' most column, and the independent variable x to its right.

' Highlight the data in these two columns, and call the

' macro LSPoly0 or LSPoly1. Then specify the maximum poly-

' nomial order for which you want the results displayed.

'

' OUTPUT:

'

' The macro provides:

' (1) the coefficients of the least squares fit of y to the

'     power series a(1)x + a(2)x^2 + ... = Sum[a(i)x^i] for

'     i = 1(1)m for LSPoly0, or to a(0) + a(1)x + a(2)x^2

'     + ... = Sum[a(i)x^i] for i = 0(1)m for LSPoly1;

' (2) the corresponding standard deviations s

'     for these coefficients;

' (3) the standard deviation of the fit

'     for each value of m;

' (4) the F-ratios for alpha = 0.05 and 0.01

'     for each value of m.

'

' For each order the output consists of three rows. The top

' row lists the order of the polynomial, and the standard

' deviation of the fit. The second row displays the coeffi-

' cients of the fitted polynomial, and the third row the

' corresponding standard deviations. The second and third

' rows also list the values for the F-ratios for 5% and 1%

' respectively. For easy reading the rows are color-coded

' by order.

'

' If you want the output to display more digits, change

' the instruction ActiveCell.NumberFormat = "0.00E+00" to

' ActiveCell.NumberFormat = "0.000E+00", "0.0000E+00", etc.

'

' WARNINGS:

'

' The output is displayed below the highlighted input

' array, and therefore might escape your notice. If

' the macro doesn't seem to respond, look below it.

'

' If you find cells filled with ###### the macro needs

' wider columns to display its results. Click on the

' column name(s) at the top of the column(s), right-click,

' and insert a larger number in Column Width.

'

' The size of the macro output will be determined by your

' choice of the maximum polynomial order. The output will

' be displayed under the input data, and will gradually

' expand to the right as results for higher orders occupy

' increasingly more space. The first-order output will

' occupy a block of 3 rows of 5 columns for LSPoly0 (6

' for LSPoly1), and each additional order will use a

' block that is one column wider than its predecessor.

' Please move valuable data from the output region before

' running this macro, lest those data be overwritten.

'

' PROCEDURE:

'

' Before calling the macro, make sure that the output area

' is clear of valuable data (see above under WARNING)

'

' In order to start the process, highlight the (two columns

' wide, and at least 3 rows high) input data block, then

' call LSPoly0 or LSPoly1.

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

'

Sub LSPoly0()           ' for unweighted least squares

                ' fit through the origin

Dim p As Double

p = 0

Call LeastSquaresPoly(p)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub LSPoly1()           ' for general unweighted

                ' least squares fit

Dim p As Double

p = 1

Call LeastSquaresPoly(p)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub LeastSquaresPoly(p)

Dim cMax As Integer, i As Integer, j As Integer

Dim m As Integer, n As Integer, rMax As Integer

Dim Color(1 To 7) As Integer

Dim MaxOrder As Integer, MaxOrderMax As Integer

Dim FRatio1 As Double, FRatio5 As Double

Dim Root As Double, StDevY As Double

Dim u As Double, z As Double

Dim DataArray As Variant, F As Variant, lccArray As Variant

Dim outputArray As Variant, SSR As Variant

Dim varY As Variant, vArray As Variant, v0Array As Variant

Dim myRange As Range

Dim bArray, btArray, btqArray, M1, M2, pArray, piArray

Dim qArray, XArray, xtArray, YArray, ytArray, ytyArray

Dim Answer, iAnswer, hAnswer, kAnswer

' Determination of the array size:

Begin:

rMax = Selection.Rows.Count

cMax = Selection.Columns.Count

ReDim F(1 To rMax, 1 To 1) As Double

ReDim SSR(1 To rMax, 1 To 1) As Double

u = 1

z = 0

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "LSPoly")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is 2:

If cMax <> 2 Then

  MsgBox "There must be two columns," & _

  Chr(13) & "one for Y, and one for X.", _

    , "Least Squares Polynomial Fit"

  End

End If

' Check that rmax > 3, so that the number of data

' points is sufficient to define the problem:

If rMax < 4 Then

  MsgBox "There must be at least 4 input" & _

  Chr(13) & "data pairs to define the problem.", _

    , "Least Squares Polynomial Fit"

  End

End If

MaxOrderMax = 14

If rMax < 16 Then MaxOrderMax = rMax - 2

DataArray = Selection.Value

Selection.Item(1, 1).Select

' Select the maximum order:

MaxOrder = InputBox _

  ("Select as maximum order an integer" & Chr(13) & _

  "between 3 and " & MaxOrderMax & "." & Chr(13) & _

  Chr(13) & "Enter the maximum order: ", "MaxOrder")

If MaxOrder < 3 Then MaxOrder = 3

If MaxOrder > MaxOrderMax Then MaxOrder = MaxOrderMax

RedimensionArrays:

' Dimension the arrays:

ReDim YArray(1 To rMax, 1 To 1) As Double

ReDim XArray(1 To rMax, 1 To cMax) As Double

ReDim ytArray(1 To 1, 1 To rMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To cMax, 1 To rMax) As Double

ReDim pArray(1 To cMax, 1 To cMax) As Double

ReDim piArray(1 To cMax, 1 To cMax) As Double

ReDim qArray(1 To cMax, 1 To 1) As Double

ReDim bArray(1 To cMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To cMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim M1(1 To rMax, 1 To rMax) As Double

ReDim M2(1 To rMax, 1 To rMax) As Double

ReDim vArray(1 To cMax, 1 To cMax) As Double

ReDim v0Array(1 To cMax - 1 + p, 1 To cMax - 1 + p) _

  As Double

ReDim lccArray(1 To cMax, 1 To cMax) As Double

' Read the dataArray, then fill yArray and xArray.

For i = 1 To rMax

  YArray(i, 1) = DataArray(i, 1)

Next i

If cMax = 2 Then

  For i = 1 To rMax

    If IsEmpty(DataArray(i, 1)) Then

      MsgBox "Y-value(s) missing", _

        , "Least Squares Polynomial Fit"

      End

    End If

  Next i

  For i = 1 To rMax

    If IsEmpty(DataArray(i, 2)) Then

      MsgBox "X-value(s) missing", _

        , "Least Squares Polynomial Fit"

      End

    End If

  Next i

End If

' Fill the first column of xArray with zeroes

' (for p = 0) or ones (for p = 1), the rest

' with the data in the x-column(s)

For i = 1 To rMax

  XArray(i, 1) = CDbl(p)

Next i

For i = 1 To rMax

  XArray(i, 2) = DataArray(i, 2)

Next i

If cMax > 2 Then

  For j = 3 To cMax

    For i = 1 To rMax

      XArray(i, j) = XArray(i, 2) * XArray(i, j - 1)

    Next i

  Next j

End If

' Compute b = (X' X)" X' Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows,

' columns) are:

'   Y          = yArray        (  rmax,     1)

'   X          = xArray        (  rmax,  cmax)

'   X'         = xtArray       (  cmax,  rmax)

'   X' X       = pArray        (  cmax,  cmax)

'  (X' X)"     = piArray       (  cmax,  cmax)

'   X' Y       = qArray        (  cmax,     1)

'   b          = bArray        (  cmax,     1)

'   Y'         = ytArray       (     1,  rmax)

'   Y' Y       = ytyArray      (     1,     1)

'   b'         = btArray       (     1,  cmax)

'   b' X' Y    = btqArray      (     1,     1)

Call Transpose(XArray, rMax, cMax, xtArray)

Call Multiply(xtArray, cMax, rMax, XArray, cMax, pArray)

Call Invert(pArray, cMax, piArray)

Call Multiply(xtArray, cMax, rMax, YArray, 1, qArray)

Call Multiply(piArray, cMax, cMax, qArray, 1, bArray)

Call Transpose(YArray, rMax, 1, ytArray)

Call Transpose(bArray, cMax, 1, btArray)

Call Multiply(ytArray, 1, rMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, cMax, qArray, 1, btqArray)

' Calculate SSR = Y'Y - b'X'Y, and then varY, the variance

' of y, as varY = SSR/(rmax-cmax-p+1); and vArray, the co-

' variance matrix, as V = (X'X)" times varY, of which we

' here only use the diagonal elements, i.e., the variances.

SSR(cMax, 1) = ytyArray(1, 1) - btqArray(1, 1)

varY = SSR(cMax, 1) / (rMax - cMax + 1 - p)

StDevY = Sqr(Abs(varY))

If cMax > 2 Then

  F(cMax, 1) = ((SSR(cMax - 1, 1) / SSR(cMax, 1)) - 1) _

    * (rMax - cMax + 1 - p)

  FRatio5 = (F(cMax, 1)) / _

    (Application.FInv(0.05, 1, rMax - cMax + 1 - p))

  FRatio1 = (F(cMax, 1)) / _

    (Application.FInv(0.01, 1, rMax - cMax + 1 - p))

End If

For i = 1 To cMax

  For j = 1 To cMax

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(rMax + 1, 0).Select

' Paint color bands for up to 14 orders, and set the

' numerical format to scientific with 3 decimal places

Color(1) = 38

Color(2) = 40

Color(3) = 36

Color(4) = 35

Color(5) = 34

Color(6) = 37

Color(7) = 39

For i = 1 To 3

  For j = 1 To cMax + p + 2

    ActiveCell.ClearContents

    If cMax < 9 Then

      ActiveCell.Interior.ColorIndex = Color(cMax - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(cMax - 8)

    End If

    ActiveCell.Offset(0, 1).Select

    If cMax < 9 Then

      ActiveCell.Interior.ColorIndex = Color(cMax - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(cMax - 8)

    End If

    ActiveCell.Font.Bold = False

    ActiveCell.Font.Italic = False

    ActiveCell.NumberFormat = "0.00E+00"

  Next j

  ActiveCell.Offset(1, -cMax - p - 2).Select

Next i

ActiveCell.Offset(-3, 0).Select

' Display the top line of the output

With ActiveCell

  .Font.Bold = True

  .Font.Italic = False

  .HorizontalAlignment = xlLeft

  .Value = "Order " & cMax - 1

End With

For j = 1 To cMax + p - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = True

    .HorizontalAlignment = xlCenter

    .Value = "term " & j - p

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Sy:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

Selection.Value = StDevY

Selection.Offset(1, -cMax - p - 2).Select

' Display the center line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Coeff:"

End With

For j = 1 To cMax + p - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

    .Value = bArray(j + 1 - p, 1)

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR5:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If cMax = 2 Then Selection.Value = "N/A"

If cMax > 2 Then Selection.Value = FRatio5

Selection.Offset(1, -cMax - p - 2).Select

' Display the bottom line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "StDev:"

End With

For j = 1 To cMax + p - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

  End With

  If vArray(j - p + 1, j - p + 1) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = _

      Sqr(Abs(vArray(j - p + 1, j - p + 1)))

  End If

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR1:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If cMax = 2 Then Selection.Value = "N/A"

If cMax > 2 Then Selection.Value = FRatio1

Selection.Offset(1, -cMax - p - 2).Select

If cMax < MaxOrder + 2 Then cMax = cMax + 1

ActiveCell.Offset(-rMax - 1, 0).Select

If cMax = MaxOrder + 2 Then End

GoTo RedimensionArrays

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

'''''''''''''''''^                         ^'''''''''''''''

''''''''''''''''^    LEAST SQUARES MULTI    ^''''''''''''''

'''''''''''''''''^                         ^'''''''''''''''

''''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

'

' The macros LSMulti0 and LSMulti1 compute the parameters

' and their standard deviations for an unweighted least

' squares fit of a dependent variable y to a user-specified

' multivariate expression of the form y = a1x1 + a2x2 + ..

' (for LSMulti0) or y = a0 + a1x1 + a2x2 + ...(for

' LSMulti1) by gradually (one term at a time) increasing

' the number of terms included in the least squares

' analysis from left to right

'

' Note that the macro does NOT rearrange the columns, and

' therefore does not provide all possible permutations.

' If those possible permutations are needed, instead use

' the macros LSPermute0 or LSPermute1.

'

' The macro yields: the fitting coefficients a(i), their

' standard deviations s(i), the standard deviation of the

' over-all fit, s(y), and the results of two F-tests, FR5

' and FR1, for 5% and 1% probability respectively.

'

' SUBROUTINES:

'

' The macros LSMulti0 and LSMulti1 call the subroutine

' LSMulti which, in turn, calls the subroutines Invert,

' Multiply, and Transpose.

'

' INPUT:

'

' The input data must be arranged in a block of at least 3

' (for LSMulti0) or 4 (for LSMulti1) adjacent columns, with

' the dependent variable y in the left-most column, and the

' independent variables x to its right. Highlight the data

' in these columns, and call the macro LSMulti0 or LSMulti1

'

' OUTPUT:

'

' The macro provides:

' (1) the coefficients of the least squares fit of y to

'     the series a(1)x(1) + a(2)x^2 + ... + a(i)x^i

'     = Sum[a(i)x(i)] for i = 1(1)m for LSMulti0, or

'     a(0) + a(1)x(1) + a(1)x(2) + ... = Sum[a(i)x(i)]

'     for i = 0(1)m for LSMulti1;

' (2) the corresponding standard deviations s for these

'     coefficients;

' (3) the corresponding standard deviation of the fit; and

' (4) the corresponding F-ratios for alpha = 0.05 and 0.01.

'

' For each order the output consists of three rows. The top

' row lists the standard deviation of the fit. The second

' row displays the coefficients of the fitted polynomial,

' and the third row the corresponding standard deviations.

' The second and third rows also list the values for the

' F-ratios for 5% and 1% respectively. For easy reading

' the rows are color-coded.

'

' If you want the output to display more digits, change

' the instruction ActiveCell.NumberFormat = "0.00E+00" to

' ActiveCell.NumberFormat = "0.000E+00", "0.0000E+00", etc.

'

' WARNING:

'

' The macro output will take up much space below and to the

' right of the input data. Please move valuable data from

' the output region before running this macro, lest those

' data be overwritten.

'

' PROCEDURE:

'

' Before calling the macro, make sure that the output area

' is clear of valuable data (see above under WARNING). The

' macro does NOT check whether that area is free of data.

'

' In order to start the process, highlight the input data,

' which should occupy at least three adjacent columns, then

' call LSMulti0 or LSMulti1.

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

'

Sub LSMulti0()      ' for unweighted least squares

                ' fit through the origin

Dim p As Double

p = 0

Call LSMulti(p)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub LSMulti1()      ' for general unweighted

                ' least squares fit

Dim p As Double

p = 1

Call LSMulti(p)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub LSMulti(p)

Dim cMax As Integer, Color(1 To 7) As Integer, i As Integer

Dim j As Integer, m As Integer, MM As Integer

Dim MaxOrder As Integer, n As Integer, rMax As Integer

Dim FRatio1 As Double, FRatio5 As Double

Dim Root As Double, StDevY As Double

Dim u As Double, z As Double

Dim DataArray As Variant, F As Variant

Dim lccArray As Variant, outputArray As Variant

Dim SSR As Variant, varY As Variant

Dim vArray As Variant, v0Array As Variant

Dim myRange As Range

Dim bArray, btArray, btqArray, M1, M2, pArray, piArray

Dim qArray, XArray, xtArray, YArray, ytArray, ytyArray

Dim Answer, iAnswer, hAnswer, kAnswer

' Determination of the array size:

Begin:

rMax = Selection.Rows.Count

cMax = Selection.Columns.Count

ReDim F(1 To rMax, 1 To 1) As Double

ReDim SSR(1 To rMax, 1 To 1) As Double

u = 1

z = 0

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "LSMulti")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least (3+p):

If cMax < 3 + p Then

  MsgBox "There must be at least " & 3 + p & " columns.", _

    , "Least Squares Multivariate Fit"

  End

End If

' Check that rmax > cmax+1, so that the number of data

' points is sufficient to define the problem:

If rMax < cMax + 1 Then

  MsgBox "     There must be at least " & cMax + 1 & _

    Chr(13) & "input data pairs to define the problem.", _

    , "Least Squares Multivariate Fit"

  End

End If

DataArray = Selection.Value

Selection.Item(1, 1).Select

MM = 2

RedimensionArrays:

' Dimension the arrays:

ReDim YArray(1 To rMax, 1 To 1) As Double

ReDim XArray(1 To rMax, 1 To MM) As Double

ReDim ytArray(1 To 1, 1 To rMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To MM, 1 To rMax) As Double

ReDim pArray(1 To MM, 1 To MM) As Double

ReDim piArray(1 To MM, 1 To cMax) As Double

ReDim qArray(1 To MM, 1 To 1) As Double

ReDim bArray(1 To MM, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To MM) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim M1(1 To rMax, 1 To rMax) As Double

ReDim M2(1 To rMax, 1 To rMax) As Double

ReDim vArray(1 To MM, 1 To MM) As Double

ReDim v0Array(1 To MM - 1 + p, 1 To MM - 1 + p) As Double

ReDim lccArray(1 To MM, 1 To MM) As Double

' Read the dataArray, then fill yArray and xArray.

For i = 1 To rMax

  YArray(i, 1) = DataArray(i, 1)

Next i

' Check the input data for contiguity.

If MM = 2 Then

  For i = 1 To rMax

    If IsEmpty(DataArray(i, 1)) Then

      MsgBox "Y-value(s) missing", _

        , "Least Squares Multivariate Fit"

      End

    End If

  Next i

  For i = 1 To rMax

    For j = 2 To cMax

      If IsEmpty(DataArray(i, j)) Then

        MsgBox "X-value(s) missing", _

          , "Least Squares Multivariate Fit"

        End

      End If

    Next j

  Next i

End If

' Fill the first column of xArray with either

' zeros (for p = 0) or ones (for p = 1),

' the rest with the data in the x-column(s)

For i = 1 To rMax

  XArray(i, 1) = CDbl(p)

Next i

For i = 1 To rMax

  XArray(i, 2) = DataArray(i, 2)

Next i

If MM > 2 Then

  For i = 1 To rMax

    For j = 3 To MM

      XArray(i, j) = DataArray(i, j)

    Next j

  Next i

End If

' Compute b = (X' X)" X' Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows,

' columns) are:

'   Y          = yArray        (  rmax,     1)

'   X          = xArray        (  rmax,    mm)

'   X'         = xtArray       (  mm,    rmax)

'   X' X       = pArray        (  mm,      mm)

'  (X' X)"     = piArray       (  mm,      mm)

'   X' Y       = qArray        (  mm,       1)

'   b          = bArray        (  mm,       1)

'   Y'         = ytArray       (     1,  rmax)

'   Y' Y       = ytyArray      (     1,     1)

'   b'         = btArray       (     1,    mm)

'   b' X' Y    = btqArray      (     1,     1)

Call Transpose(XArray, rMax, MM, xtArray)

Call Multiply(xtArray, MM, rMax, XArray, MM, pArray)

Call Invert(pArray, MM, piArray)

Call Multiply(xtArray, MM, rMax, YArray, 1, qArray)

Call Multiply(piArray, MM, MM, qArray, 1, bArray)

Call Transpose(YArray, rMax, 1, ytArray)

Call Transpose(bArray, MM, 1, btArray)

Call Multiply(ytArray, 1, rMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, MM, qArray, 1, btqArray)

' Calculate SSR = Y'Y - b'X'Y, and then varY, the variance

' of y, as varY = SSR/(rmax-cmax-p+1); and vArray, the

' covariance matrix, as V = (X'X)" times varY, of which we

' here only use the diagonal elements, i.e., the variances

For i = 1 To MM

  For j = 1 To MM

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(rMax + 1, 0).Select

' Paint color bands for up to 14 orders, and set the

' numerical format to scientific with 3 decimal places

Color(1) = 38

Color(2) = 40

Color(3) = 36

Color(4) = 35

Color(5) = 34

Color(6) = 37

Color(7) = 39

For i = 1 To 3

  For j = 1 To MM + p + 2

    ActiveCell.ClearContents

    If MM < 9 Then

      ActiveCell.Interior.ColorIndex = Color(MM - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(MM - 8)

    End If

    ActiveCell.Offset(0, 1).Select

    If MM < 9 Then

      ActiveCell.Interior.ColorIndex = Color(MM - 1)

    Else

      ActiveCell.Interior.ColorIndex = Color(MM - 8)

    End If

    ActiveCell.Font.Bold = False

    ActiveCell.Font.Italic = False

    ActiveCell.NumberFormat = "0.00E+00"

  Next j

  ActiveCell.Offset(1, -MM - p - 2).Select

Next i

ActiveCell.Offset(-3, 0).Select

' Display the top line of the output

With ActiveCell

  .Font.Bold = True

  .Font.Italic = False

  .HorizontalAlignment = xlLeft

  .Value = "Set # " & MM - 1

End With

For j = 1 To MM + p - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = True

    .HorizontalAlignment = xlCenter

    .Value = "term " & j - p

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Sy:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

Selection.Value = StDevY

Selection.Offset(1, -MM - p - 2).Select

' Display the center line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "Coeff:"

End With

For j = 1 To MM + p - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

    .Value = bArray(j + 1 - p, 1)

  End With

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR5:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If MM = 2 Then Selection.Value = "N/A"

If MM > 2 Then Selection.Value = FRatio5

Selection.Offset(1, -MM - p - 2).Select

' Display the bottom line of the output

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "StDev:"

End With

For j = 1 To MM + p - 1

  Selection.Offset(0, 1).Select

  With ActiveCell

    .Font.Bold = False

    .Font.Italic = False

    .HorizontalAlignment = xlCenter

  End With

  If vArray(j - p + 1, j - p + 1) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = _

      Sqr(Abs(vArray(j - p + 1, j - p + 1)))

  End If

Next j

Selection.Offset(0, 2).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .HorizontalAlignment = xlRight

  .Value = "FR1:"

End With

Selection.Offset(0, 1).Select

Selection.HorizontalAlignment = xlLeft

If MM = 2 Then Selection.Value = "N/A"

If MM > 2 Then Selection.Value = FRatio1

Selection.Offset(1, -MM - p - 2).Select

If MM < cMax + 1 Then MM = MM + 1

ActiveCell.Offset(-rMax - 1, 0).Select

If MM = cMax + 1 Then End

GoTo RedimensionArrays

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^     LEAST SQUARES PERMUTE     ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

'

' The macro LSPermute computes the standard deviations of

' the fit, sy, for all possible combinations and permuta-

' tions of unweighted least squares fits of a dependent

' variable y to a user-specified multivariate expression of

' the form y = a1x1 + a2x2 + .. (for "Thru 0", i.e., for a0

' = 0) or y = a0 + a1x1 + a2x2 + .. (for the "General" case

' where the value of a0 is not constrained), for up to six

' linear terms xi.

'

' SUBROUTINES:

'

' The macro LSPermute calls the subroutine LLSS which,

' in turn, calls the subroutines Invert, Multiply, and

' Transpose.

'

' INPUT:

'

' The input data must be arranged in a block of at least 3

' contiguous columns, with the dependent variable y in the

' left-most column, and the independent variables xi to

' its right.

'

' OUTPUT:

'

' In order to keep the output compact, the macro only

' provides three columns of output information:

' (1) in its first column the indices i of the parameters

'     xi considered;

' (2) in its second column the values of the standard

'     deviation of the fit, Sy, assuming a0 = 0; and

' (3) in its third column the values of the standard

'     deviation of the fit Sy for the general case.

' However, the output will sweep clear an area as wide as

' the input array, and extending by rMax lines below the

' bottom line of the output

'

' WARNING:

'

' Make sure that there are no valuable data below the

' input array, because these will be overwritten. This

' macro does NOT check whether that space is empty.

'

' PROCEDURE:

'

' Before calling the macro, make sure that the space below

' the input data array is clear of valuable data (see above

' under WARNING). The output requires a space as wide as

' the input array, even though that may not be obvious from

' the results displayed.

'

' In order to start the process, highlight the input data

' array, which should occupy at least two adjacent columns,

' then call LSPermute.

'

Sub LSPermute()

Dim C As Integer, c2 As Integer, c3 As Integer

Dim c4 As Integer, c5 As Integer, c6 As Integer

Dim cc As Integer, cMax As Integer, ccMax As Integer

Dim r As Integer, rMax As Integer

Dim StDevY As Double

Dim inputArray As Variant, outputArray As Variant

Dim hAnswer, XArray, YArray

Dim Down As Integer

Down = 0

' Read input array dataArray

Begin:

inputArray = Selection.Value

cMax = Selection.Columns.Count

rMax = Selection.Rows.Count

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "LSPermute")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check the size of the input array

If cMax < 3 Then

  MsgBox "There should be at least three input columns." _

  , , "LSPermute"

  End

End If

If cMax > 7 Then

  MsgBox "This macro can only handle" & Chr(13) & _

    "six independent parameters.", , "LSPermute"

  End

End If

If rMax < cMax + 1 Then

  MsgBox "There are too few rows for the num-" & _

  Chr(13) & "ber of independent parameters used.", _

  , "LSPermute"

  End

End If

' Check for missing input data

For i = 1 To rMax

  If IsEmpty(inputArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "LSPermute"

    End

  End If

Next i

For j = 2 To cMax

  For i = 1 To rMax

    If IsEmpty(inputArray(i, j)) Then

      MsgBox "X-value(s) missing", , "LSPermute"

      End

    End If

  Next i

Next j

' Dimension the data array and the output array

ReDim outputArray(1 To rMax, 1 To cMax)

ReDim YArray(1 To rMax, 1 To 1) As Double

ReDim XArray(1 To rMax, 1 To 1) As Double

' Fill yArray

For r = 1 To rMax

  YArray(r, 1) = inputArray(r, 1)

Next r

' Initialize xArray

For r = 1 To rMax

  For C = 1 To cMax

    outputArray(r, C) = ""

  Next C

Next r

Selection.Offset(rMax + 1, 0).Select

Down = Down + rMax + 1

' Write column headings

outputArray(1, 1) = "Standard deviation of fit"

Selection.Value = outputArray

Selection.Offset(1, 0).Select

outputArray(1, 1) = "Indices:"

outputArray(1, 2) = "Thru 0"

outputArray(1, 3) = "General"

Selection.Value = outputArray

Selection.Offset(2, 0).Select

Down = Down + 3

' Compute the output for 1 variable:

ccMax = 4

ReDim XArray(1 To rMax, 1 To ccMax) As Double

For C = 2 To cMax

  For r = 1 To rMax

    XArray(r, 1) = 0

    XArray(r, 2) = inputArray(r, C)

  Next r

  Call LLSS(ccMax, rMax, 0, YArray, XArray, StDevY)

  outputArray(1, 1) = C - 1

  outputArray(1, 2) = StDevY

  For r = 1 To rMax

    XArray(r, 1) = 1

  Next r

  Call LLSS(ccMax, rMax, 1, YArray, XArray, StDevY)

  outputArray(1, 3) = StDevY

  Selection.Value = outputArray

  Selection.Offset(1, 0).Select

  Down = Down + 1

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 2 variables:

ccMax = 5

ReDim XArray(1 To rMax, 1 To ccMax) As Double

For C = 2 To cMax

  For c2 = C + 1 To cMax

    For r = 1 To rMax

      XArray(r, 1) = 0

      XArray(r, 2) = inputArray(r, C)

      XArray(r, 3) = inputArray(r, c2)

    Next r

    Call LLSS(ccMax, rMax, 0, YArray, XArray, StDevY)

    outputArray(1, 1) = C - 1 & "," & c2 - 1

    outputArray(1, 2) = StDevY

    For r = 1 To rMax

      XArray(r, 1) = 1

    Next r

    Call LLSS(ccMax, rMax, 1, YArray, XArray, StDevY)

    outputArray(1, 3) = StDevY

    Selection.Value = outputArray

    Selection.Offset(1, 0).Select

    Down = Down + 1

  Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 3 variables:

ccMax = 6

ReDim XArray(1 To rMax, 1 To ccMax) As Double

For C = 2 To cMax

 For c2 = C + 1 To cMax

  For c3 = c2 + 1 To cMax

   For r = 1 To rMax

     XArray(r, 1) = 0

     XArray(r, 2) = inputArray(r, C)

     XArray(r, 3) = inputArray(r, c2)

     XArray(r, 4) = inputArray(r, c3)

   Next r

   Call LLSS(ccMax, rMax, 0, YArray, XArray, StDevY)

   outputArray(1, 1) = C - 1 & "," & c2 - 1 & "," & c3 - 1

   outputArray(1, 2) = StDevY

   For r = 1 To rMax

     XArray(r, 1) = 1

   Next r

   Call LLSS(ccMax, rMax, 1, YArray, XArray, StDevY)

   outputArray(1, 3) = StDevY

   Selection.Value = outputArray

   Selection.Offset(1, 0).Select

   Down = Down + 1

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 4 variables:

ccMax = 7

ReDim XArray(1 To rMax, 1 To ccMax) As Double

For C = 2 To cMax

 For c2 = C + 1 To cMax

  For c3 = c2 + 1 To cMax

   For c4 = c3 + 1 To cMax

    For r = 1 To rMax

      XArray(r, 1) = 0

      XArray(r, 2) = inputArray(r, C)

      XArray(r, 3) = inputArray(r, c2)

      XArray(r, 4) = inputArray(r, c3)

      XArray(r, 5) = inputArray(r, c4)

    Next r

    Call LLSS(ccMax, rMax, 0, YArray, XArray, StDevY)

    outputArray(1, 1) = C - 1 & "," & c2 - 1 & _

      "," & c3 - 1 & "," & c4 - 1

    outputArray(1, 2) = StDevY

    For r = 1 To rMax

      XArray(r, 1) = 1

    Next r

    Call LLSS(ccMax, rMax, 1, YArray, XArray, StDevY)

    outputArray(1, 3) = StDevY

    Selection.Value = outputArray

    Selection.Offset(1, 0).Select

    Down = Down + 1

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 5 variables:

ccMax = 8

ReDim XArray(1 To rMax, 1 To ccMax) As Double

For C = 2 To cMax

 For c2 = C + 1 To cMax

  For c3 = c2 + 1 To cMax

   For c4 = c3 + 1 To cMax

    For c5 = c4 + 1 To cMax

      For r = 1 To rMax

        XArray(r, 1) = 0

        XArray(r, 2) = inputArray(r, C)

        XArray(r, 3) = inputArray(r, c2)

        XArray(r, 4) = inputArray(r, c3)

        XArray(r, 5) = inputArray(r, c4)

        XArray(r, 6) = inputArray(r, c5)

      Next r

      Call LLSS(ccMax, rMax, 0, YArray, XArray, StDevY)

      outputArray(1, 1) = C - 1 & "," & c2 - 1 & _

        "," & c3 - 1 & "," & c4 - 1 & "," & c5 - 1

      outputArray(1, 2) = StDevY

      For r = 1 To rMax

        XArray(r, 1) = 1

      Next r

      Call LLSS(ccMax, rMax, 1, YArray, XArray, StDevY)

      outputArray(1, 3) = StDevY

      Selection.Value = outputArray

      Selection.Offset(1, 0).Select

      Down = Down + 1

    Next c5

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(1, 0).Select

Down = Down + 1

' Compute the output for 6 variables:

ccMax = 9

ReDim XArray(1 To rMax, 1 To ccMax) As Double

For C = 2 To cMax

 For c2 = C + 1 To cMax

  For c3 = c2 + 1 To cMax

   For c4 = c3 + 1 To cMax

    For c5 = c4 + 1 To cMax

     For c6 = c5 + 1 To cMax

      For r = 1 To rMax

        XArray(r, 1) = 0

        XArray(r, C) = inputArray(r, 2)

        XArray(r, c2) = inputArray(r, 3)

        XArray(r, c3) = inputArray(r, 4)

        XArray(r, c4) = inputArray(r, 5)

        XArray(r, c5) = inputArray(r, 6)

        XArray(r, c6) = inputArray(r, 7)

      Next r

      Call LLSS(ccMax, rMax, 0, YArray, XArray, StDevY)

      outputArray(1, 1) = C - 1 & "," & _

        "," & c2 - 1 & "," & c3 - 1 & "," & _

        c4 - 1 & c5 - 1 & "," & c6 - 1

      outputArray(1, 2) = StDevY

      For r = 1 To rMax

        XArray(r, 1) = 1

      Next r

      Call LLSS(ccMax, rMax, 1, YArray, XArray, StDevY)

      outputArray(1, 3) = StDevY

      Selection.Value = outputArray

      Selection.Offset(1, 0).Select

      Down = Down + 1

     Next c6

    Next c5

   Next c4

  Next c3

 Next c2

Next C

Selection.Offset(-Down, 0).Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''            ''''''''''''''''''''''''

''''''''''''''''''''''     LLSS     '''''''''''''''''''''''

'''''''''''''''''''''''            ''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub LLSS(ccMax, rMax, p, YArray, XArray, StDevY)

' This subroutine provides the standard deviation StDevY

' of the least squares fit. It calls the subroutines

' Invert, Multiply, and Transpose.

Dim btArray, btqArray

Dim SSR As Double, varY As Double

Dim pArray, piArray, qArray

Dim xtArray, ytArray, ytyArray

ReDim ytArray(1 To 1, 1 To rMax) As Double

ReDim ytyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To ccMax, 1 To rMax) As Double

ReDim pArray(1 To ccMax, 1 To ccMax) As Double

ReDim piArray(1 To ccMax, 1 To ccMax) As Double

ReDim qArray(1 To ccMax, 1 To 1) As Double

ReDim bArray(1 To ccMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To ccMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

' Compute the least squares fit

Call Transpose(XArray, rMax, ccMax, xtArray)

Call Multiply(xtArray, ccMax, rMax, XArray, ccMax, pArray)

Call Invert(pArray, ccMax, piArray)

Call Multiply(xtArray, ccMax, rMax, YArray, 1, qArray)

Call Multiply(piArray, ccMax, ccMax, qArray, 1, bArray)

Call Transpose(YArray, rMax, 1, ytArray)

Call Transpose(bArray, ccMax, 1, btArray)

Call Multiply(ytArray, 1, rMax, YArray, 1, ytyArray)

Call Multiply(btArray, 1, ccMax, qArray, 1, btqArray)

' Compute the standard deviation of the fit

SSR = ytyArray(1, 1) - btqArray(1, 1)

varY = SSR / (rMax - ccMax - p + 3)

StDevY = Sqr(Abs(varY))

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^   ORTHOGONAL POLYNOMIAL FIT   ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

' This subroutine uses Gram-Schmidt orthogonalization to

' compute and display the orthogonal polynomials correspon-

' ding to one or more independent variables x, which need

' not be equidistant. It also provides the fitting para-

' meters, their standard deviations, and their ratios.

' The latter are convenient for determining which terms

' are statistically significant and should be retained.

' SUBROUTINES:

' This macro requires the subroutines Multiply, Invert,

' and Transpose

' INPUT:

' The input data must be arranged as follows. The first

' column must contain the dependent variable y. The second

' (and subsequent) column(s) must contain the independent

' variable(s) x.

' The subroutine requires an input parameter p: p = 1

' causes a general unweighted least squares fit to the

' data, while p = 0 forces the fit to pass through the

' origin, i.e., it assumes that Y = 0 for X = 0. The sub-

' routine Ortho is therefore called by a macro, Ortho1 or

' Ortho0, that sets the value of p.

' OUTPUT:

' The macro produces its output to the right of the input,

' separated by one empty column. It therefore repeats the

' y column. It then computes the fitting parameters and

' their standard deviations. Since these are uncorrelated,

' a covariance matrix is neither needed nor provided. The

' standard deviation of the fit is shown directly below the

' label "Coeff:", as in LS1 and LS0. A third row lists the

' ratio of the absolute values of the coefficient and the

' corresponding standard deviation.

' In general you will NOT want to use the listed coeffi-

' cients and standard deviations, which apply to the ortho-

' gonalized independent parameters. The most useful part of

' the output is the row listing their ratios, which indi-

' cates which orders should be considered (green if ratio >

' 5) and which should not be (red if ratio < 1).

' PROCEDURE:

' Before calling the macro, make sure that the output area

' (a block to the right of the input data block, but exten-

' ding one more column and three more rows) contains no

' valuable data, since these will be overwritten.

' In order to start the process, highlight the entire input

' data array, i.e., the rectangular block containing the

' column for y and the adjacent column(s) for the corres-

' ponding values of x, then call either Ortho(1) or

' Ortho(0)

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

'

Sub Ortho0()    ' for unweighted LS fit through the origin

Dim p As Double

p = 0

Call Ortho(p)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub Ortho1()    ' for general unweighted least squares fit

Dim p As Double

p = 1

Call Ortho(p)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub Ortho(p)

Dim cMax As Integer, cn As Integer, i As Integer

Dim j As Integer, m As Integer, MM As Integer

Dim n As Integer, rMax As Integer

Dim Num As Double, Resid As Double

Dim Root As Double, SSR As Double, StDevY As Double

Dim Sy As Double, SYZ As Double, SZZ As Double

Dim u As Double, varY As Double

Dim bArray As Variant, coeffArray As Variant

Dim DataArray As Variant, Denom As Variant

Dim lccArray As Variant, orthoArray As Variant

Dim outputArray As Variant, stdevArray As Variant

Dim vArray As Variant, v0Array As Variant

Dim XArray As Variant, YArray As Variant

Dim z As Variant

Dim myRange As Range

Dim Answer, hAnswer, iAnswer, kAnswer

' Determination of the array size:

Begin:

rMax = Selection.Rows.Count

cMax = Selection.Columns.Count

ReDim A(0 To cMax) As Double, s(0 To cMax) As Double

u = 1

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Ortho")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least 2:

If cMax < 2 Then

  MsgBox "There must be at least two columns," & _

    Chr(13) & "one for Y, and one or more for X.", _

    , "Orthogonalization"

  End

End If

' Check that rmax > cmax, so that the number of data

' points is sufficient to define the problem:

If rMax < cMax Then

  MsgBox "There must be at least " & cMax & " input" & _

    Chr(13) & "  data to define the problem.", _

    , "Orthogonalization"

  End

End If

' Dimension the arrays:

ReDim bArray(1 To cMax) As Double

ReDim Denom(1 To cMax)

ReDim vArray(1 To cMax, 1 To cMax) As Double

ReDim XArray(1 To rMax, 1 To cMax) As Double

ReDim YArray(1 To rMax) As Double

ReDim z(1 To rMax, 1 To cMax) As Double

' Read the dataArray, then fill yArray and xArray.

DataArray = Selection.Value

For i = 1 To rMax

  YArray(i) = DataArray(i, 1)

Next i

For i = 1 To rMax

  If IsEmpty(DataArray(i, 1)) Then

    MsgBox "Y-value(s) missing", , "Orthogonalization"

    End

  End If

Next i

For j = 2 To cMax

  For i = 1 To rMax

    If IsEmpty(DataArray(i, j)) Then

      MsgBox "X-value(s) missing", , "Orthogonalization"

      End

    End If

  Next i

Next j

' Fill the first column of xArray with zeroes

' (for p = 0) or ones (for p = 1), the rest with

' the data in the x-column(s)

For i = 1 To rMax

  XArray(i, 1) = CDbl(p)

Next i

For j = 2 To cMax

  For i = 1 To rMax

    XArray(i, j) = DataArray(i, j)

  Next i

Next j

cn = 1

' Dimension and fill Z, the matrix of column vectors

' already transformed, initially Z is the 1st column of

' xArray if p = 1, the 2nd if p = 0

ReDim z(1 To rMax, 1 To cn) As Double

' Check against overwriting valuable spreadsheet

' data by the output of orthogonal polynomials

Selection.Offset(0, cMax + 1).Select

orthoArray = Selection.Value

outputArray = Selection.Value

m = 0

For i = 1 To rMax

  For j = 1 To cMax

    If IsEmpty(outputArray(i, j)) Then

      m = m

    Else

      m = m + 1

    End If

  Next j

Next i

If m > 0 Then

  Answer = MsgBox("There are data in the highlighted " _

    & "output block to the" & Chr(13) & _

    "right of the input data array. Can they be " _

    & "overwritten? ", vbYesNo, "Overwrite?")

  If Answer = vbNo Then

    End

  End If

End If

' Check against overwriting valuable spreadsheet data

' by the output of coefficients and standard deviations

Selection.Offset(4, 0).Select

outputArray = Selection.Value

MM = 0

For i = rMax - 1 To rMax

  For j = 1 To cMax

    If IsEmpty(outputArray(i, j)) Then

      MM = MM

    Else

      MM = MM + 1

    End If

  Next j

Next i

If MM > 0 Then

  Answer = MsgBox("There are data in the two bottom " _

    & "lines of the " & Chr(13) & "highlighted array. " _

    & "Can they be overwritten? ", vbYesNo, "Overwrite?")

  If Answer = vbNo Then

    End

  End If

End If

Selection.Offset(-4, 0).Select

For i = 1 To rMax

  orthoArray(i, 1) = DataArray(i, 1)

Next i

Application.ScreenUpdating = False

Do

  For i = 1 To rMax

    For j = 1 To cn

      z(i, cn) = XArray(i, j)

    Next j

  Next i

' Dimension and fill Zi, the next column vector to be

' transformed

  ReDim Zi(1 To rMax, 1 To 1) As Double

  For i = 1 To rMax

    Zi(i, 1) = XArray(i, cn + 1)

  Next i

' Dimension the other vectors and matrices

  ReDim Ztp(1 To cn, 1 To rMax) As Double       ' = Z'

  ReDim ZtpZ(1 To cn, 1 To cn) As Double        ' = Z' Z

  ReDim Z1(1 To cn, 1 To cn) As Double          ' = (Z' Z)"

  ReDim ZtpZi(1 To cn, 1 To 1) As Double        ' = Z' Zi

  ReDim Z2(1 To cn, 1 To 1) As Double

                                ' = Z1 ZtpZi = (Z' Z)" Z' Zi

  ReDim Z3(1 To rMax, 1 To 1) As Double

                                ' = Z Z2 = Z (Z' Z)" Z' Zi

' Compute Z1 = (Z' Z)"

  Call Transpose(z, rMax, cn, Ztp)

  Call Multiply(Ztp, cn, rMax, z, cn, ZtpZ)

  Call Invert(ZtpZ, cn, Z1)

  Call Multiply(Ztp, cn, rMax, Zi, 1, ZtpZi)

  Call Multiply(Z1, cn, cn, ZtpZi, 1, Z2)

  Call Multiply(z, rMax, cn, Z2, 1, Z3)

' Dimension and compute Ztf = Zi - Z3, the transformed

' vector, orthogonal to the vector(s) in Z

  ReDim Ztf(1 To rMax, 1 To 1) As Double

                                ' = Zi - Z (Z' Z)" (Z' Zi)

  For i = 1 To rMax

    Ztf(i, 1) = Zi(i, 1) - Z3(i, 1)

  Next i

  SYZ = 0

  SZZ = 0

  For i = 1 To rMax

    SYZ = SYZ + YArray(i) * Ztf(i, 1)

    SZZ = SZZ + Ztf(i, 1) * Ztf(i, 1)

    orthoArray(i, cn + 1) = Ztf(i, 1)

  Next i

  If cn < cMax Then cn = cn + 1

  ReDim Preserve z(1 To rMax, 1 To cn) As Double

Loop Until cn = cMax

' Display the orthogonal array

Selection.Value = orthoArray

' Update yArray and xArray.

For i = 1 To rMax

  YArray(i) = orthoArray(i, 1)

Next i

For i = 1 To rMax

  XArray(i, 1) = CDbl(p)

Next i

For j = 2 To cMax

  For i = 1 To rMax

    XArray(i, j) = orthoArray(i, j)

  Next i

Next j

' Compute the coefficients

For j = 2 - p To cMax

  Num = 0

  Denom(j) = 0

  For i = 1 To rMax

    Num = Num + XArray(i, j) * YArray(i)

    Denom(j) = Denom(j) + XArray(i, j) * XArray(i, j)

  Next i

  bArray(j) = Num / Denom(j)

Next j

' Compute the variance of y

SSR = 0

For i = 1 To rMax

  Resid = YArray(i)

  For j = 2 - p To cMax

    Resid = Resid - bArray(j) * XArray(i, j)

  Next j

  SSR = SSR + Resid * Resid

Next i

varY = SSR / (rMax - cMax - p + 1)

StDevY = Sqr(Abs(varY))

' Compute the variances of the coefficients

For j = 2 - p To cMax

  vArray(j, j) = varY / Denom(j)

Next j

ActiveCell.Offset(rMax, 0).Select

' Arrange for the data output

For j = 1 To cMax

  With ActiveCell.Font

    .Bold = True

    .Italic = True

    .ColorIndex = 1

  End With

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -cMax).Select

For j = 1 To cMax

  ActiveCell.Font.Italic = True

  ActiveCell.Font.ColorIndex = 1

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -cMax).Select

For j = 1 To cMax

  ActiveCell.Font.Italic = True

  ActiveCell.Font.ColorIndex = 1

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(-2, -cMax - p).Select

With ActiveCell

  .Font.Bold = True

  .Font.Italic = True

  .Font.ColorIndex = 1

  .HorizontalAlignment = xlRight

  .Value = "Coeff:"

End With

ActiveCell.Offset(0, 1).Select

For j = 2 - p To cMax

  With ActiveCell.Font

    .Bold = True

    .Italic = True

    .ColorIndex = 1

  End With

  ActiveCell.Value = bArray(j)

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -cMax - p).Select

With ActiveCell.Font

  .Bold = False

  .Italic = True

  .ColorIndex = 1

End With

ActiveCell.HorizontalAlignment = xlRight

ActiveCell.Value = "StDev:"

ActiveCell.Offset(0, 1).Select

ActiveCell.Font.ColorIndex = 1

For j = 2 - p To cMax

  If vArray(j, j) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = Sqr(Abs(vArray(j, j)))

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -cMax - p).Select

With ActiveCell

  .Font.Bold = False

  .Font.Italic = True

  .Font.ColorIndex = 1

  .HorizontalAlignment = xlRight

  .Value = "Ratio:"

End With

ActiveCell.Offset(0, 1).Select

For j = 2 - p To cMax

  If vArray(j, j) <= 0 Then

    ActiveCell.Value = "N/A"

    ActiveCell.Font.ColorIndex = 1

  End If

  If vArray(j, j) > 0 Then

    ActiveCell.Font.Bold = False

    ActiveCell.Font.Italic = True

    ActiveCell.Font.ColorIndex = 16

    ActiveCell.Value = Abs(bArray(j) _

      / Sqr(vArray(j, j)))

    If ActiveCell.Value > 5 Then _

      ActiveCell.Font.ColorIndex = 10

    If ActiveCell.Value < 1 Then _

      ActiveCell.Font.ColorIndex = 3

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, -cMax).Select

' Provide the standard deviation of the fit

ActiveCell.Offset(0, 1 - p).Select

With ActiveCell.Font

  .Bold = False

  .Italic = True

  .ColorIndex = 1

End With

Selection.Value = StDevY

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''^     EQUIDISTANT LEAST SQUARES     ^'''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''''''''''''© P. Barak, pwbarak@facstaff.wisc.edu

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

'

' This program for a least-squares fit to EQUIDISTANT data

' y,x with a moving polynomial uses the approach pioneered

' by Gram, further developed by Sheppard [Proc. London

' Math. Soc. (2) 13 (1914) 81] and Sherriff [Proc. Royal

' Soc. Edinburgh 40 (1920) 112], and subsequently advocated

' by Whittaker & Robinson [The calculus of observations,

' Blackie & Son, 1924] and by Savitzky & Golay [Anal. Chem.

' 36 (1964) 1627]. It computes smoothed values of the data

' set, or (if so desired) its first or second derivative.

'

' There are two options: in ELSfixed() the user also

' selects the order of the polynomial, whereas in ELSauto()

' the program optimizes the order of the polynomial (bet-

' ween 1 and an upper limit set by the user) each time the

' moving polynomial slides one data point along the data

' set, using an algorithm described by Barak in Anal. Chem.

' 67 (1995) 2758.

'

' The program compares the ratio of the variances for a

' given order and that for the next-lower order with the

' corresponding F test as its first criterion. Since symme-

' trical functions often contain mostly even powers, a

' single, final comparison is made between the variances of

' the next-higher and the next-lower order. If the latter

' is not desired, simply comment out the section following

' the comment line "Second test for optimum Order".

'

' FUNCTIONS & SUBROUTINES

'

' There are two drivers, ELSfixed and ELSauto, that call

' the main subroutine, ELS(iOrder). The latter uses the

' functions GenFact() and Smooth(). It also calls the sub-

' routine ConvolutionFactors, which calculates the Gram

' polynomials and the corresponding convolution weights.

'

' INPUT

'

' The user must select the moving polynomial length; the

' corresponding value is restricted to odd integers between

' 3 and 31. This choice involves a compromise: the longer

' the polynomial, the more noise is removed, and the more

' the underlying signal is distorted. A useful rule of

' thumb is to make the length such that it encompasses no

' more than the half-width of the most narrow feature in

' the signal that should be preserved.

'

' Moreover, the length of the moving polynomial, PL, cannot

' exceed the length of the data set, NPts. The selected

' (maximum) order of the polynomial, MaxOrder, must be a

' positive integer, MaxOrder>0. Moreover, for ELSfixed,

' MaxOrder<PL, while for ELSauto we have MaxOrder<(PL-1).

'

' OUTPUT

'

' The output is written in one or two columns to the right

' of the input data. The first output column contains the

' smoothed or differentiated data. The second column, which

' appears only with ELSauto(), displays the order selected

' by the program. Make sure that the output space is free,

' or can be overwritten.

'

' Some of the abbreviations and indices used:

'     DerOrder: the derivative order, selected by the

'       answer to ELS InputBox 4: Derivative Order:

'         DerOrder = 0 for smoothing,

'         DerOrder = 1 for the 1st derivative,

'         DerOrder = 2 for the 2nd derivative.

'     FValueTable: the Fisher criterion, obtained

'       from the Excel function FInv.

'     GP: the Gram Polynomial

'     j: index for the order of the polynomial.

'     k: index for the position of the center of the

'       moving polynomial in the data set;

'       k ranges from m+1 to NPts-m.

'     m: number of points on each side of the central

'       point in the moving polynomial, calculated as

'       m = (PL-1)/2.

'     MaxOrder: the maximum order of the polynomial

'       (from ELS InputBox 3: Polynomial Order) or its

'       maximum value (from ELS InputBox 3: Maximum Poly-

'       nomial Order). MaxOrder must be an integer, > 0,

'       and < PL for ELSfixed, or < (PL-1) for ELSauto.

'     NPts: number of points in the data set, computed by

'       the macro from the data range provided in ELS

'       InputBox 1: Input data.

'     OptOrder: array of optimized Order values.

'     Order: working value of the polynomial order.

'       In ELSfixed, Order = MaxOrder; in ELSauto, Order

'       starts at 1, and has a maximum value of MaxOrder.

'     OutputData: the final (smoothed or derivative) result

'     OutputOrder:  the values of OptOrder used in ELSauto.

'     PL: length (in number of points) of the moving poly-

'       nomial, selected in ELS InputBox 2: Length of Mo-

'       ving Polynomial.

'     s: index for the order of the derivative; s ranges

'       from 0 to DerivOrder.

'     t: index for the individual points in the moving

'       polynomial; t ranges from -m to +m.

'     tries:  number of attempts to enter data in input box

'     Y() and YData(): the input data containing the entire

'       data set.

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub ELSfixed()

' Selects a fixed polynomial order

' by setting iOrder equal to 1

Dim iOrder As Integer

iOrder = 1

Call ELS(iOrder)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub ELSauto()

' Selects a variable polynomial order (between 1 and a

' user-selectable maximum value,MaxOrder) by setting iOrder

' equal to -1. Note that the high-order polynomials may be

' undesirable for use with, e.g., differentiation or inter-

' polation.

Dim iOrder As Integer

iOrder = -1

Call ELS(iOrder)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub ELS(iOrder)

Dim A As Integer, B As Integer, DerOrder As Integer

Dim i As Integer, ii As Integer, j As Integer

Dim jj As Integer, k As Integer, m As Integer

Dim MaxOrder As Integer, n As Integer, NPts As Integer

Dim Order As Integer, PL As Integer, q As Integer

Dim s As Integer, t As Integer, Tries2 As Integer

Dim Tries3 As Integer, Tries4 As Integer

Dim GenFact As Double, Percentage As Long

Dim aa As Double, bb As Double, DeltaX As Double

Dim FTest1 As Double, FTest2 As Double, Length As Double

Dim Perc As Double, Sum As Double, SumSq As Double

Dim SumXY As Double, SumY As Double, SumY2 As Double

Dim GP() As Double, SumSquares() As Double

Dim SumX2() As Double, w() As Double, Y() As Double

Dim OptOrder As Variant, outputData As Variant

Dim OutputOrder As Variant, TestContents As Variant

Dim YData As Variant

Dim myRange As Range

Dim Ans As String

Dim z

' Preliminaries and data input

Set myRange = _

  Application.InputBox(Prompt:="The input data are " _

  & "located in column:", _

  Title:="ELS InputBox 1: Input data", Type:=8)

myRange.Select

If Selection.Columns.Count <> 1 Then

  MsgBox "Only enter a single column of input data."

  End

End If

NPts = Selection.Rows.Count

If NPts = 0 Then End

YData = myRange.Value

outputData = myRange.Value          'defines the array size

OutputOrder = myRange.Value         'defines the array size

OptOrder = myRange.Value            'defines the array size

' Test and prepare the default output range

n = 0

Selection.Offset(0, 1).Select

TestContents = Selection.Value

For i = 1 To NPts

  z = TestContents(i, 1)

  If IsEmpty(z) Then

    n = n

  Else

    n = n + 1

  End If

Next i

If iOrder = 1 Then

  If n > 0 Then

    Ans = MsgBox(" There are data in the " _

      & "column where the output " & Chr(13) & _

      "will be written. Proceed anyway and overwrite " _

      & "those data?", vbYesNo, _

      "Equidistant Least Squares Fit")

    If Ans = vbNo Then

      MsgBox ("Safeguard the data in the highlighted " _

        & "area by" & Chr(13) & "moving them to another " _

        & "place, then try again."), _

        , "Equidistant Least Squares Fit"

    End

    End If

  End If

  Selection.Offset(0, -1).Select

End If

If iOrder = -1 Then

  Selection.Offset(0, 1).Select

  TestContents = Selection.Value

  For i = 1 To NPts

    z = TestContents(i, 1)

    If IsEmpty(z) Then

      n = n

    Else

      n = n + 1

    End If

  Next i

  If n > 0 Then

    Ans = MsgBox("There are data in the TWO columns " _

      & "where the output " & Chr(13) & "will be " _

      & "written. Proceed anyway and overwrite " _

      & Chr(13) & "those data?", vbYesNo, "Equidistant " _

      & "Least Squares Fit")

    If Ans = vbNo Then

      MsgBox ("Safeguard the data in the highlighted " _

      & "area by" & Chr(13) & "moving them to another " _

      & "place, then try again."), _

      , "Equidistant Least Squares Fit"

    End

    End If

  End If

  Selection.Offset(0, -2).Select

End If

' Select the length of the moving polynomial, PL

Tries2 = 0

Line2:

Length = InputBox(Prompt:="The length of the moving " _

  & "polynomial is:", _

  Title:="ELS InputBox 2: Polynomial Length")

PL = CInt(Length)

' Make sure that PL is an odd integer

' larger than 0 and smaller than NPts

If (Length <= 0 Or Length >= NPts Or PL - Length <> 0 _

  Or CInt((PL - 1) / 2) - ((Length - 1) / 2) <> 0) Then

  MsgBox "The length of the moving polynomial must" & _

    Chr(13) & "    be an odd integer larger than zero " _

    & "and " & Chr(13) & "smaller than the length of " & _

    "the input column.", , "Equidistant Least Squares Fit"

  Tries2 = Tries2 + 1

  If Tries2 = 2 Then End

  GoTo Line2

End If

' Select the order of the moving polynomial, MaxOrder

Tries3 = 0

Line3:

If iOrder = 1 Then

  MaxOrder = InputBox(Prompt:="The order of the moving" _

    & "polynomial is:", _

    Title:="ELS InputBox 3: Polynomial Order")

Else

  MaxOrder = InputBox(Prompt:="The maximum order of the " _

    & "moving polynomial is:", _

    Title:="ELS InputBox 3: Polynomial Order")

End If

' Make sure that MaxOrder > 0 and that either MaxOrder < PL

' (for ELSfixed) or MaxOrder < PL - 1 (for ELSauto).

If iOrder = 1 Then

  If (MaxOrder <= 0 Or MaxOrder >= PL) Then

    MsgBox "The order of the moving polynomial" & Chr(13) _

      & "must be larger than zero, and smaller" & Chr(13) _

      & "than the length of the moving polynomial.", , _

      "Equidistant Least Squares Fit"

    Tries3 = Tries3 + 1

    If Tries3 = 2 Then End

    GoTo Line3

  End If

Else

  If (MaxOrder <= 0 Or MaxOrder >= PL - 1) Then

    MsgBox "The maximum order of the moving polynomial" & _

      Chr(13) & "must be larger than zero, and smaller " _

      & "than" & Chr(13) & "the length of the moving " _

      & "polynomial minus 1.", , _

      "Equidistant Least Squares Fit"

    Tries3 = Tries3 + 1

    If Tries3 = 2 Then End

    GoTo Line3

  End If

End If

' Select smoothing, first derivative, or second derivative

Tries4 = 0

Line4:

DerOrder = InputBox(Prompt:="Select the order of the " _

  & "derivative" & Chr(13) & "(either 1 or 2); for " _

  & "smoothing, select 0." & Chr(13) & Chr(13) _

  & "The order of the derivative is:", _

  Title:="ELS InputBox 4:" & "Derivative Order")

' Make sure that DerOrder has the value 0, 1, or 2

If DerOrder = 0 Then

  GoTo Line6

ElseIf DerOrder = 1 Then

  GoTo Line5

ElseIf DerOrder = 2 Then

  GoTo Line5

Else

  MsgBox " The order of the moving polynomial must be" & _

  Chr(13) & "either 0 (for smoothing), 1 (for the first" _

  & Chr(13) & "derivative), or 2 (for the second " & _

  "derivative).", , "Equidistant Least Squares Fit"

  Tries4 = Tries4 + 1

  If Tries4 = 2 Then End

  GoTo Line4

End If

Line5:

DeltaX = InputBox(Prompt:="The data spacing in x is:", _

  Title:="ELS InputBox 5: X Increment")

Line6:

m = (PL - 1) / 2

ReDim Y(1 To NPts), OptOrder(1 To NPts, 1)

For i = 1 To NPts

  Y(i) = YData(i, 1)

  OptOrder(i, 1) = MaxOrder

Next i

Call ConvolutionFactors(PL, MaxOrder, DerOrder, GP, w, Y)

' THE FOLLOWING SECTION IS USED ONLY BY ELSauto

If iOrder = -1 Then

  ReDim SumX2(1 To MaxOrder)

  For j = 1 To MaxOrder

    Sum = 0

    For i = -m To m

      Sum = Sum + GP(i, j, 0) ^ 2

    Next i

    SumX2(j) = Sum

  Next j

' Calculate FValueTable(MaxOrder,PL)

  ReDim FValueTable(1 To MaxOrder, 1 To PL)

  jj = 0

  Do

    jj = jj + 1

    For ii = 1 To MaxOrder

      FValueTable(ii, jj) = Application.FInv(0.05, ii, jj)

    Next ii

  Loop Until jj = PL

  For k = m + 1 To NPts - m

    ReDim SumSquares(0 To MaxOrder)

' Calculate SumSquares for Order = 0

    Order = 0

    SumY = 0

    SumY2 = 0

    For t = -m To m

      SumY = SumY + Y(k + t)

      SumY2 = SumY2 + Y(k + t) ^ 2

    Next t

    SumSquares(0) = SumY2 - SumY ^ 2 / (2 * t + 1)

' Calculate SumSquares for Order = 1

    Order = 1

    SumSq = 0

    For t = -m To m

      SumSq = SumSq + (Smooth(PL, k, Order, t, 0, w, Y) _

        - Y(k + t)) ^ 2

    Next t

    SumSquares(1) = SumSq

' Test whether one-higher order satisfies the criterion

    Do

      Order = Order + 1

      If Order > MaxOrder Then GoTo line10

' Calculate SumSquares for Order > 1

      SumXY = 0

      For t = -m To m

        SumXY = SumXY + Y(k + t) * GP(t, Order, 0)

      Next t

      SumSquares(Order) = SumSquares(Order - 1) _

        - SumXY ^ 2 / SumX2(Order)

' First test for optimum Order

      FTest1 = _

        (SumSquares(Order - 1) - SumSquares(Order)) * _

        (PL - Order) / (SumSquares(Order))

    Loop Until _

      (FTest1 / FValueTable(1, PL - Order - 1)) < 1

' Second test for optimum Order

    If Order < MaxOrder Then

      Order = Order + 1

      SumXY = 0

      For t = -m To m

        SumXY = SumXY + Y(k + t) * GP(t, Order, 0)

      Next t

      SumSquares(Order) = SumSquares(Order - 1) _

        - SumXY ^ 2 / SumX2(Order)

      FTest2 = _

        (SumSquares(Order - 2) - SumSquares(Order)) * _

        (PL - Order) / (SumSquares(Order))

      If (FTest2 / FValueTable(2, PL - Order - 1)) < 1 _

        Then Order = Order - 1

    End If

line10:

    OptOrder(k, 1) = Order - 1

    Perc = 100 * (k / NPts)

    Percentage = Int(Perc)

    Application.StatusBar = _

      "Calculation " & Percentage & "% done."

  Next k

End If

' THIS ENDS THE PART USED ONLY BY ELSauto

' Prepare the output files

For k = m + 1 To NPts - m

  If k = m + 1 Then

    For t = -m To -1

      outputData(k + t, 1) = _

        Smooth(PL, k, OptOrder(k + t, 1), _

        t, DerOrder, w, Y) / (DeltaX ^ DerOrder)

      OutputOrder(k + t, 1) = OptOrder(k, 1)

    Next t

  End If

  outputData(k, 1) = _

    Smooth(PL, k, OptOrder(k, 1), 0, DerOrder, w, Y) _

    / (DeltaX ^ DerOrder)

  OutputOrder(k, 1) = OptOrder(k, 1)

  If k = NPts - m Then

    For t = 1 To m

      outputData(k + t, 1) = _

        Smooth(PL, k, OptOrder(k + t, 1), _

        t, DerOrder, w, Y) / (DeltaX ^ DerOrder)

      OutputOrder(k + t, 1) = OptOrder(k, 1)

    Next t

  End If

Next k

' Write the output files

Selection.Offset(0, 1).Select

Selection.Value = outputData

If iOrder = -1 Then

  Selection.Offset(0, 1).Select

  Selection.Value = OutputOrder

  Selection.Offset(0, -1).Select

End If

Application.StatusBar = False

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Function GenFact(A, B)

' Computes the generalized factorial

Dim gf As Double, j As Integer

gf = 1

For j = (A - B + 1) To A

  gf = gf * j

Next j

GenFact = gf

End Function

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Public Function Smooth(PL, k, j, t, s, w, Y)

' Computes the appropriately weighted sum of the Y-values

Dim i As Integer, m As Integer, Sum As Double

m = (PL - 1) / 2

Sum = 0

For i = -m To m

  Sum = Sum + w(i, t, j, s) * Y(k + i)

Next i

Smooth = Sum

End Function

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub ConvolutionFactors(PL, MaxOrder, DerOrder, GP, w, Y)

' Calculates tables of GP(i = -m to m, k = -1 to MaxOrder,

' s = 1 to DerOrder), and of W(i = -m to m, t = -m to m,

' k = -1 to MaxOrder, s = -1 to DerOrder)

' Abbreviations used:

'   DerOrder: Derivative Order

'   GP:       Gram Polynomial

'   MaxOrder: Maximum Order,

'   PL:       Polynomial Length

'   W:        Weight

Dim i As Integer, k As Integer, m As Integer

Dim s As Integer, t As Integer

Dim aa As Double, bb As Double

m = (PL - 1) / 2

ReDim GP(-m To m, -1 To MaxOrder, -1 To DerOrder)

'Evaluate the Gram polynomials for DerOrder=0

For i = -m To m

  GP(i, 0, 0) = 1

  GP(i, 1, 0) = 0

Next i

For i = -m To -1

  GP(i, 1, 0) = i / m

Next i

For i = 1 To m

  GP(i, 1, 0) = i / m

Next i

For k = 2 To MaxOrder

  aa = 2 * (2 * k - 1) / (k * (2 * m - k + 1))

  bb = ((k - 1) * (2 * m + k)) / (k * (2 * m - k + 1))

  For i = 0 To m

    GP(i, k, 0) = aa * i * GP(i, k - 1, 0) - _

      bb * GP(i, k - 2, 0)

  Next i

  For i = -m To -1

    If k Mod 2 = 0 Then

      GP(i, k, 0) = GP(-i, k, 0)

    Else

      GP(i, k, 0) = -GP(-i, k, 0)

    End If

  Next i

Next k

'Evaluate the Gram polynomials for DerOrder>0

If DerOrder > 0 Then

  For s = 1 To DerOrder

    For i = -m To m

      GP(i, -1, s) = 0

      GP(i, 0, s) = 0

    Next i

    For k = 1 To MaxOrder

      aa = 2 * (2 * k - 1) / (k * (2 * m - k + 1))

      bb = ((k - 1) * (2 * m + k)) / (k * (2 * m - k + 1))

      For i = -m To m

        GP(i, k, s) = aa * (i * GP(i, k - 1, s) + _

        s * GP(i, k - 1, s - 1)) - bb * GP(i, k - 2, s)

      Next i

    Next k

  Next s

End If

'Calculate the convolution weights

ReDim w(-m To m, -m To m, -1 To MaxOrder, -1 To DerOrder)

For k = 0 To MaxOrder

  aa = (2 * k + 1) * GenFact(2 * m, k) / _

    GenFact(2 * m + k + 1, k + 1)

  For s = 0 To DerOrder

    For i = -m To m

      For t = -m To m

        w(i, t, k, s) = w(i, t, k - 1, s) _

        + aa * GP(i, k, 0) * GP(t, k, s)

      Next t

    Next i

  Next s

Next k

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

'''''''''''''^                             ^'''''''''''''''

''''''''''''^    WEIGHTED LEAST SQUARES     ^''''''''''''''

'''''''''''''^                             ^'''''''''''''''

''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''' (c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' The function of the following two drivers is merely to

' set the value of one parameter, p, equal to either one or

' zero, in order to choose between a general least squares

' fitting (p = 1) or one that forces the curve through the

' origin (p = 0).

'

Sub WLS0()          ' for least squares fit

                ' through the origin

Dim p As Double

p = 0

Call WeightedLeastSquares(p)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub WLS1()          ' for general least squares fit

Dim p As Double

p = 1

Call WeightedLeastSquares(p)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub WeightedLeastSquares(p)

' PURPOSE:

' The macros WLS1 and WLS0 compute the parameters and their

' standard deviations for a weighted least squares fit to

' data in 2 or more columns. The macros set the input para-

' meter p for the subroutine WeightedLeastSquares: p = 1

' causes a general weighted least squares fit to the data,

' while p = 0 forces the fit to pass through the origin,

' i.e., it assumes that y = 0 for x = 0.

' SUBROUTINES:

' This macro requires the subroutines Multiply, Invert, and

' Transpose

' INPUT:

' The input data must be organized in columns, arranged as

' follows. The first column must contain the dependent va-

' riable y. The second column should contain the weights w

' of the individual points. The third (and any subsequent)

' column(s) must contain the independent variable(s) x.

' If an unweighted least squares is desired, leave the

' weights blank, or enter the same number (say, 1, or 13)

' for all individual weights.

' Do NOT delete the second column, as in that case the

' macro will misinterpret the input.

' OUTPUT:

' The macro labels the coefficients and standard devia-

' tions, except when there is no space available for them

' because the first data column is at the left edge of the

' spreadsheet.

' The macro also provides the covariance matrix and,

' optionally, the corresponding linear correlation

' coefficient array.

' If the data to be analyzed are the result of a trans-

' formation (such as taking logarithms of an exponential

' decay), the standard deviation of the overall fit will

' not yield the correct answer with respect to those

' original, untransformed data, because the macro does not

' read those data. If that standard deviation is desired,

' it is best computed separately on the spreadsheet.

' PROCEDURE:

' Before calling the macro, make sure that the output area

' is free of valuable data. To find out the reach of the

' output, try out the macro but, in the "Overwrite?"

' message box, note the number of rows required, answer

' "No" to whether the data can be overwritten, move the

' sensitive data, then try the macro again.

' In order to start the process, highlight the entire input

' data block, then call WLS1 or WLS0.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' NOTATION:

' This subroutine computes the parameters and their

' standard deviations for a weighted or unweighted least

' squares fit to data in 3 or more columns. The columns

' must be arranged as follows:

' The first column must contain the dependent variable y.

' The second column must contain the weights w for y. For

' an unweighted least squares fit, leave this column blank,

' or fill it with ones, but do not omit it. The third (and

' possibly subsequent) column(s) must contain the indepen-

' dent variable(s) x.

' The subroutine requires an input parameter p: p = 1

' causes a general weighted least squares fit to the data,

' while p = 0 forces the fit to pass through the origin,

' i.e., it assumes that Y = 0 for X = 0. The subroutine is

' therefore called by a driver (WLS0 or WLS1) that sets the

' value of p.

Dim ccMax As Integer, cMax As Integer

Dim i As Integer, j As Integer

Dim m As Integer, n As Integer

Dim rMax As Integer, Root As Double

Dim SSR As Double, StDevY As Double, sumW As Double

Dim u As Double, varY As Double

Dim DataArray As Variant, outputArray As Variant

Dim lccArray As Variant

Dim vArray As Variant, v0Array As Variant

Dim myRange As Range

Dim Answer, hAnswer, iAnswer, jAnswer, kAnswer

Dim bArray, btArray, btqArray

Dim M1, M2, pArray, piArray, qArray

Dim XArray, xtArray, YArray, ytArray, ytyArray

' Determination of the array size:

Begin:

rMax = Selection.Rows.Count

cMax = Selection.Columns.Count

ccMax = cMax - 1

u = 1

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Weighted Least Squares")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that the number of columns is at least 3:

If cMax < 3 Then

  MsgBox "There must be at least three columns, one" & _

  Chr(13) & "for Y, one for W, and one or more for X.", _

  , "Weighted Least Squares Fit"

  End

End If

' Check that there are more data than coefficients :

If rMax - cMax - p + 2 <= 0 Then

  MsgBox "With " & rMax & " data, WLS" & p & " can only deter-" & _

    Chr(13) & "mine " & rMax - 1 & " least squares coefficients." & _

    Chr(13) & Chr(13) & "Add more data, or reduce the re-" & _

    Chr(13) & "quested number of coefficients."

  End

End If

' Dimension the arrays:

ReDim YArray(1 To rMax, 1 To 1) As Double

ReDim wArray(1 To rMax, 1 To rMax) As Double

ReDim XArray(1 To rMax, 1 To ccMax) As Double

ReDim ytArray(1 To 1, 1 To rMax) As Double

ReDim ytwArray(1 To 1, 1 To rMax) As Double

ReDim ytwyArray(1 To 1, 1 To 1) As Double

ReDim xtArray(1 To ccMax, 1 To rMax) As Double

ReDim xtwArray(1 To ccMax, 1 To rMax) As Double

ReDim pArray(1 To ccMax, 1 To ccMax) As Double

ReDim piArray(1 To ccMax, 1 To ccMax) As Double

ReDim qArray(1 To ccMax, 1 To 1) As Double

ReDim bArray(1 To ccMax, 1 To 1) As Double

ReDim btArray(1 To 1, 1 To ccMax) As Double

ReDim btqArray(1 To 1, 1 To 1) As Double

ReDim M1(1 To rMax, 1 To rMax) As Double

ReDim M2(1 To rMax, 1 To rMax) As Double

ReDim vArray(1 To ccMax, 1 To ccMax) As Double

ReDim v0Array(1 To ccMax + p - 1, 1 To ccMax + p - 1) _

  As Double

ReDim lccArray(1 To ccMax + p - 1, 1 To ccMax + p - 1) _

  As Double

' Read the dataArray, then fill the various input arrays:

' yArray, wArray, and xArray. The wArray contains zeros

' except that it has the individual, normalized weights as

' its diagonal elements.

DataArray = Selection.Value

For i = 1 To rMax

  YArray(i, 1) = DataArray(i, 1)

Next i

For i = 1 To rMax

  If IsEmpty(DataArray(i, 1)) Then

    MsgBox "Y-value(s) missing", _

      , "Weighted Least Squares"

    End

  End If

Next i

For i = 1 To rMax

  If IsEmpty(DataArray(i, 2)) Then DataArray(i, 2) = u

Next i

For j = 3 To cMax

  For i = 1 To rMax

    If IsEmpty(DataArray(i, j)) Then

      MsgBox "X-value(s) missing", _

        , "Weighted Least Squares"

      End

    End If

  Next i

Next j

' First zero the entire wArray, then fill its

' main diagonal with the listed normalized weights,

' with the normalization factor sumW / rMax

sumW = 0

For i = 1 To rMax

  sumW = sumW + DataArray(i, 2)

Next i

For i = 1 To rMax

  For j = 1 To rMax

    wArray(i, j) = 0

  Next j

  wArray(i, i) = DataArray(i, 2) * rMax / sumW

Next i

' Fill the first column of xArray with zeroes (for p = 0)

' or ones (for p = 1), the rest with the data in the

' x-column(s)

For i = 1 To rMax

  XArray(i, 1) = p

Next i

For j = 3 To cMax

  For i = 1 To rMax

    XArray(i, (j - 1)) = DataArray(i, j)

  Next i

Next j

' Compute b = (X' W X)" X' W Y , where ' or t denote

' transposition, and " or i indicate inversion

' The various arrays and their dimensions (rows, columns)

' are:

'   Y          = yArray        (  rmax,     1)

'   W          = wArray        (  rmax,  rmax)

'   X          = xArray        (  rmax, ccmax)

'   X'         = xtArray       ( ccmax,  rmax)

'   X' W       = xtwArray      ( ccmax,  rmax)

'   X' W X     = pArray        ( ccmax, ccmax)

'  (X' W X)"   = piArray       ( ccmax, ccmax)

'   X' W Y     = qArray        ( ccmax,     1)

'   b          = bArray        ( ccmax,     1)

Call Transpose(XArray, rMax, ccMax, xtArray)

Call Multiply(xtArray, ccMax, rMax, wArray, rMax, xtwArray)

Call Multiply(xtwArray, ccMax, rMax, XArray, ccMax, pArray)

Call Invert(pArray, ccMax, piArray)

Call Multiply(xtwArray, ccMax, rMax, YArray, 1, qArray)

Call Multiply(piArray, ccMax, ccMax, qArray, 1, bArray)

' Check against overwriting spreadsheet data

Selection.Offset(2, 0).Select

outputArray = Selection.Value

Selection.Offset(-2, 0).Select

m = 0

If (p = 0 And cMax = 3) Then

  For i = 1 To 3

    Selection.Offset(1, 0).Select

    outputArray = Selection.Value

    For j = 1 To cMax

      If IsEmpty(outputArray(rMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-3, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & "three lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

Else

  For i = 1 To 3 + p + cMax

    Selection.Offset(1, 0).Select

    outputArray = Selection.Value

    For j = 1 To cMax

      If IsEmpty(outputArray(rMax, j)) Then

        m = m

      Else

        m = m + 1

      End If

    Next j

  Next i

  Selection.Offset(-3 - p - cMax, 0).Select

  If m > 0 Then Answer = MsgBox("There are data in the " _

    & 1 + p + cMax & " lines below the" & Chr(13) & _

    "input data array. " & "Can they be overwritten?", _

    vbYesNo, "Overwrite?")

  If Answer = vbNo Then End

End If

' The additional arrays and their dimensions (rows,

' columns) are:

'   Y'         = ytArray       (     1,  rmax)

'   Y' W       = ytwArray      (     1,  rmax)

'   Y' W Y     = ytwyArray     (     1,     1)

'   b'         = btArray       (     1, ccmax)

'   b' X' W Y  = btqArray      (     1,     1)

Call Transpose(YArray, rMax, 1, ytArray)

Call Transpose(bArray, ccMax, 1, btArray)

Call Multiply(ytArray, 1, rMax, wArray, rMax, ytwArray)

Call Multiply(ytwArray, 1, rMax, YArray, 1, ytwyArray)

Call Multiply(btArray, 1, ccMax, qArray, 1, btqArray)

' Calculate SSR = Y'WY - b'X'WY, and then varY, the vari-

' ance of y, as varY = SSR/(rmax-ccmax); and vArray, the

' covariance matrix, as V = (X'WX)" times varY, of which

' we here only use the diagonal elements, i.e., the vari-

' ances.

SSR = ytwyArray(1, 1) - btqArray(1, 1)

varY = SSR / (rMax - ccMax - p + 1)

If SSR >= 0 Then StDevY = Sqr(varY)

If SSR < 0 Then

  StDevY = Sqr(-varY)

  MsgBox "SSR is " & SSR & ", which is negative!"

End If

For i = 1 To ccMax

  For j = 1 To ccMax

    vArray(i, j) = varY * piArray(i, j)

  Next j

Next i

Application.ScreenUpdating = False

ActiveCell.Offset(rMax, 0).Select

For j = 1 To cMax

  ActiveCell.Font.Bold = True

  ActiveCell.Font.Italic = True

  ActiveCell.Value = ""

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(0, -cMax).Select

ActiveCell.Value = "Coeff:"

ActiveCell.Offset(0, 1).Select

For j = 2 - p To ccMax

  ActiveCell.Value = bArray(j, 1)

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, 1 - p - cMax).Select

For j = 1 To cMax

  ActiveCell.Font.Bold = False

  ActiveCell.Font.Italic = True

  ActiveCell.Value = ""

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(0, -cMax).Select

ActiveCell.Value = "StDev:"

ActiveCell.Offset(0, 1).Select

For j = 2 - p To ccMax

  If vArray(j, j) < 1E-40 Then

    ActiveCell.Value = "<1E-20"

  Else

    ActiveCell.Value = Sqr(vArray(j, j))

  End If

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(1, 1 - p - cMax).Select

For j = 1 To cMax

  ActiveCell.Font.Bold = False

  ActiveCell.Font.Italic = True

  ActiveCell.Value = ""

  ActiveCell.Offset(0, 1).Select

Next j

ActiveCell.Offset(0, 1 - cMax).Select

  ActiveCell.Value = StDevY

ActiveCell.Offset(1, -1).Select

' Compute and display the covariance matrix

If (p = 0 And cMax = 3) Then GoTo LastLine

If p = 0 Then

  For i = 1 To cMax - 2

    For j = 1 To cMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.Font.ColorIndex = 1

      ActiveCell.Value = ""

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -cMax).Select

  Next i

  ActiveCell.Offset(2 - cMax, 0).Select

  ActiveCell.Font.Bold = True

  ActiveCell.Font.ColorIndex = 11

  ActiveCell.Value = "CM:"

  ActiveCell.Offset(0, 1).Select

  For i = 1 To cMax - 2

    For j = 1 To cMax - 2

      ActiveCell.Font.ColorIndex = 11

      ActiveCell.Value = vArray(i, j)

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, 2 - cMax).Select

  Next i

Else

  For i = 1 To cMax - 1

    For j = 1 To cMax

      ActiveCell.Font.Bold = False

      ActiveCell.Font.Italic = True

      ActiveCell.Font.ColorIndex = 1

      ActiveCell.Value = ""

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, -cMax).Select

  Next i

  ActiveCell.Offset(1 - cMax, 0).Select

  ActiveCell.Font.Bold = True

  ActiveCell.Font.ColorIndex = 11

  ActiveCell.Value = "CM:"

  ActiveCell.Offset(0, 1).Select

  For i = 1 To cMax - 1

    For j = 1 To cMax - 1

      ActiveCell.Font.ColorIndex = 11

      ActiveCell.Value = vArray(i, j)

      ActiveCell.Offset(0, 1).Select

    Next j

    ActiveCell.Offset(1, 1 - cMax).Select

  Next i

End If

Application.ScreenUpdating = True

' Provide as optional output the array of linear

' correlation coefficients. The user specifies

' the cell block in which to write this array

If p = 0 And cMax = 3 Then GoTo LastLine

jAnswer = MsgBox("Do you want to see the " _

  & "matrix of linear correlation" _

  & Chr(13) & "coefficients? It will need a " _

  & "block of " & cMax + p - 2 _

  & " by " & cMax + p - 2 & " cells.", vbYesNo, _

  "Least Squares Fit")

OutlineMatrix:

If jAnswer = vbYes Then

  Set myRange = Application.InputBox(Prompt:= _

    "The array should be located in:", Type:=8)

  myRange.Select

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> cMax + p - 2 Then

    MsgBox "The selected range does not have " _

      & cMax + p - 2 & " rows. Please correct.", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

  If Selection.Columns.Count <> cMax + p - 2 Then

    MsgBox "The selected range does not have " _

      & cMax + p - 2 & " columns. Please correct", _

      , "Least Squares Fit"

    GoTo OutlineMatrix

  End If

' Draw a box around the reserved area

    Selection.Borders(xlEdgeTop).LineStyle = xlDashDot

    Selection.Borders(xlEdgeTop).Weight = xlThin

    Selection.Borders(xlEdgeRight).LineStyle = xlDashDot

    Selection.Borders(xlEdgeRight).Weight = xlThin

    Selection.Borders(xlEdgeBottom).LineStyle = xlDashDot

    Selection.Borders(xlEdgeBottom).Weight = xlThin

    Selection.Borders(xlEdgeLeft).LineStyle = xlDashDot

    Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the array of linear correlation coefficients

  If p = 0 Then

    For i = 1 To cMax - 2

      For j = 1 To cMax - 2

        v0Array(i, j) = vArray(i + 1, j + 1)

      Next j

    Next i

    For i = 1 To cMax - 2

      For j = 1 To cMax - 2

        Root = Sqr(v0Array(i, i) * v0Array(j, j))

        lccArray(i, j) = v0Array(i, j) / Root

    Next j

    Next i

    Selection.Value = lccArray

  End If

  If p = 1 Then

    For i = 1 To cMax - 1

      For j = 1 To cMax - 1

        v0Array(i, j) = vArray(i, j)

      Next j

    Next i

    For i = 1 To cMax - 1

      For j = 1 To cMax - 1

        Root = Sqr(vArray(i, i) * vArray(j, j))

        lccArray(i, j) = vArray(i, j) / Root

      Next j

    Next i

    Selection.Value = lccArray

  End If

End If

LastLine:

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^           SOLVER AID          ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

' SolverAid takes the results of Solver and computes the

' corresponding standard deviations. The macro displays the

' covariance matrix as well, and (optionally) the corres-

' sponding array of linear correlation coefficients.

' REQUIREMENTS:

' This macro requires the subroutine Invert

' INPUT:

' The input information must be provided in response to

' three input boxes, which ask for (1) the location of the

' row or column containing the parameters determined by

' Solver; (2) the location of the cell containing the para-

' meter to be minimized, such as SSR; and (3) the COLUMN

' containing the y-values calculated with the Solver-deter-

' mined parameters. Note that these data must contain the

' formulas referring to the Solver-determined parameters;

' a column containing merely their numerical values will

' NOT do.

' OUTPUT:

' Provided that these spaces are unoccupied, the standard

' deviations of the fit, and the standard deviation(s) of

' the parameter(s), will be placed directly, in italics, to

' the right of the corresponding parameters (when these

' were entered as a column), or below them (for row input).

' Otherwise, the result(s) will be displayed in message

' boxes. It is therefore most convenient to leave blank

' the spreadsheet cells to the right of (or below) the

' parameters and of SSR.

' Warning: If space to be used for output contains text or

' other non-numerical information (such as time or date),

' the macro will crash.

' The covariance matrix shows the variances on its main

' (top-left to bottom-right) diagonal, and the correspon-

' ding covariances in the off-diagonal positions. The user

' must specify where on the spreadsheet it should be

' displayed.

' The user also specifies the location on the spreadsheet

' of the optional array of the correlation coefficients,

' for which all elements on the main diagonal are 1, and

' all off-diagonal elements are calculated as r(i,j) =

' v(i,j)/sqr[(v(i)*v(j)].

' PROCEDURE:

' In order to start the process, call the macro, and pro-

' vide the requested addresses (either by typing them, or

' by using the 'point & shoot' method) to the input boxes.

Sub SolverAid()

' NOTATION USED:

'   P1:  single Parameter determined by Solver

'   PP:  multiple parameters determined by Solver

'        These must be in either a single, contiguous row

'        or in a single, contiguous column.

'   SP:  standard deviations on those parameters

'   SSR: the sum of the residuals squared

'        used to optimize Solver

'   SY:  the standard deviation on the function

'   YC:  the Y-values computed with the parameters P

'        NOTE: THESE MUST BE IN A SINGLE, CONTIGUOUS COLUMN

'   c:   prefix denoting the number of columns:

'        cP = columns of PP, cX = columns of XE

'        Note: cP and cX should be 1

'   r:   prefix denoting the number of rows:

'        rP = rows of PP, rY = rows of YC

' Dimension all parameters

Dim cP As Integer, cSSR As Integer, cY As Integer

Dim i As Integer, ii As Integer, j As Integer

Dim n1 As Integer, n2 As Integer, n3 As Integer

Dim rP As Integer, rY As Integer, rrY As Integer

Dim rSSR As Integer

Dim P1Value As Double, Root As Double

Dim SDD1 As Double, SP1 As Double

Dim SP1Value As Double, SSRValue As Double

Dim Sy As Double, SYValue As Double, z As Double

Dim CorrelCoeff As Variant, CovarValue As Variant

Dim PPValue As Variant, SPValue As Variant

Dim YCValue As Variant

Dim YYValue As Variant, YYValue1 As Variant

Dim myRange1 As Range, myRange2 As Range

Dim myRange3 As Range, myRange4 As Range

Dim Answer, iAnswer, iiAnswer, SP1Address, SYAddress

Dim TryAgainA, TryAgainB, TryAgainC, TryAgainD

' Select the computed Solver parameter P1 or parameters

Set myRange1 = Application.InputBox(Prompt:= _

  "The parameters determined by Solver are located in:", _

  Title:="SolverAid InputBox 1: Solver parameters", _

  Type:=8)

myRange1.Select

cP = Selection.Columns.Count

rP = Selection.Rows.Count

If cP = 0 Then End

If cP > 1 And rP > 1 Then

  MsgBox "The Solver-determined parameters" & Chr(13) & _

  "must either be in a single contiguous" & Chr(13) & _

  "column, or in a single contiguous row."

  End

End If

If cP = 1 And rP = 1 Then P1Value = Selection.Value

If cP = 1 And rP > 1 Then PPValue = Selection.Value

If cP > 1 And rP = 1 Then PPValue = Selection.Value

' Replace any zeros in the input parameters

' by small non-zero numbers

If cP = 1 And rP = 1 Then

  P1Value = Selection.Value

  If (IsEmpty(P1Value) Or P1Value = 0) Then P1Value = 1E-20

ElseIf rP > 1 Then

  PPValue = Selection.Value

  For i = 1 To rP

    If (IsEmpty(PPValue(i, 1)) Or PPValue(i, 1) = 0) Then _

      PPValue(i, 1) = 1E-20

  Next i

ElseIf cP > 1 Then

  PPValue = Selection.Value

  For i = 1 To cP

    If (IsEmpty(PPValue(1, i)) Or PPValue(1, i) = 0) Then _

      PPValue(1, i) = 1E-20

  Next i

End If

' Test and prepare the default output range for the

' standard deviations of the parameters, SP, to the

' right of the parameters for a single or columnar

' input, below the parameter for a row input.

n1 = 0

If cP = 1 And rP = 1 Then

  Selection.Offset(0, 1).Select

  SP1Address = Selection.Address

  SP1Value = Selection.Value

  If IsEmpty(SP1Value) Then

    n1 = n1

  Else

    n1 = n1 + 1

  End If

  Selection.Offset(0, -1).Select

ElseIf rP > 1 Then

  Selection.Offset(0, 1).Select

  SPValue = Selection.Value

  For i = 1 To rP

    If IsEmpty(SPValue(i, 1)) Then

      n1 = n1

    Else

      n1 = n1 + 1

    End If

  Next i

  Selection.Offset(0, -1).Select

Else

  Selection.Offset(1, 0).Select

  SPValue = Selection.Value

  For i = 1 To cP

    If IsEmpty(SPValue(1, i)) Then

      n1 = n1

    Else

      n1 = n1 + 1

    End If

  Next i

  Selection.Offset(-1, 0).Select

End If

' Select the computed chi-squared value, SSR

Set myRange2 = Application.InputBox(Prompt:= _

  "The sum of squares of the residuals is located in:", _

  Title:="SolverAid InputBox 2: SSR", Type:=8)

myRange2.Select

cSSR = Selection.Columns.Count

rSSR = Selection.Rows.Count

If cSSR <> 1 Then End

If rSSR <> 1 Then End

SSRValue = Selection.Value

' Test the default output range for the standard devi-

' ation of the fit, SY, to the right of SSR for single

' or columnnar parameter input, below SSR for row input

n2 = 0

If cP = 1 Then

  Selection.Offset(0, 1).Select

  SYValue = Selection.Value

  If (IsEmpty(SYValue) Or SYValue = 0) Then

    n2 = n2

    SYAddress = Selection.Address

  Else

    n2 = n2 + 1

  End If

  Selection.Offset(0, -1).Select

End If

If cP > 1 Then

  Selection.Offset(1, 0).Select

  SYValue = Selection.Value

  If (IsEmpty(SYValue) Or SYValue = 0) Then

    n2 = n2

    SYAddress = Selection.Address

  Else

    n2 = n2 + 1

  End If

  Selection.Offset(-1, 0).Select

End If

' Select the computed Y-values, YC

Set myRange3 = Application.InputBox(Prompt:= _

  "The column containing Ycalc is:", _

  Title:="SolverAid InputBox 3: Ycalc", Type:=8)

myRange3.Select

rY = Selection.Rows.Count

cY = Selection.Columns.Count

If cY <> 1 Then

  MsgBox "The Ycalc values should be in" & _

    Chr(13) & "a single, contiguous COLUMN"

  End

End If

If rY <= rP + 1 Then

  MsgBox " The number N of data pairs must be at least" & _

    Chr(13) & "larger by one than the number of " _

    & "parameters P.", , "SolverAid"

  End

End If

YCValue = Selection.Value

Application.ScreenUpdating = False

' Compute the partial differentials and the

' standard deviations for the one-parameter case

If cP = 1 And rP = 1 Then

  myRange1.Select

  P1Value = P1Value * 1.000001

  Selection = P1Value

  myRange3.Select

  YYValue1 = Selection.Value

' The following loop avoids counting empty rows in the

' computation of the variances and standard deviations

  n3 = 0

  For j = 1 To rY

    If IsEmpty(YYValue1(j, 1)) Then n3 = n3 + 1

  Next j

  rrY = rY - n3

  Sy = Sqr(SSRValue / (rrY - rP))

' Resume to compute the partial differentials and

' the standard deviations for the one-parameter case

  ReDim D1(1 To rY) As Double, DD1(1 To rY) As Double

  SDD1 = 0

  For j = 1 To rY

    D1(j) = (YYValue1(j, 1) - YCValue(j, 1)) / _

      (0.000001 * P1Value)

    DD1(j) = D1(j) * D1(j)

    SDD1 = SDD1 + DD1(j)

  Next j

  P1Value = P1Value / 1.000001

  myRange1.Select

  Selection.Value = P1Value

  SP1 = Sy / Sqr(SDD1)

  If n2 = 0 Then

    myRange2.Select

    Selection.Offset(0, 1).Select

    ActiveCell.Font.Italic = True

    Selection.Value = Sy

  Else

    MsgBox "The standard deviation of the fit is ", _

      , "SolverAid" & Sy

  End If

  If n1 = 0 Then

    myRange1.Select

    Selection.Offset(0, 1).Select

    ActiveCell.Font.Italic = True

    Selection.Value = SP1

  Else

    MsgBox "The standard deviation of the parameter is " _

      & SP1, , "SolverAid"

  End If

' Compute the partial differentials in

' case of a multi-parameter COLUMNAR input

ElseIf rP > 1 Then

  ReDim d(1 To rP, 1 To rY) As Double

  For i = 1 To rP

    myRange1.Select

    PPValue(i, 1) = PPValue(i, 1) * 1.000001

    Selection.Value = PPValue

    myRange3.Select

    YYValue = Selection.Value

' The following loop avoids counting empty rows in the

' computation of the variances and standard deviations

    n3 = 0

    For j = 1 To rY

      If IsEmpty(YYValue(j, 1)) Then n3 = n3 + 1

    Next j

    rrY = rY - n3

    Sy = Sqr(SSRValue / (rrY - rP))

' Resume the computation of the partial differentials and

' the standard deviations for the multi-parameter case

    For j = 1 To rY

      d(i, j) = (YYValue(j, 1) - YCValue(j, 1)) / _

        (0.000001 * PPValue(i, 1))

    Next j

    PPValue(i, 1) = PPValue(i, 1) / 1.000001

  Next i

  myRange1.Select

  Selection.Value = PPValue

  ReDim DD(1 To rP, 1 To rP, 1 To rY) As Double

  For i = 1 To rP

    For ii = 1 To rP

      For j = 1 To rY

        DD(i, ii, j) = d(i, j) * d(ii, j)

      Next j

    Next ii

  Next i

  ReDim SDD(1 To rP, 1 To rP) As Double

  ReDim SDDInv(1 To rP, 1 To rP) As Double

  For i = 1 To rP

    For ii = 1 To rP

      SDD(i, ii) = 0

    Next ii

  Next i

  For i = 1 To rP

    For ii = 1 To rP

      For j = 1 To rY

        SDD(i, ii) = SDD(i, ii) + DD(i, ii, j)

      Next j

    Next ii

  Next i

  Call Invert(SDD, rP, SDDInv)

' Select the option, and the block

' in which to write the matrix

TryAgainA:

  Application.ScreenUpdating = True

  Set myRange4 = Application.InputBox(Prompt:= _

    "The covariance matrix needs a block" & Chr(13) & _

    "of " & rP & " by " & rP & " cells." & _

    " It should be located in:", Type:=8)

  myRange4.Select

  CovarValue = Selection.Value

  CorrelCoeff = Selection.Value

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> rP Then

    MsgBox "The selected range does not have " & _

      rP & " rows." & Chr(13) & _

      "Push OK and then select a proper block."

    GoTo TryAgainA

  End If

  If Selection.Columns.Count <> rP Then

    MsgBox "The selected range does not have " & rP & _

      " columns." & Chr(13) & _

      "Push OK and then select a proper block."

    GoTo TryAgainA

  End If

' Draw a box around the selected block

  Selection.Borders(xlEdgeTop).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeTop).Weight = xlThin

  Selection.Borders(xlEdgeRight).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeRight).Weight = xlThin

  Selection.Borders(xlEdgeBottom).LineStyle = _

    xlDashDotDot

  Selection.Borders(xlEdgeBottom).Weight = xlThin

  Selection.Borders(xlEdgeLeft).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the covariance matrix in the selected block

  For i = 1 To rP

    For j = 1 To rP

      CovarValue(i, j) = Sy * Sy * SDDInv(i, j)

    Next j

  Next i

  For i = 1 To rP

    For j = 1 To rP

      Root = Sqr(CovarValue(i, i) * _

        CovarValue(j, j))

      CorrelCoeff(i, j) = CovarValue(i, j) / Root

    Next j

  Next i

  Selection.Value = CovarValue

' Provide the optional display of the

' linear correlation coefficients

  iAnswer = MsgBox("Do you also want to see the" _

    & Chr(13) & "linear correlation coefficients?", _

    vbYesNo, "SolverAid")

TryAgainB:

  If iAnswer = vbYes Then

    Application.ScreenUpdating = True

    Set myRange4 = Application.InputBox(Prompt:= _

      "The array of linear correlation coefficients" & _

      "also needs" & Chr(13) & "a block of " & rP & _

      " by " & rP & " cells." & _

      " It should be located in:", Type:=8)

    myRange4.Select

' Make sure that the selected block has the correct size

    If Selection.Rows.Count <> rP Then

      MsgBox "The selected range does not have " & _

        rP & " rows." & Chr(13) & _

        "Push OK and then select a proper block."

      GoTo TryAgainB

    End If

    If Selection.Columns.Count <> rP Then

      MsgBox "The selected range does not have " & rP & _

        " columns." & Chr(13) & _

        "Push OK and then select a proper block."

      GoTo TryAgainB

    End If

' Draw a box around the selected block

    Selection.Borders(xlEdgeTop).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeTop).Weight = xlThin

    Selection.Borders(xlEdgeRight).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeRight).Weight = xlThin

    Selection.Borders(xlEdgeBottom).LineStyle = _

      xlDashDotDot

    Selection.Borders(xlEdgeBottom).Weight = xlThin

    Selection.Borders(xlEdgeLeft).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the linear correlation coefficients

' in the selected block

    Selection.Value = CorrelCoeff

  End If

  Application.ScreenUpdating = True

' Display the standard deviations of the

' Solver parameters

  For i = 1 To rP

    SPValue(i, 1) = Sy * Sqr(SDDInv(i, i))

  Next i

  If n1 > 0 Then

    Answer = MsgBox("There are data in the cells to the " _

      & "right of the" & Chr(13) & "Solver parameters. " _

      & "Can they be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then n1 = 0

  End If

  If n1 = 0 Then

    myRange1.Select

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    For i = 1 To rP

      Selection.Value = SPValue

    Next i

  Else

    For i = 1 To rP

      MsgBox "The standard deviation of parameter #" & _

        i & " is " & SPValue(i, 1), , "SolverAid"

    Next i

  End If

' Display the standard deviation of the over-all fit

  If n2 > 0 Then

    Answer = MsgBox("There are data in the cells to the" _

      & Chr(13) & "right of the sum of squares of the" _

      & Chr(13) & "residuals. Can they be overwritten?", _

      vbYesNo, "Overwrite?")

    If Answer = vbYes Then n2 = 0

  End If

  If n2 = 0 Then

    myRange2.Select

    Selection.Offset(0, 1).Select

    ActiveCell.Font.Italic = True

    Selection.Value = Sy

  Else

    MsgBox "The standard deviation of the fit is " & Sy

  End If

  myRange1.Select

' Compute the partial differentials

' in case of a multi-parameter ROW input

ElseIf cP > 1 Then

  ReDim d(1 To cP, 1 To rY) As Double

  For i = 1 To cP

    myRange1.Select

    PPValue(1, i) = PPValue(1, i) * 1.000001

    Selection.Value = PPValue

    myRange3.Select

    YYValue = Selection.Value

' The following loop avoids counting empty rows in the

' computation of the variances and standard deviations

    n3 = 0

    For j = 1 To rY

      If IsEmpty(YYValue(j, 1)) Then n3 = n3 + 1

    Next j

    rrY = rY - n3

    Sy = Sqr(SSRValue / (rrY - cP))

' Resume the computation of the partial differentials and

' the standard deviations for the multi-parameter case

    For j = 1 To rY

      d(i, j) = (YYValue(j, 1) - YCValue(j, 1)) / _

        (0.000001 * PPValue(1, i))

    Next j

    PPValue(1, i) = PPValue(1, i) / 1.000001

  Next i

  myRange1.Select

  Selection.Value = PPValue

  ReDim DD(1 To cP, 1 To cP, 1 To rY) As Double

  For i = 1 To cP

    For ii = 1 To cP

      For j = 1 To rY

        DD(i, ii, j) = d(i, j) * d(ii, j)

      Next j

    Next ii

  Next i

  ReDim SDD(1 To cP, 1 To cP) As Double

  ReDim SDDInv(1 To cP, 1 To cP) As Double

  For i = 1 To cP

    For ii = 1 To cP

      SDD(i, ii) = 0

    Next ii

  Next i

  For i = 1 To cP

    For ii = 1 To cP

      For j = 1 To rY

        SDD(i, ii) = SDD(i, ii) + DD(i, ii, j)

      Next j

    Next ii

  Next i

  Call Invert(SDD, cP, SDDInv)

' Select the option, and the block

' in which to write the matrix

TryAgainC:

  Application.ScreenUpdating = True

  Set myRange4 = Application.InputBox(Prompt:= _

    "The covariance matrix needs a block" & Chr(13) & _

    "of " & cP & " by " & cP & " cells." & _

    " It should be located in:", Type:=8)

  myRange4.Select

  CovarValue = Selection.Value

  CorrelCoeff = Selection.Value

' Make sure that the selected block has the correct size

  If Selection.Rows.Count <> cP Then

    MsgBox "The selected range does not have " & _

      cP & " rows." & Chr(13) & _

      "Push OK and then select a proper block."

    GoTo TryAgainC

  End If

  If Selection.Columns.Count <> cP Then

    MsgBox "The selected range does not have " & cP & _

      " columns." & Chr(13) & _

      "Push OK and then select a proper block."

    GoTo TryAgainC

  End If

' Draw a box around the selected block

  Selection.Borders(xlEdgeTop).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeTop).Weight = xlThin

  Selection.Borders(xlEdgeRight).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeRight).Weight = xlThin

  Selection.Borders(xlEdgeBottom).LineStyle = _

    xlDashDotDot

  Selection.Borders(xlEdgeBottom).Weight = xlThin

  Selection.Borders(xlEdgeLeft).LineStyle = xlDashDotDot

  Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the covariance matrix in the selected block

  For i = 1 To cP

    For j = 1 To cP

      CovarValue(i, j) = Sy * Sy * SDDInv(i, j)

    Next j

  Next i

  For i = 1 To cP

    For j = 1 To cP

      Root = Sqr(CovarValue(i, i) * _

        CovarValue(j, j))

      CorrelCoeff(i, j) = CovarValue(i, j) / Root

    Next j

  Next i

  Selection.Value = CovarValue

' Provide the optional display of the

' linear correlation coefficients

  iAnswer = MsgBox("Do you also want to see the" _

    & Chr(13) & "linear correlation coefficients?", _

    vbYesNo, "SolverAid")

TryAgainD:

  If iAnswer = vbYes Then

    Application.ScreenUpdating = True

    Set myRange4 = Application.InputBox(Prompt:= _

      "The array of linear correlation coefficients" & _

      "also needs" & Chr(13) & "a block of " & cP & _

      " by " & cP & " cells." & _

      " It should be located in:", Type:=8)

    myRange4.Select

' Make sure that the selected block has the correct size

    If Selection.Rows.Count <> cP Then

      MsgBox "The selected range does not have " & _

        cP & " rows." & Chr(13) & _

        "Push OK and then select a proper block."

      GoTo TryAgainD

    End If

    If Selection.Columns.Count <> cP Then

      MsgBox "The selected range does not have " & cP & _

        " columns." & Chr(13) & _

        "Push OK and then select a proper block."

      GoTo TryAgainD

    End If

' Draw a box around the selected block

    Selection.Borders(xlEdgeTop).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeTop).Weight = xlThin

    Selection.Borders(xlEdgeRight).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeRight).Weight = xlThin

    Selection.Borders(xlEdgeBottom).LineStyle = _

      xlDashDotDot

    Selection.Borders(xlEdgeBottom).Weight = xlThin

    Selection.Borders(xlEdgeLeft).LineStyle = xlDashDotDot

    Selection.Borders(xlEdgeLeft).Weight = xlThin

' Write the linear correlation coefficients

' in the selected block

    Selection.Value = CorrelCoeff

  End If

  Application.ScreenUpdating = True

' Display the standard deviations of the

' Solver parameters

  For i = 1 To cP

    SPValue(1, i) = Sy * Sqr(SDDInv(i, i))

  Next i

  If n1 > 0 Then

    Answer = MsgBox("There are data in the cells " _

      & "below the" & Chr(13) & "Solver parameters. " _

      & "Can they be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then n1 = 0

  End If

  If n1 = 0 Then

    myRange1.Select

    Selection.Offset(1, 0).Select

    Selection.Font.Italic = True

    For i = 1 To cP

      Selection.Value = SPValue

    Next i

  Else

    For i = 1 To cP

      MsgBox "The standard deviation of parameter #" & _

        i & " is " & SPValue(1, i), , "SolverAid"

    Next i

  End If

' Display the standard deviation of the over-all fit

  If n2 > 0 Then

    Answer = MsgBox("There are data in the cells" _

      & Chr(13) & "below the sum of squares of the" _

      & Chr(13) & "residuals. Can they be overwritten?", _

      vbYesNo, "Overwrite?")

    If Answer = vbYes Then n2 = 0

  End If

  If n2 = 0 Then

    myRange2.Select

    Selection.Offset(1, 0).Select

    ActiveCell.Font.Italic = True

    Selection.Value = Sy

  Else

    MsgBox "The standard deviation of the fit is " & Sy

  End If

  myRange1.Select

End If

End Sub

Sub Propagation()

' PURPOSE:

' This macro computes the propagated standard deviation in

' a single function F based on N input parameters, based on

' their standard deviations or on the corresponding covari-

' ance matrix. When only the standard deviations are given,

' the macro will assume that the input parameters are mutu-

' ally independent. No such assumption will be made when

' the covariance matrix is provided. For a single input

' parameter, there is no distinction between the two ap-

' proaches. For more than one input parameter, the macro

' recognizes which of the two computations to perform

' because the standard deviations will be provided as a

' vector, while the covariance matrix has the form of a

' square data array. The components of the N by N covari-

' ance matrix are assumed to be in the same order as those

' of the N input parameters.

' SUBROUTINES:

' This macro does not require any subroutines

' INPUT:

' The N independent input parameter values must be placed

' either in a contiguous row or in a contiguous column.

' THEY MUST BE NUMBERS, i.e., they cannot be formulas,

' i.e., equations.

' The N standard deviations must follow the same format,

' again either in a contiguous row or column, consistent

' with the format of the input data, but can be either

' values or formulas.

' The ORDER of the input parameters and of the standard

' deviations must be the same, because the (partial)

' derivative of the function and the corresponding standard

' deviation are combined strictly on the basis of their

' sequential order. The same applies to the covariance

' matrix if this is used instead.

' OUTPUT:

' The standard deviation of the single function F will be

' placed directly to the right of (or below) that function,

' in italics, provided that this cell is either unoccupied

' or its contents can be overwritten. Otherwise, the result

' will be displayed in a message box.

' PROCEDURE:

' In order to start this macro, call it. There is no need

' to highlight anything beforehand.

' You will see an input box in which to place (either

' by typing or by the 'point-and-shoot' method) the

' address(es) of the input parameter(s). After you have

' entered these, a second input box will request the

' addresses of either the standard deviations or the

' covariance matrix. These should have been arranged in

' the same order as the earlier-entered parameters.

' Finally, a third input box will ask for the address of

' the function. The output will be provided either on the

' spreadsheet, or through message box(es).

' NOTATION:

'   N:   the number of input parameters

'   X:   single input parameter (for N=1)

'   S:   the corresponding, single standard deviation of X

'   Xi:  multiple input parameters (for N>1) NOTE: THESE

'        MUST BE IN A SINGLE, CONTIGUOUS ROW OR COLUMN

'   Si:  standard deviations of the multiple input para-

'        meters. NOTE: THESE MUST BE IN A SINGLE, CONTI-

'        GUOUS ROW OR COLUMN

'   CM:  the covariance matrix

'   F:   the single function through which the error(s)

'        propagate(s)

'   VF:  the propagated variance of the function F

'   SF:  the propagated standard deviation of function F

' We distinguish five cases:

' C = 1  one parameter P, one uncertainty U

' C = 2  parameters P and uncertainties U in column format

' C = 3  parameters P and uncertainties U in row format

' For C = 1 to 3, the uncertainty is the standard deviation

' C = 4  parameters P in column, uncertainty U in matrix

' C = 5  parameters P in row, uncertainty U in matrix

' For C = 4 or 5, the uncertainty must be in the form of

'   a covariance matrix, i.e., the uncertainties are in

'   terms of variances (the squares of the standard

'   deviations) and covariances (which may be zero).

Dim C As Integer    'Case selector

Dim i As Integer, j As Integer, m As Integer

Dim number As Integer, LCCTest As Integer

Dim n As Integer        'Larger dimension of

        'input parameter set

Dim NCF As Integer  'Number of Columns of output Function

Dim NCP As Integer  'Number of Columns of input Parameters

Dim NCU As Integer  'Number of Columns of the Uncertainty

        'estimate

Dim NRF As Integer  'Number of Rows of output Function

Dim NRP As Integer  'Number of Rows of input Parameters

Dim NRU As Integer      'Number of Rows in the Uncertainty

        'estimate

Dim FValue As Double, FFValue As Double

Dim newFiValue As Double, newFValue As Double

Dim newXValue As Double, SFValue As Double

Dim SValue As Double, VFiValue As Double

Dim VFValue As Double, XValue As Double

Dim DelValue As Variant, newXiValue As Variant

Dim SFiValue As Variant, SiValue As Variant

Dim CMValue As Variant, CMiValue As Variant

Dim XiValue As Variant

Dim myRange2 As Range, myRange3 As Range

Dim myRange1 As Range

Dim Answer, Prompt, Title

' Select the input parameters of the function

Prompt = "The input parameters of the" & Chr(13) & _

  "function are located in:"

Title = "Uncertainty Propagation InputBox 1: Input" _

  & " Parameters "

Set myRange1 = Application.InputBox(Prompt, Title, _

  Type:=8)

myRange1.Select

NRP = Selection.Rows.Count

NCP = Selection.Columns.Count

' Check the type of input

If NRP = 0 Then End

If NRP <> 1 And NCP <> 1 Then

  MsgBox "The input parameters should be placed" _

  & Chr(13) & "either in a single contiguous row," _

  & Chr(13) & "or in a single contiguous column.", _

    , "Propagation of uncertainty"

  End

ElseIf NRP = 1 And NCP = 1 Then

  n = 1

  XValue = Selection.Value

ElseIf NRP > 1 And NCP = 1 Then

  n = NRP

  XiValue = Selection.Value

  DelValue = Selection.Value        ' dimensioning the array

  newXiValue = Selection.Value      ' dimensioning the array

  SFiValue = Selection.Value        ' dimensioning the array

ElseIf NCP > 1 And NRP = 1 Then

  n = NCP

  XiValue = Selection.Value

  DelValue = Selection.Value        ' dimensioning the array

  newXiValue = Selection.Value      ' dimensioning the array

  SFiValue = Selection.Value        ' dimensioning the array

End If

' Select the uncertainty estimates (standard deviations

' or covariance matrix) of the input parameters

A:

Prompt = " The standard deviations, or" _

  & Chr(13) & "the covariance matrix, of the" _

  & Chr(13) & " input parameters are located in:"

Title = "Uncertainty Propagation InputBox 2: " _

  & "Uncertainty estimates"

Set myRange2 = Application _

  .InputBox(Prompt, Title, Type:=8)

myRange2.Select

NRU = Selection.Rows.Count

NCU = Selection.Columns.Count

' Verify that the number of parameters

' matches that of the uncertainty estimates

number = NCP + NRP - 1

If NCU = NRU Then

  If (NCU + NRU) / 2 <> number Then

    MsgBox "The dimension of the covariance matrix does" _

      & Chr(13) & "not match the number of input " _

      & "parameters"

    End

  End If

End If

If NCU <> NRU Then

  If NCU + NRU - 1 <> number Then

    MsgBox "The number of uncertainty estimates does" _

      & Chr(13) & "not match the number of input " _

      & "parameters"

    End

  End If

End If

' Verify and categorize the inputs,

' then read the uncertainties

If NCU = 0 Then End

If n = 1 And NRU = 1 And NCU = 1 Then

  C = 1     'only one input parameter

  SValue = Selection.Value

End If

If n > 1 Then   'multiple input parameters:

  If n = NRU And NCU = 1 Then

    C = 2   'the independent parameters and

            'their standard deviations are

            'located in single columns

    SiValue = Selection.Value

  End If

  If n <> NRU And NCU = 1 Then

    MsgBox "The number of entries in the two" _

      & Chr(13) & "input columns must be the same.", _

      , "Propagation of uncertainty"

    End

  End If

  If n = NCU And NRU = 1 Then

    C = 3   'the independent parameters and

            'their standard deviations are

            'located in single rows

    SiValue = Selection.Value

    If n <> NCU And NRU = 1 Then

      MsgBox "The number of entries in the two" _

        & Chr(13) & "input rows must be the same.", _

        , "Propagation of uncertainty"

      End

    End If

  End If

  If n = NCU And n = NRU Then

                'mutually dependent parameters:

    If NCP <> NCU And NRP <> NRU Then

      MsgBox "The number of entries for the " _

        & "uncertainties does" & Chr(13) & _

        "not match the number of input parameters." _

        , , "Propagation of uncertainty"

      End

    End If

    If n = NRP And NCP = 1 Then

      C = 4 'the dependent parameters are located

            'in a single column of N items, and

            'the covariance matrix is N by N

      CMValue = Selection.Value

      CMiValue = Selection.Value

    End If

    If n = NCP And NRP = 1 Then

      C = 5 'the dependent parameters are located

            'in a single row of N items, and

            'the covariance matrix is N by N

      CMValue = Selection.Value

      CMiValue = Selection.Value

    End If

  End If

End If

If NCU > 1 And NRU > 1 And NCU <> NRU Then

  MsgBox "The covariance matrix has an" & _

    Chr(13) & " incorrect dimension. Try again.", _

    , "Propagation of uncertainty"

  GoTo A

End If

If NCU > 1 And NRU > 1 Then

  LCCTest = 0

  For i = 1 To n

    If CMValue(i, i) > 0.99999 And _

      CMValue(i, i) < 1.00001 Then LCCTest = LCCTest + 1

  Next i

  If LCCTest = n Then

    Answer = MsgBox("You appear to have entered the" _

      & Chr(13) & "array of linear correlation " _

      & "coefficients" & Chr(13) & "rather than the " _

      & "covariance matrix." & Chr(13) _

      & "Do you want to correct the input?", _

      vbYesNo, "Propagation of uncertainty")

    If Answer = vbYes Then GoTo A

  End If

End If

' Select the function

Set myRange3 = Application.InputBox(Prompt:= _

  "The function is located in:", Title:= _

  "Uncertainty Propagation InputBox 3: Function", Type:=8)

myRange3.Select

NRF = Selection.Rows.Count

NCF = Selection.Columns.Count

If NRF = 0 Then End

If NRF = 1 And NCF = 1 Then FValue = Selection.Value

If NRF <> 1 Or NCF <> 1 Then

  MsgBox "The propagation is computed only for a SINGLE " _

    & "function.", , "Propagation of uncertainty"

  End

End If

' Test and prepare the default output cell for the

' standard deviation of the function F

' Place the standard deviation to the right of the cell

' containing F for a single input parameter, or when the

' input parameters are in a single column

If n = 1 Or NCP = 1 Then

  Selection.Offset(0, 1).Select

  m = 0

  FFValue = Selection.Value

  If (IsEmpty(FFValue) Or FFValue = 0) Then

    m = m

  Else

    m = m + 1

  End If

  Selection.Offset(0, -1).Select

End If

' Place the standard deviation below F when the

' input parameters are organized in a single row.

If n > 1 And NRP = 1 Then

  Selection.Offset(1, 0).Select

  m = 0

  FFValue = Selection.Value

  If (IsEmpty(FFValue) Or FFValue = 0) Then

    m = m

  Else

    m = m + 1

  End If

  Selection.Offset(-1, 0).Select

End If

Application.ScreenUpdating = False

VFValue = 0

myRange1.Select

' Compute and display the sought standard deviation SFValue

Select Case C

Case 1

  newXValue = XValue * 1.000001

  Selection.Replace XValue, newXValue

  myRange3.Select

  newFValue = Selection.Value

  myRange1.Select

  Selection.Replace newXValue, XValue

  SFValue = Abs(1000000 * (newFValue - FValue) * SValue _

    / XValue)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox _

      ("The cell to the right of the function is " _

      & "not empty." & Chr(13) & "Can its contents " _

      & "be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    Selection.Value = SFValue

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & SFValue

  End If

Case 2

  For i = 1 To n

    newXiValue(i, 1) = XiValue(i, 1) * 1.000001

    Selection.Item(i, 1).Replace XiValue(i, 1), newXiValue(i, 1)

    myRange3.Select

    newFiValue = Selection.Value

    myRange1.Select

    Selection.Item(i, 1).Replace newXiValue(i, 1), XiValue(i, 1)

    VFiValue = 1000000 * (newFiValue - FValue) _

      * SiValue(i, 1) / XiValue(i, 1)

    VFValue = VFValue + VFiValue ^ 2

  Next i

  SFValue = Sqr(VFValue)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox _

      ("The cell to the right of the function is " _

      & "not empty." & Chr(13) & "Can its contents " _

      & "be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    Selection.Value = SFValue

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & SFValue

  End If

Case 3

  For j = 1 To n

    newXiValue(1, j) = XiValue(1, j) * 1.000001

    Selection.Item(1, j).Replace XiValue(1, j), newXiValue(1, j)

    myRange3.Select

    newFiValue = Selection.Value

    myRange1.Select

    Selection.Item(1, j).Replace newXiValue(1, j), XiValue(1, j)

    VFiValue = 1000000 * (newFiValue - FValue) _

      * SiValue(1, j) / XiValue(1, j)

    VFValue = VFValue + VFiValue ^ 2

  Next j

  SFValue = Sqr(VFValue)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox("The cell below the function is " _

    & "not empty." & Chr(13) & "Can its contents be " _

    & "overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(1, 0).Select

    Selection.Font.Italic = True

    Selection.Value = SFValue

  Else

  End If

Case 4

  For i = 1 To n

    newXiValue(i, 1) = XiValue(i, 1) * 1.000001

    Selection.Item(i, 1).Replace XiValue(i, 1), newXiValue(i, 1)

    myRange3.Select

    newFiValue = Selection.Value

    myRange1.Select

    Selection.Item(i, 1).Replace newXiValue(i, 1), XiValue(i, 1)

    DelValue(i, 1) = 1000000 * (newFiValue - FValue) _

      / XiValue(i, 1)

  Next i

  For j = 1 To n

    For i = 1 To n

      VFValue = VFValue + DelValue(i, 1) _

        * DelValue(j, 1) * CMValue(i, j)

    Next i

  Next j

  SFValue = Sqr(VFValue)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox _

    ("The cell to the right of the function is " _

    & "not empty." & Chr(13) & "Can its contents " _

    & "be overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(0, 1).Select

    Selection.Font.Italic = True

    Selection.Value = SFValue

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & SFValue

  End If

Case 5

  For i = 1 To n

    newXiValue(1, i) = XiValue(1, i) * 1.000001

    Selection.Item(1, i).Replace XiValue(1, i), newXiValue(1, i)

    myRange3.Select

    newFiValue = Selection.Value

    myRange1.Select

    Selection.Item(1, i).Replace newXiValue(1, i), XiValue(1, i)

    DelValue(1, i) = 1000000 * (newFiValue - FValue) _

      / XiValue(1, i)

  Next i

  For i = 1 To n

    For j = 1 To n

      VFValue = VFValue + DelValue(1, i) _

        * DelValue(1, j) * CMValue(j, i)

    Next j

  Next i

  SFValue = Sqr(VFValue)

  myRange3.Select

  If m > 0 Then

    Answer = MsgBox("The cell below the function is " _

      & "not empty." & Chr(13) & "Can its contents be " _

      & "overwritten?", vbYesNo, "Overwrite?")

    If Answer = vbYes Then m = 0

  End If

  If m = 0 Then

    Selection.Offset(1, 0).Select

    Selection.Font.Italic = True

    Selection.Value = SFValue

  Else

    MsgBox "The standard deviation of the function is ", _

      , "Propagation of uncertainty" & SFValue

  End If

End Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''''''              '''''''''''''''''''''

'''''''''''''''''''''''     INVERT     ''''''''''''''''''''

''''''''''''''''''''''''              '''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Invert(Min, nrc, Mout)

' The square input and output matrices are Min and Mout

' respectively; nrc is the number of rows and columns in

' Min and Mout

Dim i As Integer, icol As Integer, irow As Integer

Dim j As Integer, k As Integer, L As Integer, LL As Integer

Dim big As Double, dummy As Double

Dim n As Integer, pivinv As Double

Dim u As Double

n = nrc + 1

ReDim bb(1 To n, 1 To n) As Double

ReDim ipivot(1 To n) As Double

ReDim Index(1 To n) As Double

ReDim indexr(1 To n) As Double

ReDim indexc(1 To n) As Double

u = 1

' Copy the input matrix in order to retain it

For i = 1 To nrc

  For j = 1 To nrc

    Mout(i, j) = Min(i, j)      'Min rather than M1

  Next j

Next i

' The following is the Gauss-Jordan elimination routine

' GAUSSJ from J. C. Sprott, "Numerical Recipes: Routines

' and Examples in BASIC", Cambridge University Press,

' Copyright (C)1991 by Numerical Recipes Software. Used by

' permission. Use of this routine other than as an integral

' part of the present book requires an additional license

' from Numerical Recipes Software. Further distribution is

' prohibited. The routine has been modified to yield

' double-precision results.

For j = 1 To nrc

  ipivot(j) = 0

Next j

For i = 1 To nrc

  big = 0

  For j = 1 To nrc

    If ipivot(j) <> u Then

      For k = 1 To nrc

        If ipivot(k) = 0 Then

          If Abs(Mout(j, k)) >= big Then

            big = Abs(Mout(j, k))

            irow = j

            icol = k

          End If

          ElseIf ipivot(k) > 1 Then Exit Sub

        End If

      Next k

    End If

  Next j

  ipivot(icol) = ipivot(icol) + 1

  If irow <> icol Then

    For L = 1 To nrc

      dummy = Mout(irow, L)

      Mout(irow, L) = Mout(icol, L)

      Mout(icol, L) = dummy

    Next L

    For L = 1 To nrc

      dummy = bb(irow, L)

      bb(irow, L) = bb(icol, L)

      bb(icol, L) = dummy

    Next L

  End If

  indexr(i) = irow

  indexc(i) = icol

  If Mout(icol, icol) = 0 Then Exit Sub

  pivinv = u / Mout(icol, icol)

  Mout(icol, icol) = u

  For L = 1 To nrc

    Mout(icol, L) = Mout(icol, L) * pivinv

    bb(icol, L) = bb(icol, L) * pivinv

  Next L

  For LL = 1 To nrc

    If LL <> icol Then

      dummy = Mout(LL, icol)

      Mout(LL, icol) = 0

      For L = 1 To nrc

        Mout(LL, L) = Mout(LL, L) - Mout(icol, L) * dummy

        bb(LL, L) = bb(LL, L) - bb(icol, L) * dummy

      Next L

    End If

  Next LL

Next i

For L = nrc To 1 Step -1

  If indexr(L) <> indexc(L) Then

    For k = 1 To nrc

      dummy = Mout(k, indexr(L))

      Mout(k, indexr(L)) = Mout(k, indexc(L))

      Mout(k, indexc(L)) = dummy

    Next k

  End If

Next L

Erase indexc, indexr, ipivot

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''''                '''''''''''''''''''''

'''''''''''''''''''''     MULTIPLY     ''''''''''''''''''''

''''''''''''''''''''''                '''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Multiply(M1, r1, c1, M2, c2, Mout)

' Computes the product of two matrices: Mout = M1 times M2

'   r1: number of rows in M1

'   c1: number of columns in M1

'   c2: number of columns in M2

' M2 must have c1 rows

' Mout will have r1 rows and c2 columns

Dim i As Integer, j As Integer, k As Integer

For i = 1 To r1

  For j = 1 To c2

    Mout(i, j) = 0

    For k = 1 To c1

      Mout(i, j) = Mout(i, j) + M1(i, k) * M2(k, j)

    Next k

  Next j

Next i

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''''                 ''''''''''''''''''''

'''''''''''''''''''''     TRANSPOSE     '''''''''''''''''''

''''''''''''''''''''''                 ''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Transpose(M1, r1, c1, Mout)

' Computes the transpose Mout of matrix M1

'   r1: number of rows in M1

'   c1: number of columns in M1

' Mout will have c1 rows and r1 columns

Dim i As Integer, j As Integer

For i = 1 To c1

  For j = 1 To r1

    Mout(i, j) = M1(j, i)

  Next j

Next i

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^     FOURIER TRANSFORMATION    ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v4, May 1, 2003

' PURPOSE:

' The macros ForwardFT and InverseFT perform fast Fourier

' transformations on an array of complex data, arranged in

' a single block of three adjacent columns.

' SUBROUTINES:

' The macros are merely drivers that set iSign to +1 (for

' the forward transform) or to -1 (for inverse transforma-

' tion). The main subroutine Fourier then reads the input

' information, makes a number of checks, and computes the

' output. For the actual Fourier transformation it then

' calls the subroutine FT. This latter part is placed in a

' separate subroutine in order to make it readily available

' to other macros, such as ConvolveFT, DeconvolveFT,

' DeconvolveIt, and Gabor.

' INPUT:

' The first column must contain the variable (e.g., time,

' or frequency), the second column the real components of

' the input data, the third column their imaginary com-

' ponents. For purely real input data, either leave the

' third column blank, or fill it with zeros. Likewise, for

' purely imaginary input data, the second column should

' only contain zeros, blanks, or combinations thereof.

' The macro accepts two input formats: (1) data with time

' or frequency values centered around zero (jSign = -1),

' and (2) data with time or frequency values starting at

' zero (jSign = 1). The output format is commensurate with

' the input format used.

' OUTPUT:

' The output is written in the three columns to the right

' of the input data block, thereby overwriting any prior

' data in that region. The output columns contain (from

' left to right) frequency or time, the real components

' of the output data, and their imaginary components.

' PROCEDURE:

' To initiate the transform, highlight the three columns

' of the input array, and call one of the two macros,

' ForwardFT() or InverseFT().

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub ForwardFT()

Dim iSign As Integer

iSign = 1

Call Fourier(iSign)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub InverseFT()

Dim iSign As Integer

iSign = -1

Call Fourier(iSign)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Fourier(iSign)

' Check the array length n, which must be a power of 2,

' and be at least 2

Dim cn As Integer, cnMax As Integer, jSign As Integer

Dim n As Integer, NN As Integer

Dim rn As Integer, rnMax As Integer

Dim Length As Single

Dim Check1 As Double, Check2 As Double

Dim Denom As Double, Interval As Double, z As Double

Dim DataArray As Variant, outputArray As Variant

Dim myRange As Range

Dim Ans, hAnswer

Begin:

n = 0

rnMax = Selection.Rows.Count

Length = CSng(rnMax)

cnMax = Selection.Columns.Count

' If area was not highlighted

If rnMax = 1 And cnMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Fourier transformation")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

If Length < 2 Then

  MsgBox "There must be at least two rows."

  End

End If

Do While Length > 1

  Length = Length / 2

Loop

If Length <> 1 Then

  MsgBox "The current number of rows is " & rnMax _

    & "which is not an integer power of two."

  End

End If

' Check that there are three input columns

If cnMax <> 3 Then

  MsgBox "There must be three input columns," _

    & Chr(13) & "one for the variable (time, frequen-," _

    & Chr(13) & "cy, etc.), the next two for the real" _

    & Chr(13) & "and imaginary parts of the input data."

  End

End If

' Read the input data

DataArray = Selection.Value

' Check that the first column has its first two elements

Check1 = VarType(DataArray(1, 1))

If Check1 = 0 Then

  MsgBox "Enter the top left value."

  End

End If

Check2 = VarType(DataArray(2, 1))

If Check2 = 0 Then

  MsgBox "Enter a value in row 2 of the first column."

  End

End If

' Determine what input convention is used:

' jSign = -1 for input data centered around zero,

' jSign = 1 for input data starting at zero

jSign = 0

Interval = (DataArray(2, 1) - DataArray(1, 1)) * rnMax

If DataArray(1, 1) > (-0.5 * Interval / rnMax) And _

  DataArray(1, 1) < (0.5 * Interval / rnMax) Then jSign = 1

If DataArray(1, 1) < (-0.5 * Interval / rnMax) And _

  DataArray(rnMax / 2 + 1, 1) > (-0.5 * Interval / rnMax) _

  And DataArray(rnMax / 2 + 1, 1) < (0.5 * Interval / _

  rnMax) Then jSign = -1

If jSign = 0 Then

  MsgBox "The input format is incorrect." _

    & Chr(13) & "It should either be centered" _

    & Chr(13) & "around zero, or start at zero."

  End

End If

' Read and rearrange the input data

NN = 2 * rnMax

ReDim Term(NN) As Double

If jSign = 1 Then

  For rn = 1 To rnMax

    Term(2 * rn - 1) = DataArray(rn, 2)

    Term(2 * rn) = DataArray(rn, 3)

  Next rn

End If

If jSign = -1 Then

  For rn = 1 To rnMax / 2

    Term(2 * rn - 1) = DataArray(rnMax / 2 + rn, 2)

    Term(2 * rn) = DataArray(rnMax / 2 + rn, 3)

    Term(rnMax + 2 * rn - 1) = DataArray(rn, 2)

    Term(rnMax + 2 * rn) = DataArray(rn, 3)

  Next rn

End If

' Check that the output does not overwrite valuable data

Selection.Offset(0, 3).Select

outputArray = Selection.Value

For rn = 1 To rnMax

  For cn = 1 To cnMax

    z = outputArray(rn, cn)

    If (IsEmpty(z) Or z = 0) Then

      n = n

    Else

      n = n + 1

    End If

  Next cn

Next rn

If n > 0 Then

  Ans = MsgBox("  There are data in the space where" _

    & Chr(13) & "the output will be written. Proceed " _

    & Chr(13) & "  anyway and overwrite those data?", _

    vbYesNo)

  If Ans = vbNo Then

    Selection.Offset(0, -3).Select

    End

  End If

End If

' Calculate and write the frequency or time scale

If jSign = 1 Then

  For rn = 1 To rnMax / 2

    DataArray(rn, 1) = (rn - 1) / Interval

  Next rn

  For rn = (rnMax / 2 + 1) To rnMax

    If iSign > 0 Then

      DataArray(rn, 1) = (rn - rnMax - 1) / Interval

    Else

      DataArray(rn, 1) = (rn - 1) / Interval

    End If

  Next rn

End If

If jSign = -1 Then

  For rn = 1 To rnMax

    DataArray(rn, 1) = (-(rnMax / 2) + rn - 1) / Interval

  Next rn

End If

' Calculate the Fourier transform

Call FT(Term, NN, iSign)

' Arrange and write the output data

Denom = (rnMax + 1 + iSign * (rnMax - 1)) / 2

If jSign = 1 Then

  For rn = 1 To rnMax

    DataArray(rn, 2) = Term(2 * rn - 1) / Denom

    DataArray(rn, 3) = Term(2 * rn) / Denom

  Next rn

End If

If jSign = -1 Then

  For rn = 1 To rnMax / 2

    DataArray(rn, 2) = Term(rnMax + 2 * rn - 1) / Denom

    DataArray(rn, 3) = Term(rnMax + 2 * rn) / Denom

    DataArray((rnMax / 2) + rn, 2) = _

      Term(2 * rn - 1) / Denom

    DataArray((rnMax / 2) + rn, 3) = Term(2 * rn) / Denom

  Next rn

End If

Application.ScreenUpdating = False

Selection.Value = DataArray

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub FT(Term, NN, iSign)

' The following is the Fourier transform routine FOUR1 from

' J. C. Sprott, "Numerical Recipes: Routines and Examples

' in BASIC", Cambridge University Press, Copyright (C)1991

' by Numerical Recipes Software. Used here by permission.

' Use of this routine other than as an integral part of the

' present book requires an additional license from Numeri-

' cal Recipes Software. Further distribution is prohibited.

' The routine has been modified to yield double-precision

' results, and to conform to the standard mathematical sign

' convention for Fourier transformation.

Dim i As Integer, istep As Integer, j As Integer

Dim m As Integer, mmax As Integer

Dim tr As Double, ti As Double, theta As Double

Dim wtemp As Double, wi As Double, wr As Double

Dim wpi As Double, wpr As Double

j = 1

For i = 1 To NN Step 2

  If j > i Then

    tr = Term(j)

    ti = Term(j + 1)

    Term(j) = Term(i)

    Term(j + 1) = Term(i + 1)

    Term(i) = tr

    Term(i + 1) = ti

  End If

  m = Int(NN / 2)

  While m >= 2 And j > m

    j = j - m

    m = Int(m / 2)

  Wend

  j = j + m

Next i

mmax = 2

While NN > mmax

  istep = 2 * mmax

  theta = 2 * [Pi()] / (-iSign * mmax)

  wpr = -2 * Sin(0.5 * theta) ^ 2

  wpi = Sin(theta)

  wr = 1

  wi = 0

  For m = 1 To mmax Step 2

    For i = m To NN Step istep

      j = i + mmax

      tr = wr * Term(j) - wi * Term(j + 1)

      ti = wr * Term(j + 1) + wi * Term(j)

      Term(j) = Term(i) - tr

      Term(j + 1) = Term(i + 1) - ti

      Term(i) = Term(i) + tr

      Term(i + 1) = Term(i + 1) + ti

    Next i

    wtemp = wr

    wr = wr * wpr - wi * wpi + wr

    wi = wi * wpr + wtemp * wpi + wi

  Next m

  mmax = istep

Wend

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''^       DIRECT (DE)CONVOLUTION      ^'''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v4, May 1, 2004

' PURPOSE:

' These macros perform a straightforward (i.e., non-Fourier

' transform) convolution or deconvolution of 2 columns of

' real data, arranged in a single block of 3 adjacent

' columns, the first column containing the variable (time),

' the second column the signals to be convolved, and the

' third column the transfer function. The signal in the

' third column is normalized to have an average of 1.

' The number of points in the data set is arbitrary, but

' the data must be equidistant, since the signal and trans-

' fer functions will be sliding past each other.

' Convolution tends to spread the function. Make sure that

' there is enough space for it to spread into, e.g., by ad-

' ding zeros before and/or after the signal for expansion.

' Likewise, for a symmetrical transfer function, signal

' shifting can be prevented by centering the transfer

' function at t = 0.

' The deconvolution routine works well when the deconvol-

' ving function (in the third column) is a monotonically

' decreasing function of time, but otherwise may lead to

' oscillations.

' SUBROUTINES:

' The macros Convolve and Deconvolve set the value of d

' to +1 (for convolution) or -1 (for deconvolution), then

' call the subroutine Convolution. The latter requires no

' further subroutines.

' INPUT:

' The macro accepts any number of input data, but requires

' that they be equidistant. The output is independent of

' the starting value of the time scale, but does depend on

' the phase relationship between the two input signals.

' OUTPUT:

' The output is written in one column, immediately to the

' right of the input data block, thereby overwriting any

' prior data in that region. After writing the output, the

' activated area returns to its original position and

' contents.

' PROCEDURE:

' Before starting the macro, make sure that the column to

' the right of the input data block contains no valuable

' data.

' To initiate the transform, highlight the three columns of

' the input array, and call the macro.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Convolve()

Dim d As Integer

d = 1

Call Convolution(d)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub Deconvolve()

Dim d As Integer

d = -1

Call Convolution(d)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub Convolution(d)

Dim cMax As Integer, rM As Integer, rMax As Integer

Dim i As Integer, j As Integer, n As Integer

Dim Incr As Double, mySum As Double

Dim Sum As Double, z As Double

Dim Inp As Variant, myInput As Variant

Dim myInput2 As Variant, myOutput As Variant

Dim Outp As Variant, Trans As Variant

Dim Ans, hAnswer

' Read the input

Begin:

cMax = Selection.Columns.Count

rMax = Selection.Rows.Count

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "(De)Convolution")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check that there are three input columns

If cMax <> 3 Then

  MsgBox "There must be three input columns," _

    & Chr(13) & "one each for time, signal s, and" _

    & Chr(13) & "transfer function t, in that order."

  End

End If

' Read the input data

myInput = Selection.Value

myInput2 = Selection.Formula

Incr = myInput(2, 1) - myInput(1, 1)

rM = rMax - 1

ReDim Inp(-rM To rM)

ReDim Outp(-rM To rM)

ReDim Trans(-rM To rM)

' Check whether the fourth column can be overwritten

n = 0

Selection.Offset(0, 1).Select

myOutput = Selection.Value

For i = 1 To rMax

  z = myOutput(i, 3)

  If (IsEmpty(z) Or z = 0) Then

    n = n

  Else

    n = n + 1

  End If

Next i

If n > 0 Then

  Ans = MsgBox("There are data in the space where" _

  & Chr(13) & "the output will be written. Proceed" _

  & Chr(13) & "anyway and overwrite those data?", vbYesNo)

  If Ans = vbNo Then

    Selection.Offset(0, -1).Select

    End

  End If

End If

Selection.Offset(0, -1).Select

' Normalize and invert the transfer function

Sum = 0

For i = 1 To rMax

  Sum = Sum + myInput(i, 3)

Next i

For i = 1 To rMax

  Inp(1 - i) = 0

  Inp(i - 1) = myInput(i, 2)

  Trans(i - 1) = 0

  Trans(1 - i) = myInput(i, 3)

  Outp(1 - i) = 0

  Outp(i - 1) = 0

Next i

' Compute the convolution (for d = 1)

If d = 1 Then

  For j = 0 To rM

    mySum = 0

    For i = 0 To rM

      mySum = mySum + Inp(j - i) * Trans(-i)

    Next i

  Outp(j) = mySum / Sum

  myOutput(j + 1, 3) = Outp(j)

  Next j

End If

' Compute the deconvolution (for d = -1)

If d = -1 Then

  For j = 0 To rM

    mySum = Inp(j)

    For i = 1 To rM

      mySum = mySum - Outp(j - i) * Trans(-i)

    Next i

  Outp(j) = mySum

  Next j

  For j = 1 To rMax

    myOutput(j, 3) = Outp(j - 1) * Sum

  Next j

End If

' Output the result

Selection.Offset(0, 1).Select

Selection.Value = myOutput

Selection.Offset(0, -1).Select

Selection.Formula = myInput2

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''

'''''''^                                           ^'''''''

''''''^      FOURIER TRANSFORM (DE)CONVOLUTION      ^''''''

'''''''^                                           ^'''''''

''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

' These macros perform either a convolution or a deconvolu-

' tion of two sets of real data, arranged in a single block

' of three adjacent columns, the first column containing

' their common variable (e.g., time), and the next columns

' the data sets to be convolved or deconvolved.

' For convolution the order of the second and third column

' does not matter, and interchanging the data sets in the

' second and third column will yield the same result.

' For deconvolution, the second column should contain the

' data to be deconvolved, and the third column the data set

' with which the data in the second column are to be decon-

' volved. In this case, exchanging the second and third

' column will in general yield quite different results.

' For deconvolution the macro incorporates an adjustable

' von Hann window with a single, continually adjustable

' positive parameter, w. The default, w = 0, yields a

' rectangular window (i.e., the absence of filtering), but

' allows the user to select a sharp high-frequency cut-off.

' Setting w = 1 selects the traditional von Hann window,

' while other w-values >0 produce filtering that is less

' severe (0 < w < 1) or more severe (w > 1) than the stan-

' dard von Hann filter. Note that the filters have not been

' normalized.

' SUBROUTINES:

' The macros Convolve and Deconvolve require the subroutine

' Convolution which, in turn, calls the subroutine FT.

' INPUT:

' The macro accepts any input format in which the time

' scale contains 2^n equidistant values (where n is a posi-

' tive integer larger than 1), regardless of its starting

' value. While the output is independent of the starting

' value of the time scale, it does depend on the phase

' relationship between the two input signals.

' Provided that the function in the third column is symme-

' trical, such phase shifts can be avoided by starting that

' function at its point of symmetry (typically: at its

' maximum), and placing the part before that point of

' symmetry at the very end of the column. What makes this

' 'wrap-around' method work is the assumption of the

' Fourier transform method that the functions are self-

' repeating.

' OUTPUT:

' The output is written in one column, immediately to the

' right of the input data block, thereby overwriting any

' prior data in that region. After writing the output,

' the activated area returns to its original position and

' contents.

' PROCEDURE:

' Before starting the macro, make sure that the column to

' the right of the input data block contains no valuable

' data.

' To initiate the transform, highlight the three columns

' of the input array, and call the macro.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' The following are the two drivers for the (de)convolution

' subroutine. They set kSign to either +1 (for convolution)

' or -1 (for deconvolution).

Sub ConvolveFT()

Dim kSign As Integer

kSign = 1

Call ConvolutionFT(kSign)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub DeconvolveFT()

Dim kSign As Integer

kSign = -1

Call ConvolutionFT(kSign)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub ConvolutionFT(kSign)

Dim cn As Integer, cnMax As Integer

Dim n As Integer, NN As Integer, nfZ As Integer

Dim rn As Integer, rnMax As Integer, rrn As Integer

Dim nfZero As Single, q As Single

Dim Average3 As Double, Check1 As Double, Check2 As Double

Dim d As Double, dataSpacing As Double, Length As Double

Dim Min As Double, Norm As Double, Pi As Double

Dim Ratio As Double, Rnm2 As Double, s As Double

Dim Sum As Double, Sum3 As Double

Dim WindowFunction As Double, z As Double

Dim DataArray As Variant, inputArray As Variant

Dim outputArray As Variant

Dim Answer, hAnswer, vAnswer, Default1, Default2

Dim Message1, Message2, Title1, Title2

' Read the input

Begin:

n = 0

z = 0

rnMax = Selection.Rows.Count

NN = 2 * rnMax

cnMax = Selection.Columns.Count

Length = CDbl(rnMax)

' If area was not highlighted

If rMax = 1 And cMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Fourier transform (de)convolution")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

' Check the array length n, which must be a power of 2,

' and be at least 2

If Length < 2 Then

  MsgBox "There must be at least two rows."

  End

End If

Do While Length > 1

  Length = Length / 2

Loop

If Length <> 1 Then

  MsgBox "The number of rows must be a power of two."

  End

End If

' Check that there are three input columns

cnMax = Selection.Columns.Count

If cnMax <> 3 Then

  If kSign = 1 Then

    MsgBox " There must be 3 input columns," _

      & Chr(13) & "one for the independent variable" _

      & Chr(13) & " (e.g., time), the next for the" _

      & Chr(13) & " two functions to be convolved."

    End

  End If

  If kSign = -1 Then

    MsgBox " There must be 3 input columns," _

      & Chr(13) & "one for the independent variable" _

      & Chr(13) & " (e.g., time), the next for the" _

      & Chr(13) & " two functions to be deconvolved."

    End

  End If

End If

' Read the input data

DataArray = Selection.Value

inputArray = Selection.Formula

' Check that the first column has its first two elements

Check1 = VarType(DataArray(1, 1))

If Check1 = 0 Then

  MsgBox "Enter the top left value."

  End

End If

Check2 = VarType(DataArray(2, 1))

If Check2 = 0 Then

  MsgBox "Enter a value in row 2 of the first column."

  End

End If

dataSpacing = DataArray(2, 1) - DataArray(1, 1)

' Read and rearrange the input data

' from the second and third columns

ReDim Term2(NN) As Double, Term3(NN) As Double

ReDim Term4(NN) As Double

For rn = 1 To rnMax

  Term2(2 * rn - 1) = DataArray(rn, 2)

  Term2(2 * rn) = z

  Term3(2 * rn - 1) = DataArray(rn, 3)

  Term3(2 * rn) = z

Next rn

' Normalize the data in the third column

Sum = 0

For rn = 1 To rnMax

  Sum = Sum + Term3(2 * rn - 1)

Next rn

For rn = 1 To rnMax

' If normalization is NOT desired,

' comment out the next line:

  Term3(2 * rn - 1) = Term3(2 * rn - 1) * rnMax / Sum

Next rn

' Check that the output does not overwrite valuable data.

Selection.Offset(0, 1).Select

outputArray = Selection.Value

For rn = 1 To rnMax

  q = outputArray(rn, 3)

  If (IsEmpty(q) Or q = 0) Then

    n = n

  Else

    n = n + 1

  End If

Next rn

Selection.Offset(0, -1).Select

If n > 0 Then

  Answer = MsgBox(" There are data in the column where" _

    & Chr(13) & "the output will be written. Proceed" & _

    Chr(13) & " anyway and overwrite those data?", vbYesNo)

  If Answer = vbNo Then End

End If

' Transform the data from the second and third columns

Call FT(Term2, NN, 1)

Call FT(Term3, NN, 1)

' Multiply in the frequency domain for convolution, or

' divide in the frequency domain for deconvolution. For

' deconvolution, a minimum value "min" (arbitrarily set

' here to 1E-6) prevents division by zero. Moreover, an

' input box is provided for optional noise filtering.

Rnm2 = CDbl(rnMax) * CDbl(rnMax)

' For convolution, multiply in the frequency domain:

' (a + jb) (c + jd) = (ac - bd) + j(ad + bc)

If kSign = 1 Then

  For rn = 1 To rnMax

    Term4(2 * rn - 1) = _

      (Term2(2 * rn - 1) * Term3(2 * rn - 1) _

      - Term2(2 * rn) * Term3(2 * rn)) / rnMax

    Term4(2 * rn) = _

      (Term2(2 * rn - 1) * Term3(2 * rn) _

      + Term2(2 * rn) * Term3(2 * rn - 1)) / rnMax

  Next rn

End If

' For deconvolution, select the window parameter s

If kSign = -1 Then

 Min = 0.000001

 Pi = [Pi()]

 Message1 = "Enter the window parameter, a " & _

   "non-negative number." & Chr(13) & "The default " & _

   "is 0, which corresponds to no filtering," & Chr(13) & _

   "but allows you to insert a sharp frequency cut-off." _

   & Chr(13) & Chr(13) & "For the standard von Hann " _

   & "filter select 1." & Chr(13) & Chr(13) _

   & "For an adjustable von Hann / Tukey window," _

   & Chr(13) & " select a value other than 0 or 1."

  Title1 = "Adjustable von Hann / Tukey window"

  Default1 = "0"

  s = InputBox(Message1, Title1, Default1)

  If s < Min Then s = 0

' Provide the option of removing

' high-frequency data for M1 = 0

  If s = 0 Then

    Message2 = "Enter the number of frequencies" _

      & Chr(13) & "(an integer smaller than " _

      & rnMax & ") for which" & Chr(13) _

      & "you want the signal to be set to zero." & _

      Chr(13) & Chr(13) & "Press ENTER or OK for the " _

      & "default, 0," & Chr(13) & "in which case no " _

      & "data will be set to zero."

    Title2 = "Zeroing out high frequencies"

    Default2 = "0"

    nfZero = InputBox(Message2, Title2, Default2)

    If nfZero < 0 Then nfZ = 0

    If nfZero > 0 And nfZero < rnMax / 2 Then _

      nfZ = CInt(nfZero)

    If nfZero > rnMax / 2 Then nfZ = rnMax / 2

    If nfZ > 0 Then

      For rn = (rnMax / 2 + 2 - nfZ) To (rnMax / 2 + nfZ)

        Term4(2 * rn - 1) = 0

        Term4(2 * rn - 2) = 0

      Next rn

    End If

  End If

' For deconvolution, divide in the frequency domain:

' (a+jb)/(c+jd) = (a+jb)(c-jd)/(cc+dd)

'   = [(ac+bd)+j(bc-ad)]/(cc+dd)

  For rn = 1 To rnMax

    d = (Term3(2 * rn - 1) * Term3(2 * rn - 1) + _

      Term3(2 * rn) * Term3(2 * rn)) * dataSpacing / rnMax

    If d < Min Then d = Min

    Term4(2 * rn - 1) = _

      (Term2(2 * rn - 1) * Term3(2 * rn - 1) _

      + Term2(2 * rn) * Term3(2 * rn)) / d

    Term4(2 * rn) = (Term2(2 * rn) * Term3(2 * rn - 1) _

      - Term2(2 * rn - 1) * Term3(2 * rn)) / d

  ' Compute and apply the window function

    Ratio = 2 * CDbl(rn - 1) / CDbl(rnMax)

    If s < 1 Then WindowFunction = 1

    If s > 0 And s < 1 Then

      If Abs(Ratio - 1) < s Then WindowFunction = _

        0.5 + 0.5 * Cos(Pi * (Abs(Ratio) - 1 + s) / s)

    End If

    If s >= 1 Then

      WindowFunction = (0.5 + 0.5 * Cos(Pi * Ratio)) ^ s

    End If

    Term4(2 * rn - 1) = Term4(2 * rn - 1) * WindowFunction

    Term4(2 * rn) = Term4(2 * rn) * WindowFunction

  Next rn

End If

' Calculate output data

Call FT(Term4, NN, -1)

' Arrange and write the output data

For rn = 1 To rnMax

  outputArray(rn, 3) = Term4(2 * rn - 1) / rnMax

Next rn

Application.ScreenUpdating = False

Selection.Offset(0, 1).Select

Selection.Value = outputArray

Selection.Offset(0, -1).Select

Selection.Value = inputArray

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''^      ITERATIVE DECONVOLUTION      ^'''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

' This macro perform an iterative deconvolution along the

' lines of the van Cittert approach. It convolves (rather

' than deconvolves) the data in the second (signal) column

' with the filter function in the third column, then

' corrects the signal by the difference between it and its

' convolution. The macro provides the option of smoothing

' intermediate results with a moving quadratic of selec-

' table length, using the recursive expressions given by

' Gorry in Anal. Chem. 62 (1990) 570. The iteration stops

' when the sum of the squares of the residuals has reached

' a minimum, or the number of iterations has reached a pre-

' set limit.

' There are two choices: DeconvolveIt0 involves no

' constraints, while DeconvolveIt1 assumes one constraint,

' namely that the function is always positive. It therefore

' replaces all negative intermediate results by zero.

' INPUT:

' The input requires three adjacent columns of equal

' length. The first column should contain the independent

' parameter, such as time, wavelength, etc. The second

' column should contain the (real) input data that need to

' be deconvolved with the (real) filter data in the third

' column.

' The macro accepts any number of input data, but requires

' that they be equidistant. The output is independent of

' the starting value of the time scale, but does depend on

' the phase relationship between the two input signals.

' An optional parabolic noise filter is provided to reduce

' the effect of noise. Use of such a filter introduces its

' own distortion, the more so the longer the length of the

' moving parabola. If used at all, the parabola length

' should therefore be kept to a minimum.

' INTERRUPT:

' Because the iteration can take more time than you may

' have available or care to spend, the process can be

' interrupted, by pressing either the Escape key or the

' Control+Break key combination. A message box then

' provides the option of ending the computation and

' displaying the result obtained so far, or of resuming

' the calculation where it was interrupted.

' OUTPUT:

' The output is written in one column, immediately to the

' right of the input data block, thereby overwriting any

' prior data in that region. After writing the output,

' the activated area returns to its original position and

' contents. Moreover, the number of iterations used, and

' the value of the termination criterion are shown at the

' bottom of the data set.

' SUBROUTINES AND FUNCTIONS:

' The macros DeconvIt0 and DeconvolveIt1 require the

' subroutine DeconvolveIteratively which, in turn, calls

' the subroutine FT. Moreover, when filtering is used,

' the functions ConvolutionWeight, GramPolynomial, and

' GenFact are also required.

' PROCEDURE:

' Before starting the macro, make sure that the column to

' the right of the input data block contains no valuable

' data.

' To initiate the macro, highlight the three columns of the

' input array, and call the macro. The macro will then ask

' you to specify the maximum number of iterations; its

' default value is 10, and its maximum value has been set

' at 100,000. (If that takes you too long, interrupt the

' iterations with Esc.) The iterations will stop whenever

' the sum of the squares of the residuals has reached a

' minimum, or the set maximum number of iterations has been

' performed, whichever comes first.

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' For iterative deconvolution without constraints:

'

Sub DeconvolveIt0()

Dim zz As Integer

zz = 0

Call DeconvolveIteratively(zz)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

' For iterative deconvolution constrained at x = 0:

'

Sub DeconvolveIt1()

Dim zz As Integer

zz = 1

Call DeconvolveIteratively(zz)

End Sub

'

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'

Sub DeconvolveIteratively(zz)

Dim C As Integer, cn As Integer, cnMax As Integer

Dim ItNo As Integer, n As Integer

Dim NN As Integer, nfZ As Integer

Dim rn As Integer, rnMax As Integer, rrn As Integer

Dim nfZero As Single, q As Single

Dim Average3 As Double, Check1 As Double, Check2 As Double

Dim d As Double, dataSpacing As Double, Delta As Double

Dim Factor As Double, Length As Double, MaxItNo As Double

Dim Min As Double, myScale As Double, Norm As Double

Dim p As Double, Pi As Double, previousSSR As Double

Dim Ratio As Double, Rnm2 As Double, SSR As Double

Dim s As Double, Sum As Double

Dim Sum3 As Double, WindowFunction As Double, z As Double

Dim DataArray As Variant, inputArray As Variant

Dim outputArray As Variant

Dim aa, bb, cc, Answer, hAnswer, vAnswer, ContinueHere

Dim Default, InterruptHandler, Message, myFilter

Dim myFunction, Reply, Title, TryAgain

' Read the input

Begin:

n = 0

z = 0

rnMax = Selection.Rows.Count

NN = 2 * rnMax

cnMax = Selection.Columns.Count

Length = CDbl(rnMax)

Pi = [Pi()]

' If area was not highlighted

If rnMax = 1 And cnMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Iterative (de)convolution")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

On Error GoTo InterruptHandler

' Check the array length n, which must

' be a power of 2, and be at least 2

If Length < 2 Then

  MsgBox "There must be at least two rows."

  End

End If

Do While Length > 1

  Length = Length / 2

Loop

If Length <> 1 Then

  MsgBox "The number of rows must be a power of two."

  End

End If

' Check that there are three input columns

If cnMax <> 3 Then

  MsgBox "There must be three input columns, one for " _

    & "the" & Chr(13) & "   independent variable (e.g., " _

    & "time), the next" & Chr(13) _

    & "   two for the two functions to be deconvolved."

  End

End If

' Read the input data

DataArray = Selection.Value

inputArray = Selection.Formula

' Check that the first column has its first two elements

Check1 = VarType(DataArray(1, 1))

If Check1 = 0 Then

  MsgBox "Enter the top left value."

  End

End If

Check2 = VarType(DataArray(2, 1))

If Check2 = 0 Then

  MsgBox "Enter a value in row 2 of the first column."

  End

End If

dataSpacing = DataArray(2, 1) - DataArray(1, 1)

' Check that the output does not overwrite valuable data.

Selection.Offset(0, 1).Select

outputArray = Selection.Value

For rn = 1 To rnMax

  q = outputArray(rn, 3)

  If (IsEmpty(q) Or q = 0) Then

    n = n

  Else

    n = n + 1

  End If

Next rn

Selection.Offset(0, -1).Select

If n > 0 Then

  Answer = MsgBox("There are data in the column where" & _

    Chr(13) & "the output will be written. Proceed" & _

    Chr(13) & " anyway and overwrite those data?", vbYesNo)

  If Answer = vbNo Then End

End If

' Select a filter

TryAgain:

Message = "You can select a moving parabola to filter" & _

Chr(13) & "the data. Specify a moving parabola with an" & _

Chr(13) & "odd integer number of data points, at least" & _

Chr(13) & "5 but always much smaller than the number" & _

Chr(13) & "of points in your data set, or select the" & _

Chr(13) & "default, 0, if you do NOT want any filtering."

Title = "Deconvolve Iteratively: Select moving polynomial length"

Default = "0"

myFilter = InputBox(Message, Title, Default)

If myFilter = 0 Then GoTo aa

m = CInt((myFilter - 1) / 2)

If (m <> 0 And m < 2) Then

  MsgBox "Select either 0 (for no filtering)," & _

  Chr(13) & "or an odd number larger than 3 for" & _

  Chr(13) & "the length of the moving parabola."

  GoTo TryAgain

End If

If rnMax < 2 * m + 3 Then

  MsgBox "Select a shorter moving parabola length."

  GoTo TryAgain

End If

aa:

' Save the input signal

ReDim signal(1 To rnMax) As Double

ReDim newSignal(1 To rnMax) As Double

For rn = 1 To rnMax

  signal(rn) = DataArray(rn, 2)

Next rn

' Read and rearrange the input data from the third column

ReDim Term3(1 To NN) As Double

For rn = 1 To rnMax

  Term3(2 * rn - 1) = DataArray(rn, 3)

  Term3(2 * rn) = z

Next rn

' Normalize the data in the third column

Sum = 0

For rn = 1 To rnMax

  Sum = Sum + DataArray(rn, 3)

Next rn

myScale = rnMax / Sum

' Select the maximum number of iterations; the

' default value is 10, and its maximum value is

' limited at 10000 to prevent an endless loop.

Message = "Select the maximum number of iterations"

Title = _

  "Deconvolve Iteratively: Select max. # of iterations"

Default = "10"

MaxItNo = InputBox(Message, Title, Default)

If MaxItNo > 10000 Then MaxItNo = 10000

' Make code uninterruptible for next section

Application.EnableCancelKey = xlDisabled

' Set the iteration number and start the iteration loop

ItNo = 1

bb:

previousSSR = SSR

' Read and rearrange the input data from the second column

ReDim Term2(NN) As Double, Term4(NN) As Double

For rn = 1 To rnMax

  Term2(2 * rn - 1) = DataArray(rn, 2)

  Term2(2 * rn) = z

  Term3(2 * rn - 1) = DataArray(rn, 3) * myScale

  Term3(2 * rn) = z

Next rn

' Transform the data from the second and third columns

Call FT(Term2, NN, 1)

Call FT(Term3, NN, 1)

' Multiply in the frequency domain for convolution

Rnm2 = CDbl(rnMax) * CDbl(rnMax)

' Multiply in the frequency domain, using

' (a + jb) * (c + jd) = (ac - bd) + j(ad + bc)

' where j is the square root of -1

For rn = 1 To rnMax

  Term4(2 * rn - 1) = _

    (Term2(2 * rn - 1) * Term3(2 * rn - 1) _

    - Term2(2 * rn) * Term3(2 * rn)) / rnMax

  Term4(2 * rn) = (Term2(2 * rn - 1) * Term3(2 * rn) _

    + Term2(2 * rn) * Term3(2 * rn - 1)) / rnMax

Next rn

' Inverse transform to obtain the convolution

Call FT(Term4, NN, -1)

For rn = 1 To rnMax

  outputArray(rn, 3) = Term4(2 * rn - 1) / rnMax

Next rn

' Compute the corrected signal value

If zz < 2 Then

  For rn = 1 To rnMax

    p = 1

    If zz = 1 And signal(rn) < 0 Then p = 0

    newSignal(rn) = p * signal(rn) + _

      DataArray(rn, 2) - outputArray(rn, 3)

  Next rn

ElseIf zz = 2 Then

  For rn = 1 To rnMax

    newSignal(rn) = signal(rn) + (1 - 2 * _

      Abs(signal(rn) - 0.5)) * (DataArray(rn, 2) - _

      outputArray(rn, 3))

  Next rn

End If

' Compute SSR

SSR = 0

For rn = 1 To rnMax

  SSR = SSR + (outputArray(rn, 3) - signal(rn)) ^ 2

Next rn

If myFilter = 0 Then GoTo cc

' Apply the filter

ReDim SmoothedArray(rnMax) As Double

For rn = 1 To rnMax

  SmoothedArray(rn) = DataArray(rn, 2)

Next rn

k = 2           ' for a parabola

s = 0           ' for smoothing

For rn = 1 To m

  SmoothedArray(rn) = 0

  For i = -m To m

    SmoothedArray(rn) = SmoothedArray(rn) _

    + ConvolutionWeight(i, rn - m - 1, m, k, s) _

    * DataArray(i + m + 1, 2)

  Next i

Next rn

For rn = m + 1 To rnMax - m

  SmoothedArray(rn) = 0

  For i = -m To m

    SmoothedArray(rn) = SmoothedArray(rn) _

    + ConvolutionWeight(i, 0, m, k, s) _

    * DataArray(rn + i, 2)

  Next i

Next rn

For rn = rnMax - m + 1 To rnMax

  SmoothedArray(rn) = 0

  For i = -m To m

    SmoothedArray(rn) = SmoothedArray(rn) _

    + ConvolutionWeight(i, rn - rnMax + m, m, k, s) _

    * DataArray(rnMax - m + i, 2)

  Next i

Next rn

For rn = 1 To rnMax

  DataArray(rn, 2) = SmoothedArray(rn)

Next rn

cc:

' Set negative values equal to zero for DeconvolveIt1

If zz = 1 Then

  For rn = 1 To rnMax

    If DataArray(rn, 2) < 0 Then DataArray(rn, 2) _

      = -DataArray(rn, 2)

  Next rn

End If

ContinueHere:

' Make code briefly interruptible

Application.EnableCancelKey = xlErrorHandler

' Update the second (signal) column of dataArray

For rn = 1 To rnMax

  DataArray(rn, 2) = signal(rn) _

    + DataArray(rn, 2) - outputArray(rn, 3)

Next rn

' increment ItNo and loop back

If ItNo = 1 Then previousSSR = SSR

ItNo = ItNo + 1

Application.StatusBar = "Iteration # " & _

  ItNo & ", SSR = " & SSR

If (ItNo < MaxItNo And previousSSR >= SSR) Then GoTo bb

Output:

' Prepare for landing

For rn = 1 To rnMax

  outputArray(rn, 3) = signal(rn) _

    + DataArray(rn, 2) - outputArray(rn, 3)

Next rn

If zz > 0 Then

  For rn = 1 To rnMax

    If outputArray(rn, 3) < 0 Then _

      outputArray(rn, 3) = 0

  Next rn

End If

' Summarize the parameters used

' If you do not want a reminder of the parameters chosen,

' and a display of the final score on the status bar, place

' the apostrophe before the word MsgBox

MsgBox "The program DeconvolveIt" & zz & "," & _

  Chr(13) & "using a " & myFilter & _

  "-point parabolic filter," & Chr(13) & _

  "found an answer after " & ItNo & " iterations" & _

  Chr(13) & "with an SSR value of " & SSR & "."

Application.StatusBar = False

Application.ScreenUpdating = False

Selection.Offset(0, 1).Select

Selection.Value = outputArray

Selection.Offset(0, -1).Select

Selection.Value = inputArray

End

InterruptHandler:

If Err.number = 18 Then

  Reply = MsgBox("Do you want to terminate the" & _

    vbCr & "computation? If so, press Yes", vbYesNo, _

    "DeconvolveIt:   Stop iteration?")

  If Reply = vbYes Then GoTo Output

  If Reply = vbNo Then

    Resume ContinueHere

  End If

End If

End Sub

Function ConvolutionWeight(i, t, m, k, s) As Double

Dim h As Integer

Dim Sum As Double

Sum = 0

For h = 0 To k

  Sum = Sum + (2 * h + 1) * (GenFact(2 * m, h) _

  / GenFact(2 * m + h + 1, h + 1)) _

  * GramPolynomial(i, m, h, 0) _

  * GramPolynomial(t, m, h, s)

Next h

ConvolutionWeight = Sum

End Function

Function GramPolynomial(i, m, k, s) As Double

If k > 0 Then

  GramPolynomial = ((4 * k - 2) / (k * (2 * m - k + 1))) _

  * (i * GramPolynomial(i, m, k - 1, s) _

  + s * GramPolynomial(i, m, k - 1, s - 1)) _

  - ((k - 1) / k) * ((2 * m + k) / (2 * m - k + 1)) _

  * GramPolynomial(i, m, k - 2, s)

ElseIf k = 0 And s = 0 Then

  GramPolynomial = 1

Else

  GramPolynomial = 0

End If

End Function

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^      GABOR TRANSFORMATION     ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

' Gabor transformation analyzes a time-dependent signal in

' terms of its frequency content as a function of time. It

' does this by multiplying the time-dependent signal by a

' moving Gaussian window, and by then Fourier-transforming

' the resulting product. The result is stored in two

' arrays, ReOut() and ImOut(), which contain the real and

' imaginary parts of the Fourier transform as a function of

' time and frequency. The contents of one of these arrays,

' or of the corresponding magnitude or phase angle, can

' then be displayed in a color map (using Mapper) or a 3-D

' plot.

' SUBROUTINES:

' This macro requires the subroutine FT.

' INPUT:

' In input information should be placed in two adjacent

' columns, with the left-most column containing time, the

' right-most the corresponding signal. The macro is re-

' stricted to equidistant data, and the time increments

' must therefore be equal. The width of the moving window

' must be an odd power of 2 (here set to be at least a

' factor 2 smaller than the number of data in the set),

' and is selected with an input box.

' OUTPUT:

' The result is displayed with a time axis on top, and a

' frequency axis to the left, ready for immediate visuali-

' zation with Mapper.

' PROCEDURE:

' In order to start the Gabor transform, call the macro

' and enter the requested information in the input boxes.

' WARNING:

' This macro produces a large output array to the right of

' the input data. It does NOT check whether that array

' contains any information that should not be overwritten.

' Therefore make sure that there is no valuable information

' on the spreadsheet to the right of the input columns.

' dataArray   input data set, two columns wide, the left-

'               most for time, the rightmost for signal

' M           the number of data in the set, i.e., the

'               number of rows in dataArray

' N           the number of data used for the window,

'               smaller than about half of M

' Gaussian    the window function used, containing N points

' ReOut       the real component of the output signal,

'               as a function of time and frequency

' ImOut       the imaginary component of the output signal,

'               as a function of time and frequency

' MagOut      the magnitude (absolute value) of the output

'               signal, as a function of time and frequency

' LogOut      the 10-based logarithm of the magnitude of

'               the output signal, as a function of time

'               and frequency

' PhaseOut    the phase angle (in radians) of the output

'               signal, as a function of time and frequency

Sub Gabor()

Dim i As Integer, j As Integer, jj As Integer

Dim m As Integer, MM As Integer, n As Integer

Dim NN As Integer, MinSpacing As Integer

Dim p As Integer, Spacing As Integer

Dim Gaussian() As Double, ImOut() As Double

Dim Interval As Double, LogOut() As Double

Dim MagOut() As Double, Norm As Double

Dim PhaseOut() As Double, Pi As Double, ReOut() As Double

Dim s As Double, Term() As Double

Dim DataArray As Variant, freqArray As Variant

Dim outputArray As Variant

Dim myRange As Range

Dim Message1, Message3, Title1, Title3, Default3

' Read the time interval

Message1 = "The time interval between successive data is:"

Title1 = "Gabor InputBox 1: Time interval"

Interval = InputBox(Message1, Title1)

' Read the input data

Set myRange = Application.InputBox(Prompt:= _

  "The input data are located in column:", Title:= _

  "Gabor InputBox 2: Input data", Type:=8)

myRange.Select

' Check that there is only one data input column

If Selection.Columns.Count <> 1 Then

  MsgBox "There can only be ONE data input column."

  End

End If

DataArray = Selection.Value

freqArray = Selection.Value

outputArray = Selection.Value

m = Selection.Rows.Count

' Determine the size of the window width N

MM = m / 4

n = 2

Do

n = 2 * n

Loop Until n > MM

' Set the minimum sample spacing

MinSpacing = Int((m - n + 1) / 125)

jj = MinSpacing

' Select a sample spacing

Message3 = "The minimum sample spacing is:"

Title3 = "Gabor InputBox 3: Sample spacing"

Default3 = MinSpacing

Spacing = InputBox(Message3, Title3, Default3)

If Spacing >= MinSpacing Then jj = Spacing

' Compute the frequency scale

For i = 1 To n

  freqArray(i, 1) = " "

Next i

For i = 1 + n / 2 To n

  freqArray(i, 1) = (-n + 2 * i - 2) / (2 * n * Interval)

Next i

For i = n + 1 To n + 1

  freqArray(i, 1) = (-n + 2 * i - 2) / (2 * n * Interval)

Next i

' Display the frequency scale

Selection.Offset(0, 1).Select

Selection.Value = freqArray

Application.ScreenUpdating = False

' Compute the window function Gaussian()

ReDim Gaussian(1 To n)

Pi = [Pi()]

s = n / 6

Norm = 12 / Sqr(2 * Pi)

For i = 1 To n

  Gaussian(i) = _

    Norm * Exp(-(i - (n + 1) / 2) ^ 2 / (2 * s * s))

Next i

' The main loop, repeating the analysis M-N+1 times

For j = jj To m - n + 1 Step jj

' Window the input data fragment and rearrange it for FFT

  NN = 2 * n

  ReDim Term(NN) As Double

  For i = 1 To n

    Term(2 * i - 1) = DataArray(i + j - 1, 1) * Gaussian(i)

    Term(2 * i) = 0

  Next i

'  Call the forward Fourier transform subroutine

  Call FT(Term, NN, 1)

' Compute the output data. The following section presents

' several output options, of which only one combination can

' be selected at any one time. For general visualization of

' the result, MagOut is usually the preferred output format

  ReDim MagOut(n, m - n + 1)

'  ReDim ReOut(N, M - N + 1)

'  ReDim ImOut(N, M - N + 1)

'  ReDim LogMagOut(N, M - N + 1)

'  ReDim PhaseOut(N, M - N + 1)

  For i = 1 To n

    MagOut(i, j) = _

      (Sqr(Term(2 * i - 1) ^ 2 + Term(2 * i) ^ 2)) / n

'    ReOut(i, j) = Term(2 * i - 1) / N

'    ImOut(i, j) = Term(2 * i) / N

'    LogMagOut(i, j) = 0.5 * Log((Term(2 * i - 1) ^ 2 _

       + Term(2 * i) ^ 2) / N)

'    PhaseOut(i, j) = Atn(Term(2 * i) / Term(2 * i - 1))

  Next i

  p = Int(100 * (j / (m - n + 1)))

  Application.StatusBar = "Calculation " & p & "% done"

' Data output

' (shown here for MagOut; modify by replacing MagOut

' by alternative output file name XXX(i, j) if so desired)

  For i = 1 To n / 2 - 1

    outputArray(i, 1) = " "

  Next i

  For i = n / 2 To n / 2

    outputArray(i, 1) = j

  Next i

  For i = 1 + n / 2 To n

    outputArray(i, 1) = MagOut(i, j)

  Next i

  For i = n + 1 To n + 2

    outputArray(i, 1) = MagOut(1, j)

  Next i

  For i = n + 2 To m

    outputArray(i, 1) = " "

  Next i

  Selection.Offset(0, 1).Select

  Selection.Value = outputArray

Next j

Application.StatusBar = False

End Sub

''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^     SEMI-INTEGRATION AND      ^'''''''''''''

'''''''''''''^     SEMI-DIFFERENTIATION      ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

' This macro performs semi-integration and semi-differen-

' tiation, specialized (de)convolution methods applicable

' to problems of planar heat conduction, planar diffusion,

' and signal transmission along a cable with uniformly

' distributed resistance and capacitance, see K. B. Oldham,

' Anal. Chem. 41 (1969) 1121; 44 (1972) 196.

' The macro uses a simple semi-integration algorithm; for

' more sophisticated algorithms see K. B. Oldham & J. C.

' Myland, Fundamentals of electrochemical science, Academic

' Press 1994. The semi-differential is then found by taking

' first differences.

' In the context of linear sweep voltammetry or cyclic

' voltammetry (which are both methods used in electro-

' chemical analysis), semi-integration can convert a

' current-time-voltage curve into the corresponding

' steady-state current-voltage curve, which is often

' much easier to interpret quantitatively. For a discussion

' of such an application, see any modern electrochemical

' textbook, or consult my How to use Excel in analytical

' chemistry, Cambridge Univ. Press, 2001, pp. 257-263.

' SUBROUTINES:

' The drivers SemiIntegrate() and SemiDifferentiate()

' call the subroutine Semi

' INPUT:

' The input data must be equidistant, and should be

' stored in a single, continuous column.

' OUTPUT:

' The resulting transformed output data will be written

' in the column immediately to the right of the input

' data. That column is therefore best left free.

' PROCEDURE:

' Call the macro, and input the data by following the

' instructions with its input boxes.

Sub SemiIntegrate()

Dim iIntDif As Integer

iIntDif = 1

Call Semi(iIntDif)

End Sub

Sub SemiDifferentiate()

Dim iIntDif As Integer

iIntDif = -1

Call Semi(iIntDif)

End Sub

Sub Semi(iIntDif)

' The macro uses a simple semi-integration algorithm due

' to Oldham which is adequate to illustrate the approach.

' The semi-differential is then found by taking first

' differences.

' The input data should be stored in a single, continuous

' column. The resulting transformed output data will be

' written in the column immediately to the right of the

' input data. That column is therefore best left free.

' Parameter names used:

'     cnmax         number of columns in

'                   highlighted data block

'     rnmax         number of rows in

'                   highlighted data block

'     dataArray     array of all input data

'     rn            index specifying row number

'                   in input data array

Dim cnMax As Integer, iIntDif As Integer

Dim j As Integer, k As Integer

Dim n As Integer, rnMax As Integer

Dim A As Double, s As Double, deltaTime As Double

Dim myRange As Range

Dim inputData As Variant, semiInt As Variant

Dim outputData As Variant, testValue As Variant

Dim z As Variant

Dim myAddress, Ans

n = 0

' Input the time increment and use it

' to compute the parameter s

If iIntDif = 1 Then

  deltaTime = _

    InputBox("Enter the time increments, in seconds.", _

    "SemiIntegration InputBox 1: Time increments")

Else

  deltaTime = _

    InputBox("Enter the time increments, in seconds.", _

    "SemiDifferentiation InputBox 1: Time increments")

End If

s = Sqr(Abs(deltaTime))

' Enter the input data

If iIntDif = 1 Then

  Set myRange = Application.InputBox(Prompt:= _

  "The current values are located in:", Title:= _

  "SemiIntegration InputBox 2: Input data", Type:=8)

Else

  Set myRange = Application.InputBox(Prompt:= _

  "The current values are located in:", Title:= _

  "SemiDifferentiation InputBox 2: Input data", Type:=8)

End If

myRange.Select

If Selection.Columns.Count <> 1 Then

  End

Else

  rnMax = Selection.Rows.Count

  inputData = Selection.Value

  semiInt = Selection.Value

  outputData = Selection.Value

  myAddress = Selection.Address

End If

' Check that the output will not overwrite valuable data

Selection.Offset(0, 1).Select

testValue = Selection.Value

For j = 1 To rnMax

  z = testValue(j, 1)

  If IsEmpty(z) Then

    n = n

  Else

    n = n + 1

  End If

Next j

If n > 0 Then

  Ans = MsgBox("There are data in the space where " _

    & "the output will be" & Chr(13) & " written. " _

    & "Proceed anyway and overwrite those data?", _

  vbYesNo)

  If Ans = vbNo Then

    Selection.Offset(0, -1).Select

    End

  End If

End If

' Compute the semi-integration

ReDim p(1 To rnMax) As Double

p(1) = 1

For j = 2 To rnMax

  p(j) = p(j - 1) * (j - 1.5) / (j - 1)

Next j

For k = 1 To rnMax

  A = 0

  For j = 1 To k

    A = A + p(k - j + 1) * inputData(j, 1)

  Next j

  semiInt(k, 1) = A * s

Next k

If iIntDif = -1 Then

  For k = 2 To rnMax

    outputData(k, 1) = _

      (semiInt(k, 1) - semiInt(k - 1, 1)) / deltaTime

  Next k

  semiInt(1, 1) = ""

Else

  For k = 1 To rnMax

    outputData(k, 1) = semiInt(k, 1)

  Next k

End If

' Write the result

Selection.Value = outputData

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''^     MOVIE DEMONSTRATION MACROS    ^'''''''''''

''''''''''''^                                 ^''''''''''''

'''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' The following macros merely demonstrate how to make

' simple Excel 'movies'

'

''''''''''''''''''''''' MOVIEDEMO 1 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the

' number 0 in cells A1 and B1, and the number 10 in A2 and

' B2. Then highlight the area A1:B2, and make a correspon-

' ding graph. Format both axes to have manual scales from 0

' to 10. Make sure that the graph displays B1:B2 vs. A1:A2

' as individual points ('markers'). Then call the macro,

' and enjoy the show.

Sub MovieDemo1()

Range("A1") = 0

Range("A2") = 0

For i = 1 To 400

  Range("A1") = 10 - 0.05 * Abs(i - 200)

  Range("A2") = 10 * Exp(-0.001 * (i - 300) ^ 2)

  Application.ScreenUpdating = True

Next i

Range("A1") = 0

Range("A2") = 0

End Sub

''''''''''''''''''''''' MOVIEDEMO 2 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the num-

' ber 0 in cells H1 and I1, and the number 10 in H2 and I2.

' Then highlight the area H1:I2, and make a corresponding

' graph. Format both axes to have manual scales from 0 to

' 10. Make sure that the graph displays I1:I2 vs. H1:H2 as

' individual points ('markers') connected by a line. Then

' call the macro.

Sub MovieDemo2()

Range("H1") = 5: Range("I1") = 1

Range("H2") = 5: Range("I2") = 5

'Application.ScreenUpdating = True

For i = 1 To 1000

  Range("H1") = 5 - 5 * Sin(0.06 * i)

  Range("I1") = 5 - 5 * Cos(0.07 * i)

  Application.ScreenUpdating = True

Next i

Range("H1") = 5: Range("I1") = 1

Range("H2") = 5: Range("I2") = 5

End Sub

''''''''''''''''''''''' MOVIEDEMO 3 '''''''''''''''''''''''

' This demonstration macro requires that you enter the

' numbers 0 through 10 in cells A15 through A25, and the

' number 5 in cells B15:B25. Again use a graph with manual

' scales from 0 to 10, and display the data in B15:B25 vs.

' A15:A25 as markers plus (smoothed) connecting lines.

Sub MovieDemo3()

Range("A15") = 0: Range("B15") = 5

Range("A16") = 1: Range("B16") = 5

Range("A17") = 2: Range("B17") = 5

Range("A18") = 3: Range("B18") = 5

Range("A19") = 4: Range("B19") = 5

Range("A20") = 5: Range("B20") = 5

Range("A21") = 6: Range("B21") = 5

Range("A22") = 7: Range("B22") = 5

Range("A23") = 8: Range("B23") = 5

Range("A24") = 9: Range("B24") = 5

Range("A25") = 10: Range("B25") = 5

Application.ScreenUpdating = True

For i = 1 To 100

  Application.ScreenUpdating = False

  Range("B15") = 5 + 0.04 * i * Sin(0.01 * i)

  Range("B16") = 5 + 0.04 * i * Sin(0.01 * i - 100)

  Range("B17") = 5 + 0.04 * i * Sin(0.01 * i - 200)

  Range("B18") = 5 + 0.04 * i * Sin(0.01 * i - 300)

  Range("B19") = 5 + 0.04 * i * Sin(0.01 * i - 400)

  Range("B20") = 5 + 0.04 * i * Sin(0.01 * i - 500)

  Range("B21") = 5 + 0.04 * i * Sin(0.01 * i - 600)

  Range("B22") = 5 + 0.04 * i * Sin(0.01 * i - 700)

  Range("B23") = 5 + 0.04 * i * Sin(0.01 * i - 800)

  Range("B24") = 5 + 0.04 * i * Sin(0.01 * i - 900)

  Range("B25") = 5 + 0.04 * i * Sin(0.01 * i - 1000)

  Application.ScreenUpdating = True

Next i

For i = 101 To 900

  Application.ScreenUpdating = False

  Range("B15") = 5 + 4 * Sin(0.01 * i)

  Range("B16") = 5 + 4 * Sin(0.01 * i - 100)

  Range("B17") = 5 + 4 * Sin(0.01 * i - 200)

  Range("B18") = 5 + 4 * Sin(0.01 * i - 300)

  Range("B19") = 5 + 4 * Sin(0.01 * i - 400)

  Range("B20") = 5 + 4 * Sin(0.01 * i - 500)

  Range("B21") = 5 + 4 * Sin(0.01 * i - 600)

  Range("B22") = 5 + 4 * Sin(0.01 * i - 700)

  Range("B23") = 5 + 4 * Sin(0.01 * i - 800)

  Range("B24") = 5 + 4 * Sin(0.01 * i - 900)

  Range("B25") = 5 + 4 * Sin(0.01 * i - 1000)

  Application.ScreenUpdating = True

Next i

For i = 901 To 1000

  Application.ScreenUpdating = False

  Range("B15") = 5 + 0.04 * (1000 - i) * Sin(0.01 * i)

  Range("B16") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 100)

  Range("B17") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 200)

  Range("B18") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 300)

  Range("B19") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 400)

  Range("B20") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 500)

  Range("B21") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 600)

  Range("B22") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 700)

  Range("B23") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 800)

  Range("B24") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 900)

  Range("B25") = _

    5 + 0.04 * (1000 - i) * Sin(0.01 * i - 1000)

  Application.ScreenUpdating = True

Next i

Range("A15") = 0: Range("B15") = 5

Range("A16") = 1: Range("B16") = 5

Range("A17") = 2: Range("B17") = 5

Range("A18") = 3: Range("B18") = 5

Range("A19") = 4: Range("B19") = 5

Range("A20") = 5: Range("B20") = 5

Range("A21") = 6: Range("B21") = 5

Range("A22") = 7: Range("B22") = 5

Range("A23") = 8: Range("B23") = 5

Range("A24") = 9: Range("B24") = 5

Range("A25") = 10: Range("B25") = 5

End Sub

''''''''''''''''''''''' MOVIEDEMO 4 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the

' numbers 0, 0, 5, 5, 5, 5, 10 in cells A29:A36, and the

' number 0,1,7,5,5,5,5,0 in cells B29:B36. Again use a

' graph with manual scales from 0 to 10, and display the

' data as individually distinct points, each of a different

' type and/or color. For example, point A32,B32 might be a

' hollow black circle of size 18, point A33,B33 a hollow

' red circle of size 14, point A34,B34 a hollow black cir-

' cle of size 9, and point A35,B35 be a solid red circle of

' size 6. Also make the other data markers distinct.

Sub MovieDemo4()

Range("A1") = 0: Range("B1") = 0

Range("A2") = 0: Range("B2") = 1

Range("A3") = 5: Range("B3") = 9

Range("A4") = 5: Range("B4") = 5

Range("A5") = 5: Range("B5") = 5

Range("A6") = 5: Range("B6") = 5

Range("A7") = 5: Range("B7") = 5

Range("A8") = 10: Range("B8") = 0

Application.ScreenUpdating = True

Application.Volatile True

For i = 1 To 1000

  Application.ScreenUpdating = False

  Range("A1") = 5 - 5 * Sin(0.01 * (i - 500))

  Range("B1") = 5 - 5 * Sin(0.01 * (i - 500))

  Range("A2") = Range("A2") + 0.01

  Range("B2") = 5 - 4 * Cos(0.03 * i)

  Range("A3") = 5 + 4 * Sin(0.2 * i)

  Range("B3") = 5 - 4 * Cos(0.2 * i)

  If CDbl(i) / 160 - CInt(i / 160) = 0 Then

    Range("A4") = 5 + (5 * Rnd() - 2.5)

    Range("B4") = 5 + (5 * Rnd() - 2.5)

  End If

  If CDbl(i) / 170 - CInt(i / 170) = 0 Then

    Range("A5") = Range("A4")

    Range("B5") = Range("B4")

  End If

  If CDbl(i) / 180 - CInt(i / 180) = 0 Then

    Range("A6") = Range("A4")

    Range("B6") = Range("B4")

  End If

  If CDbl(i) / 190 - CInt(i / 190) = 0 Then

    Range("A7") = Range("A4")

    Range("B7") = Range("B4")

  End If

  If CDbl(i) / 10 - CInt(i / 10) = 0 Then

    Range("A8") = 5 + 5 * Sin(0.01 * (i - 500))

  End If

  Range("B8") = Range("B8") + 0.01

  Application.ScreenUpdating = True

Next i

Range("A1") = 0: Range("B1") = 0

Range("A2") = 0: Range("B2") = 1

Range("A3") = 5: Range("B3") = 9

Range("A4") = 5: Range("B4") = 5

Range("A5") = 5: Range("B5") = 5

Range("A6") = 5: Range("B6") = 5

Range("A7") = 5: Range("B7") = 5

Range("A8") = 10: Range("B8") = 0

End Sub

''''''''''''''''''''''' MOVIEDEMO 5 '''''''''''''''''''''''

'

' This demonstration macro requires that you enter the

' numbers 4, 4, 4, 4, 6, 6, 6, and 6 in cells H15:H22,

' and the number 5 in cells I15:I22. Again use a graph

' with manual scales from 0 to 10, and display the points

' H15, I15, and H19:I19 as small solid circles, preferably

' of different color, and the others as larger, hollow

' (concentric) rings around them.

Sub MovieDemo5()

Range("H15") = 4: Range("I15") = 5

Range("H16") = 4: Range("I16") = 5

Range("H17") = 4: Range("I17") = 5

Range("H18") = 4: Range("I18") = 5

Range("H19") = 6: Range("I19") = 5

Range("H20") = 6: Range("I20") = 5

Range("H21") = 6: Range("I21") = 5

Range("H22") = 6: Range("I22") = 5

For i = 1 To 252

  Range("H15") = 4 + 0.2 * Sin(0.1 * i)

  Range("H16") = 4 + 0.3 * Sin(0.1 * i)

  Range("H17") = 4 + 0.4 * Sin(0.1 * i)

  Range("H18") = 4 + 0.5 * Sin(0.1 * i)

  Range("H19") = 6 - 0.2 * Sin(0.1 * i)

  Range("H20") = 6 - 0.3 * Sin(0.1 * i)

  Range("H21") = 6 - 0.4 * Sin(0.1 * i)

  Range("H22") = 6 - 0.5 * Sin(0.1 * i)

  Application.ScreenUpdating = True

Next i

For i = 253 To 502

  Range("H15") = _

    5 + Cos(0.025 * i) * (-1 + 0.2 * Sin(0.1 * i))

  Range("I15") = _

    5 + Sin(0.025 * i) * (-1 + 0.2 * Sin(0.1 * i))

  Range("H16") = _

    5 + Cos(0.025 * i) * (-1 + 0.3 * Sin(0.1 * i))

  Range("I16") = _

    5 + Sin(0.025 * i) * (-1 + 0.3 * Sin(0.1 * i))

  Range("H17") = _

    5 + Cos(0.025 * i) * (-1 + 0.4 * Sin(0.1 * i))

  Range("I17") = _

    5 + Sin(0.025 * i) * (-1 + 0.4 * Sin(0.1 * i))

  Range("H18") = _

    5 + Cos(0.025 * i) * (-1 + 0.5 * Sin(0.1 * i))

  Range("I18") = _

    5 + Sin(0.025 * i) * (-1 + 0.5 * Sin(0.1 * i))

  Range("H19") = _

    5 + Cos(0.025 * i) * (1 - 0.2 * Sin(0.1 * i))

  Range("I19") = _

    5 + Sin(0.025 * i) * (1 - 0.2 * Sin(0.1 * i))

  Range("H20") = _

    5 + Cos(0.025 * i) * (1 - 0.3 * Sin(0.1 * i))

  Range("I20") = _

    5 + Sin(0.025 * i) * (1 - 0.3 * Sin(0.1 * i))

  Range("H21") = _

    5 + Cos(0.025 * i) * (1 - 0.4 * Sin(0.1 * i))

  Range("I21") = _

    5 + Sin(0.025 * i) * (1 - 0.4 * Sin(0.1 * i))

  Range("H22") = _

    5 + Cos(0.025 * i) * (1 - 0.5 * Sin(0.1 * i))

  Range("I22") = _

    5 + Sin(0.025 * i) * (1 - 0.5 * Sin(0.1 * i))

  Application.ScreenUpdating = True

Next i

For i = 503 To 630

  Range("H15") = 4 + 0.2 * Sin(0.1 * i)

  Range("H16") = 4 + 0.3 * Sin(0.1 * i)

  Range("H17") = 4 + 0.4 * Sin(0.1 * i)

  Range("H18") = 4 + 0.5 * Sin(0.1 * i)

  Range("H19") = 6 - 0.2 * Sin(0.1 * i)

  Range("H20") = 6 - 0.3 * Sin(0.1 * i)

  Range("H21") = 6 - 0.4 * Sin(0.1 * i)

  Range("H22") = 6 - 0.5 * Sin(0.1 * i)

  Application.ScreenUpdating = True

Next i

Range("H15") = 4: Range("I15") = 5

Range("H16") = 4: Range("I16") = 5

Range("H17") = 4: Range("I17") = 5

Range("H18") = 4: Range("I18") = 5

Range("H19") = 6: Range("I19") = 5

Range("H20") = 6: Range("I20") = 5

Range("H21") = 6: Range("I21") = 5

Range("H22") = 6: Range("I22") = 5

Application.ScreenUpdating = True

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''^                            ^'''''''''''''''

'''''''''''''^    LAGRANGE INTERPOLATION    ^''''''''''''''

''''''''''''''^                            ^'''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

Function Lagrange(XArray, YArray, x, m)

' m denotes the order of the polynomial used,

' and must be an integer between 1 and 14

Dim Row As Integer, i As Integer, j As Integer

Dim Term As Double, Y As Double

Row = Application.Match(x, XArray, 1)

If Row < (m + 1) / 2 Then Row = (m + 1) / 2

If Row > XArray.Count - (m + 1) / 2 Then _

  Row = XArray.Count - (m + 1) / 2

For i = Row - (m - 1) / 2 To Row + (m + 1) / 2

  Term = 1

  For j = Row - (m - 1) / 2 To Row + (m + 1) / 2

    If i <> j Then Term = Term * (x - XArray(j)) / _

      (XArray(i) - XArray(j))

  Next j

  Y = Y + Term * YArray(i)

Next i

Lagrange = Y

End Function

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''^^^^^^^^^^^^^^^^^^^^''''''''''''''''''''

''''''''''''''''''^                    ^'''''''''''''''''''

'''''''''''''''''^      SOLVER SCAN     ^''''''''''''''''''

''''''''''''''''''^                    ^'''''''''''''''''''

'''''''''''''''''''^^^^^^^^^^^^^^^^^^^^''''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''' v3, Oct.1, 2003

' PURPOSE:

'

' SolverScan uses Solver to scan possible combinations of

' two adjustable parameters, P and Q, by changing the

' initial guess values that Solver uses as its points of

' departure. Its output displays the value of the criterion

' (such as SSR) used to optimize P and Q.

'

' The user specifies the initial values, the step sizes to

' be used to vary those initial values, and the numbers of

' such steps. The output is presented as an array suitable

' for display with Mapper or as a 3-D plot.

'

' Selecting a fine search grid (i.e., many small steps)

' increases the execution time of the macro. The macro has

' built-in limits of 251 steps per parameter, but using far

' fewer steps is usually sufficient, and certainly makes

' the macro complete its task much faster.

'

' INPUT:

'

' Set up Solver as if for a single operation, including

' all necessary specifications of constraints and options,

' if any. Then call SolverScan and answer the questions in

' the input boxes. SolverScan will start once the output

' area has been specified.

'

' OUTPUT:

'

' The user selects the location where the output array will

' appear, complete with top and left-hand axis labels for

' subsequent visual display with Mapper or as a 3-D plot

' of SSR as a function of P and Q, where P and Q are the

' two Solver-adjusted parameters, and where P is plotted

' in the X-direction, and Q in the Y-direction.

'

' SUBROUTINES:

'

' SolverScan requires that both Solver and Solver.xla

' have been installed.

'

' ONE-TIME PROCEDURE FOR INSTALLING SOLVER.XLA

'

' SolverScan can only call Solver as a subroutine, when

' Solver.xla is properly referenced in the VBAProject.

' To see whether this is the case, go to the spreadsheet,

' select its VBA module, and look under Tools ->

' References to see whether SOLVER has been activated.

' It should be part of the contiguous, check-marked top

' items in the listed References - VBAProject. If it is

' not, first prepare your computer to make Solver callable.

' The procedure to do so is outlined below, and is needed

' only once until the operating system and/or Excel is

' re-installed or updated.

'

' Click on the Start button, and in the resulting menu, on

' Search, For Files or Folders...(or Find > Files or

' Folders). In the Find: All Files dialog box, under the

' tab Name & Location, in the Name: window type Solver.xla,

' and, if necessary, enter Systemdisk[C:] in the Look in:

' window, activate Include subfolders, and click on Find

' now. Note down where Solver.xla is located. (If you have

' Microsoft Office, it most likely is in Systemdisk[C:],

' Program Files\Microsoft Office\Office\Library\Solver.)

' Exit the Search or Find dialog box, which will bring you

' back to the spreadsheet, select the VBEditor with

' Alt+F11, then go to Tools -> References, which will

' display the References - VBAProject dialog box. Click on

' Browse, select Filers of type: Microsoft Excel Files

' (*.xls, *.xla), and now that you know where to find it,

' navigate your way to Solver.xla, and Open it. This

' will return you to the References - VBAProject dialog

' box, where you now use the Priority up button to move

' it up, so that it is listed contiguously with the other,

' already activated add-ins. Click OK; from then on (i.e.,

' until you reload or upgrade Excel) VBA will know how to

' find Solver when it is called from a macro.

Sub SolverScan()

Dim nc As Integer, ns As Integer, nz As Integer

Dim p As Integer, PassOut As Integer, PassP As Integer

Dim PassQ As Integer, PassTarget As Integer, q As Integer

Dim StepNumberP As Integer, StepNumberQ As Integer

Dim InitialValueP As Double, InitialValueQ As Double

Dim StepSizeP As Double, StepSizeQ As Double

Dim SSR As Variant, z As Variant

Dim RangeOut As Range, RangeP As Range

Dim RangeQ As Range, RangeTarget As Range

Dim AddressOut, AddressP, AddressQ, AddressTarget

Dim Answer, Default, Prompt, Title

' Specify the location of the Target criterion

Prompt = "Enter a CELL ADDRESS:" & Chr(13) & Chr(13) & _

  "The target cell (containing the opti-" _

  & Chr(13) & "mization criterion) is located in:"

Title = "SolverScan: Address of target cell"

Set RangeTarget = Application.InputBox(Prompt, Title, Type:=8)

RangeTarget.Select

AddressTarget = RangeTarget.Address

SSR = RangeTarget.Value

PassP = 1

AP:

' Read the address of the first adjustable parameter, P

Prompt = "Enter a CELL ADDRESS:" & Chr(13) & Chr(13) & _

  "The first adjustable parameter, P, is located in:"

Title = "SolverScan: Address of P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = RangeP.Address

Set RangeP = Application.InputBox(Prompt, Title, Default, Type:=8)

RangeP.Select

AddressP = RangeP.Address

' Specify the initial guess value for P

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The initial value of P is:"

Title = "SolverScan: Initial value of P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = InitialValueP

InitialValueP = InputBox(Prompt, Title, Default)

' Specify the step size for P

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The step size for P is:"

Title = "SolverScan: Step size for P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = StepSizeP

StepSizeP = Application.InputBox(Prompt, Title, Default)

' Specify the number of steps for P

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The number of steps for P is:"

Title = "SolverScan: Step number for P"

If PassP = 1 Then Default = " "

If PassP > 1 Then Default = StepNumberP

StepNumberP = Application.InputBox(Prompt, Title, Default)

PassP = PassP + 1

' Check that StepNumberP is smaller than 252

If StepNumberP > 251 Then

  MsgBox "The number of steps cannot be larger than 251."

  GoTo AP

End If

PassQ = 1

AQ:

' Read the address of the first adjustable parameter, Q

Prompt = "Enter a CELL ADDRESS:" & Chr(13) & Chr(13) & _

  "The first adjustable parameter, Q, is located in:"

Title = "SolverScan: Address of Q"

If PassQ = 1 Then Default = ""

If PassQ > 1 Then Default = RangeQ.Address

Set RangeQ = Application.InputBox(Prompt, Title, _

  Default, Type:=8)

RangeQ.Select

AddressQ = RangeQ.Address

' Specify the initial guess value for Q

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The initial value of Q is:"

Title = "SolverScan: Initial value of Q"

If PassQ = 1 Then Default = ""

If PassQ > 1 Then Default = InitialValueQ

InitialValueQ = InputBox(Prompt, Title, Default)

' Specify the step size for Q

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The step size for Q is:"

Title = "SolverScan: Step size for Q"

If PassQ = 1 Then Default = " "

If PassQ > 1 Then Default = StepSizeQ

StepSizeQ = Application.InputBox(Prompt, Title, Default)

' Specify the number of steps for Q

Prompt = "Enter a value:" & Chr(13) & Chr(13) & _

  "The number of steps for Q is:"

Title = "SolverScan: Step number for Q"

If PassQ = 1 Then Default = " "

If PassQ > 1 Then Default = StepNumberQ

StepNumberQ = Application.InputBox(Prompt, Title, Default)

PassQ = PassQ + 1

' Check that StepNumberQ is smaller than 252

If StepNumberQ > 251 Then

  MsgBox "The number of steps cannot be larger than 251."

  GoTo AQ

End If

' Specify the top left-hand corner of the Output array

PassOut = 1

AOut:

Prompt = "The output array, including its axes, " _

  & "will be" & Chr(13) & StepNumberP + 1 _

  & " columns wide, and " & StepNumberQ + 1 _

  & " rows high." & Chr(13) & _

  "                     Enter a CELL ADDRESS:" & Chr(13) _

  & "The top left-hand corner of the output" & Chr(13) _

  & "array should be located at:"

Title = "SolverScan: Top left-hand corner of output array"

If PassOut = 1 Then Default = " "

If PassOut > 1 Then Default = RangeOut.Address

Set RangeOut = Application.InputBox(Prompt, Title, Type:=8)

RangeOut.Select

AddressOut = RangeOut.Address

PassOut = PassOut + 1

' Count the number of columns nc to the left of RangeOut

nc = 0

AEdge:

On Error GoTo OnEdge

Selection.Offset(0, -1).Select

nc = nc + 1

GoTo AEdge

OnEdge:

RangeOut.Select

' Make sure that there is enough space for the output array

If nc + StepNumberP > 254 Then

  MsgBox "Move the left-hand corner of the" & _

    Chr(13) & "output array to the left, in order" & _

    Chr(13) & "to accommodate its projected size."

  GoTo AOut

End If

' Verify that the output will not overwrite valuable data

Application.ScreenUpdating = False

nz = 0

RangeOut.Select

For p = 0 To StepNumberP + 1

  Selection.Offset(0, 1).Select

  For q = 0 To StepNumberQ + 1

    Selection.Offset(1, 0).Select

    z = Selection.Value

    nz = nz + 1

    If (IsEmpty(z) Or z = Null) Then nz = nz - 1

  Next q

Next p

Application.ScreenUpdating = True

If nz > 0 Then

  Answer = MsgBox("There are data in the array in which" _

    & Chr(13) & "the output data will be written. " _

    & "Proceed" & Chr(13) _

    & "anyway and overwrite those data?", vbYesNo)

  If Answer = vbNo Then GoTo AOut

End If

Application.ScreenUpdating = False

ReDim SSR(StepNumberQ + 1, StepNumberP + 1) As Double

' Enter the initial guess values,

' run Solver, and write output file

ns = 0

For p = 0 To StepNumberP

  RangeP.Select

  RangeP.Value = InitialValueP + p * StepSizeP

  For q = 0 To StepNumberQ

    RangeQ.Select

    RangeQ.Value = InitialValueQ + q * StepSizeQ

    Application.DisplayStatusBar = False

    SolverOK SetCell:="addressTarget", MaxMinVal:=2, _

      ByChange:="addressP, addressQ"

    SolverSolve UserFinish:=True

    RangeTarget.Select

    SSR(q, p) = RangeTarget.Value

    ns = ns + 1

    Application.DisplayStatusBar = True

    Application.StatusBar = "SolverScan has completed " & _

      ns & " out of " & (StepNumberP + 1) * _

      (StepNumberQ + 1) & " calculations."

  Next q

Next p

' Write the axis scale labels

RangeOut.Select

For p = 1 To StepNumberP + 1

  Selection.Offset(0, p).Select

  Selection.Value = InitialValueP + (p - 1) * StepSizeP

  ActiveCell.Font.Italic = True

  ActiveCell.Font.Bold = True

  Selection.Offset(0, -p).Select

Next p

For q = 1 To StepNumberQ + 1

  Selection.Offset(q, 0).Select

  ActiveCell.Font.Italic = True

  ActiveCell.Font.Bold = True

  Selection.Value = InitialValueQ + (q - 1) * StepSizeQ

  Selection.Offset(-q, 0).Select

Next q

' Output the SSR data

RangeOut.Select

For p = 0 To StepNumberP

  For q = 0 To StepNumberQ

    Selection.Offset(q + 1, p + 1).Select

    Selection.Value = SSR(q, p)

    Selection.Offset(-q - 1, -p - 1).Select

  Next q

Next p

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

''''''''''''''''''''^^^^^^^^^^^^^^^^^^^''''''''''''''''''''

'''''''''''''''''''^                   ^'''''''''''''''''''

''''''''''''''''''^       MAPPER        ^''''''''''''''''''

'''''''''''''''''''^                   ^'''''''''''''''''''

''''''''''''''''''''^^^^^^^^^^^^^^^^^^^''''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''''v4, May 1, 2004

' PURPOSE:

'

' This macro reads a highlighted block of data, including

' axis labels at its top and left-hand side. This format

' was chosen in order to be compatible with that of Excel

' 2-D graphs and 3-D surface plots. The macro separates

' these labels, converts the remaining data into RGB colors

' with either a gray scale (for iColor = 0) or one of three

' sample color schemes provided (iColor = 1 to 3), and uses

' these to make a graph. If you wish, look at the subrou-

' tines BitMap0 through BitMap3 to see how it is done, then

' add your own color scheme.

'

' This macro is most useful when there are many data points

' in both directions. If not, Excel's 3-D plot may well be

' the more satisfactory. Both of them require the input

' data to be equidistant in x as well as in y. When the

' plot contains relatively few data points, the simple, bi-

' linear interpolation to generate pixel values can cause

' star-like artifacts, as if one looks at a street lamp

' through a finely woven cloth. This could be remedied by

' using a more sophisticated interpolation scheme.

'

' The resulting figure can be moved, enlarged or reduced,

' annotated, and modified as any other XY graph. Data

' points, curves, and text can be added, scales can be

' modified or deleted, etc. 

'

' In order to facilitate displaying scales, the macro 

' checks whether the top left cell in the highlighted

' array is empty, and the cells below it and to its right 

' are filled. Note, however, that the scales displayed 

' merely reflect the numbers provided in the top row and

' the left column, and are NOT functional, so that modi-

' fying them afterwards will not rescale or otherwise 

' affect the map. Moreover, the map does not respond to 

' subsequent changes in the input data until the macro 

' is called again to update the input information.

'

' SUBROUTINES:

'

' The drivers Mapper0 through Mapper3 call the main routine

' Mapper, which in turn calls on a number of subroutines:

' BitmapHeader, WriteAWord, WriteALong, WriteAPixel,

' WriteAByte, and Bitmap0 through Bitmap3. If you want to

' add a different color scheme, just add one as Bitmap4

' with a corresponding driver (Mapper4), using the existing

' code as example. If you use the MBToolbar or MBMenu, make

' sure to include the new driver(s) there.

'

' INPUT:

'

' The macro requires an input array together with the

' corresponding scales, which must be placed in the top

' row and in the left-hand column, leaving the top left-

' hand cell empty.

'

' OUTPUT:

'

' The resulting map can be resized, rescaled, moved, and

' otherwise manipulated as any graph. You can also insert

' text, overlay curves, add inserts, etc.

'

' PROCEDURE:

'

' Highlight the array to be mapped together with the

' corresponding scales, which must be placed in the

' top row and in the left-hand column, leaving the

' top left-hand cell empty. Then call the macro.

'

' WARNING:

'

' This macro goes OUTSIDE OF Excel in order to write

' a bitmap, which is then re-imported as 'background'

' for the graph. Depending on how the rest of the Office

' bundle is configured (or in its absence, when only

' Excel is loaded) the macro may not always run properly.

Sub Mapper0()

Dim iColor As Integer

iColor = 0

Call Mapper(0)

End Sub

Sub Mapper1()

Dim iColor As Integer

iColor = 1

Call Mapper(1)

End Sub

Sub Mapper2()

Dim iColor As Integer

iColor = 2

Call Mapper(2)

End Sub

Sub Mapper3()

Dim iColor As Integer

iColor = 3

Call Mapper(3)

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Sub Mapper(iColor As Integer)

' Nomenclature used:

'

' inputArray    the highlighted block containing the data

'                 plus (at its top and left-hand side)

'                 either axis labels or space for them.

' scaledArray   the input array scaled to fit the range

'                 from 0 to 255.

' valueArray    the interpolated array. It has as many

'                 terms as there are pixels in the plotting

'                 area. All terms are double precision, in

'                 the range from 0 to 255

' pixelArray    the value array expressed in terms of

'                 integers.

' rn            input data height.

' ph            pixel (instead or row) height, starting at

'                 the bottom with 0.

' cw            input data width.

' pw            pixel (instead of column) width, starting

'                 at the left with 0.

' cwMax-1, rhMax-1     number of input data, in terms of

'                 columns and rows respectively.

' hMax, wMax    number of pixels, in terms of height and

'                 width respectively.

' HH, WW        number of pixels per input data point, in

'                 terms of height and width respectively.

Dim cn As Integer, cw As Integer, cwMax As Integer

Dim ph As Integer, pw As Integer

Dim rh As Integer, rn As Integer, rhMax As Integer

Dim i As Integer, iH As Integer, iW As Integer

Dim j As Integer

Dim OutputFileNumber As Integer

Dim maxValue As Double, minValue As Double

Dim Product As Long

Dim XLabelMax As Double, XLabelMin As Double

Dim YLabelMax As Double, YLabelMin As Double

Dim a1 As Integer, a2 As Integer, b1 As Integer

Dim h As Integer, HH As Integer, hScale As Integer

Dim hMax As Integer, hMax0 As Integer       ' plot pixel height

Dim wMax As Integer, wMax0 As Integer   ' plot pixel width

Dim ns As Integer                  ' number of data series

Dim p As Integer

Dim w As Integer, WW As Integer, wScale As Integer

Dim h1 As Double, h2 As Double

Dim w1 As Double, w2 As Double

Dim inputArray As Variant, scaledArray As Variant

Dim pixelArray As Variant, ValueArray As Variant

Dim OutputFileName As String, Sheetname As String

Dim hAnswer, myRange, ThisSheet

' Read the inputArray

Begin:

inputArray = Selection.Value

cwMax = Selection.Columns.Count

rhMax = Selection.Rows.Count

' If input area was not highlighted

If rhMax = 1 And cwMax = 1 Then

  hAnswer = MsgBox("You forgot to highlight" _

    & Chr(13) & "the block of input data." _

    & Chr(13) & "Do you want to do so now?" _

    , vbYesNo, "Mapper")

  If hAnswer = vbNo Then End

  If hAnswer = vbYes Then

    Set myRange = Application.InputBox(Prompt:= _

      "The input data are located in:", Type:=8)

    myRange.Select

  End If

  GoTo Begin

End If

Application.ScreenUpdating = False

If rhMax < 3 Then

  MsgBox "The block, including (space for)" & _

    Chr(13) & "a scale at the top row," & _

    Chr(13) & "should be at least 3 cells high."

End

End If

If cwMax < 3 Then

  MsgBox "The block, including (space for)" & _

    Chr(13) & "a scale at the left-most column," & _

    Chr(13) & "should be at least 3 cells wide."

  End

End If

' Make sure that the top left cell is empty, but that

' its two nearest neighbors are not. Note: the top column

' and the left-most column will be used for creating the

' map scales, and will NOT be plotted.

p = 0

If IsEmpty(inputArray(1, 1)) Or IsNull(inputArray(1, 1)) _

  Then p = p + 1

If p = 0 Then

  MsgBox "Make sure that the top left cell" & _

  Chr(13) & "of the highlighted area is empty."

  End

End If

If IsEmpty(inputArray(2, 1)) Or IsNull(inputArray(2, 1)) _

  Then p = p + 1

If IsEmpty(inputArray(1, 2)) Or IsNull(inputArray(1, 2)) _

  Then p = p + 1

If p <> 1 Then

  MsgBox "   Make sure that the top row and the" & _

  Chr(13) & "left-hand column contain the plot scales," & _

  Chr(13) & "while leaving the left top cell empty."

  End

End If

' Find maximum and minimum values in XLabels and YLabels

YLabelMax = inputArray(2, 1)

YLabelMin = inputArray(2, 1)

For rn = 3 To rhMax

  If inputArray(rn, 1) > YLabelMax Then YLabelMax = _

    inputArray(rn, 1)

  If inputArray(rn, 1) < YLabelMin Then YLabelMin = _

    inputArray(rn, 1)

Next rn

XLabelMax = inputArray(1, 2)

XLabelMin = inputArray(1, 2)

For cn = 3 To cwMax

  If inputArray(1, cn) > XLabelMax Then XLabelMax = _

    inputArray(1, cn)

  If inputArray(1, cn) < XLabelMin Then XLabelMin = _

    inputArray(1, cn)

Next cn

' Find maximum and minimum values in inputArray

maxValue = inputArray(2, 2)

minValue = inputArray(2, 2)

For rn = 2 To rhMax

  For cn = 2 To cwMax

    If inputArray(rn, cn) > maxValue Then maxValue = _

      inputArray(rn, cn)

    If inputArray(rn, cn) < minValue Then minValue = _

      inputArray(rn, cn)

  Next cn

Next rn

' Scale the inputArray on a scale from 0 to 255

ReDim scaledArray(0 To rhMax, 0 To cwMax)

For rn = 2 To rhMax

  For cn = 2 To cwMax

    scaledArray(rn, cn) = 255 * (inputArray(rn, cn) _

      - minValue) / (maxValue - minValue)

  Next cn

Next rn

' Draw graph

cw = Columns(1).Width

rh = Rows(1).Height

Sheetname = ActiveSheet.Name

Charts.Add

ActiveChart.ChartType = xlXYScatter

ActiveChart.Location Where:=xlLocationAsObject, Name:=Sheetname

ActiveChart.Axes(xlValue).Select

With ActiveChart.Axes(xlValue)

  .MinimumScaleIsAuto = True

  .MaximumScaleIsAuto = True

  .MajorUnitIsAuto = True

  .MinorUnitIsAuto = True

  .Crosses = xlAutomatic

  .MajorTickMark = xlCross

  .MinorTickMark = xlNone

  .TickLabelPosition = xlLow

End With

Selection.TickLabels.NumberFormat = "General"

ActiveChart.Axes(xlCategory).Select

With ActiveChart.Axes(xlCategory)

  .MinimumScaleIsAuto = True

  .MaximumScaleIsAuto = True

  .MajorUnitIsAuto = True

  .MinorUnitIsAuto = True

  .Crosses = xlAutomatic

  .MajorTickMark = xlCross

  .MinorTickMark = xlNone

  .TickLabelPosition = xlLow

End With

With Selection.Border

   .Weight = xlHairline

   .LineStyle = xlAutomatic

End With

' Round plot area height and width to integer

' multiples of the data in the input array

wScale = 1

If wMax > cwMax - 1 Then wScale = Int(wMax0 / (cwMax - 2))

hScale = 1

If hMax > rhMax - 1 Then hScale = Int(hMax0 / (rhMax - 2))

wMax = (cwMax - 2) * wScale

hMax = (rhMax - 2) * hScale

HH = hMax / (rhMax - 2)

WW = wMax / (cwMax - 2)

' Calculate valueArray for cn = 2 to cwMax

ReDim ValueArray(0 To hMax, 0 To wMax) As Double

ReDim pixelArray(0 To hMax, 0 To wMax) As Integer

For cn = 2 To cwMax

  h = hMax

  iH = Int(h / HH)

  h1 = (iH + 1) - h / HH

  h2 = h / HH - iH

  ValueArray(hMax, (cn - 2) * WW) = _

    h1 * scaledArray(iH + 2, cn)

Next cn

For cn = 2 To cwMax

  For h = 0 To hMax - 1

    iH = Int(h / HH)

    h1 = (iH + 1) - h / HH

    h2 = h / HH - iH

    ValueArray(h, (cn - 2) * WW) = _

      h1 * scaledArray(iH + 2, cn) _

      + h2 * scaledArray(iH + 3, cn)

  Next h

Next cn

' Calculate all of valueArray

For h = 0 To hMax

  For w = 1 To wMax - 1

    iW = Int(w / WW)

    w1 = (iW + 1) - w / WW

    w2 = w / WW - iW

    ValueArray(h, w) = w1 * ValueArray(h, iW * WW) + _

       w2 * ValueArray(h, (iW + 1) * WW)

  Next w

Next h

' Convert pixel values to integers

For h = 0 To hMax

  For w = 0 To wMax

    pixelArray(h, w) = CInt(ValueArray(h, w))

  Next w

Next h

'Create, write, and close the bitmap file

OutputFileName = "ColorMap.BMP"

OutputFileNumber = 1

Open OutputFileName For Binary Access Write As _

  #OutputFileNumber

Call BitmapHeader(hMax, wMax)

If iColor = 0 Then Call BitMap0(hMax, wMax, pixelArray)

If iColor = 1 Then Call BitMap1(hMax, wMax, pixelArray)

If iColor = 2 Then Call BitMap2(hMax, wMax, pixelArray)

If iColor = 3 Then Call BitMap3(hMax, wMax, pixelArray)

Close #OutputFileNumber

ThisSheet = ActiveSheet.Name

ActiveChart.PlotArea.Fill.UserPicture OutputFileName

' Maximize the plot area

ActiveChart.PlotArea.Select

Selection.Fill.Visible = True

' Set the axis labels

ActiveChart.Axes(xlCategory).Select

ActiveChart.Axes(xlCategory).MinimumScale = XLabelMin

ActiveChart.Axes(xlCategory).MaximumScale = XLabelMax

ActiveChart.Axes(xlValue).Select

ActiveChart.Axes(xlValue).MinimumScale = YLabelMin

ActiveChart.Axes(xlValue).MaximumScale = YLabelMax

' Remove all markers

ns = ActiveChart.SeriesCollection.Count

For i = 1 To ns

  ActiveChart.SeriesCollection(i).Select

  ActiveChart.SeriesCollection(i).MarkerStyle = xlNone

Next i

ActiveChart.ChartArea.Select

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

' This routine writes a minimal 54 byte bitmap file header.

Private Sub BitmapHeader(ph As Integer, pw As Integer)

' Values for pixel width pW, pixel height pH, ImageSize-

' InByte and TotalFileSize change with the image size.

'

Dim RowLengthInBytes  As Long

Dim ImageSizeInByte As Long

Dim TotalFileSize As Long

' Calculate three values needed write header. In

' particular, row length is the number of bytes

' rounded up to the next DWORD (4 byte) boundary.

RowLengthInBytes = (((pw + 1) * 3) \ 4) * 4

ImageSizeInByte = RowLengthInBytes * ph

TotalFileSize = 54 + (ph * RowLengthInBytes)

' Write the BitmapFileHeader

WriteAWord 19778        ' bfType   The two characters BM

WriteALong TotalFileSize        ' bfSize   Headers + image

WriteALong 0        ' bfReserved1&2  Two words of zero.

WriteALong 54       ' bfOffBits      Byte offset to image.

' Write the BitmapInfoHeader

WriteALong 40   ' biSize       BITMAPINFOHEADER size = 40.

WriteALong pw       ' biWidth           Image width.

WriteALong ph       ' biHeight          Image width.

WriteAWord 1        ' biPlanes          Always 1.

WriteAWord 24       ' biBitCount   Always 24 bits per pixel.

WriteALong 0        ' biCompression     No compression = 0

WriteALong ImageSizeInByte

                ' biSizeImage       Image size in bytes

WriteALong 3000     ' biXPelsPerMeter   A safe value

WriteALong 3000     ' biYPelsPerMeter   A safe value

WriteALong 0        ' biClrUsed         Always 0

WriteALong 0        ' biClrImportant    Always 0

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap0(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

Dim h As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For h = hMax To 0 Step -1

  For w = 0 To wMax - 1

    RedVal = pixelArray(h, w)

    GreenVal = pixelArray(h, w)

    BlueVal = pixelArray(h, w)

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next h

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap1(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

Dim h As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For h = hMax To 0 Step -1

  For w = 0 To wMax - 1

' from black to dark gray

    If pixelArray(h, w) < 32 Then

      RedVal = Int(4 * pixelArray(h, w))

      GreenVal = Int(4 * pixelArray(h, w))

      BlueVal = Int(4 * pixelArray(h, w))

' from dark gray to dark brown

    ElseIf pixelArray(h, w) < 64 Then

      RedVal = 128

      GreenVal = Int(192 - 2 * pixelArray(h, w))

      BlueVal = Int(192 - 2 * pixelArray(h, w))

' from dark brown to dark purple

    ElseIf pixelArray(h, w) < 96 Then

      RedVal = 128

      GreenVal = 64

      BlueVal = Int(-64 + 2 * pixelArray(h, w))

' from dark purple to dark blue

    ElseIf pixelArray(h, w) < 128 Then

      RedVal = Int(320 - 2 * pixelArray(h, w))

      GreenVal = 64

      BlueVal = Int(-64 + 2 * pixelArray(h, w))

' from dark blue to light blue

    ElseIf pixelArray(h, w) < 160 Then

      RedVal = Int(192 - pixelArray(h, w))

      GreenVal = Int(-576 + 5 * pixelArray(h, w))

      BlueVal = Int(64 + pixelArray(h, w))

' from light blue to green

    ElseIf pixelArray(h, w) < 192 Then

      RedVal = Int(192 - pixelArray(h, w))

      GreenVal = 224

      BlueVal = Int(864 - 4 * pixelArray(h, w))

' from green to yellow green

    ElseIf pixelArray(h, w) < 208 Then

      RedVal = Int(-1152 + 6 * pixelArray(h, w))

      GreenVal = 224

      BlueVal = 96

' from yellow green to brown yellow

    ElseIf pixelArray(h, w) < 224 Then

      RedVal = Int(-1568 + 8 * pixelArray(h, w))

      GreenVal = 224

      BlueVal = 96

' from brown yellow to bright yellow

    ElseIf pixelArray(h, w) < 240 Then

      RedVal = Int(-224 + 2 * pixelArray(h, w))

      GreenVal = Int(-224 + 2 * pixelArray(h, w))

      BlueVal = 96

' from bright yellow to white

    ElseIf pixelArray(h, w) >= 240 Then

      RedVal = 255

      GreenVal = 255

      BlueVal = Int(-864 + 10 * pixelArray(h, w))

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next h

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap2(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

Dim h As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For h = hMax To 0 Step -1

  For w = 0 To wMax - 1

' from black to red

    If pixelArray(h, w) < 64 Then

      RedVal = Int(4 * pixelArray(h, w))

      GreenVal = 0

      BlueVal = 0

' from red to orange

    ElseIf pixelArray(h, w) < 128 Then

      RedVal = 255

      GreenVal = Int(2 * (pixelArray(h, w) - 64))

      BlueVal = 0

' from orange to yellow

    ElseIf pixelArray(h, w) < 192 Then

      RedVal = 255

      GreenVal = Int(2 * (pixelArray(h, w) - 64))

      BlueVal = 0

' from yellow to white

    Else

      RedVal = 255

      GreenVal = 255

      BlueVal = Int(4 * (pixelArray(h, w) - 192))

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next h

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub BitMap3(hMax As Integer, _

  wMax As Integer, pixelArray As Variant)

Dim h As Integer, w As Integer, RedVal As Integer

Dim GreenVal As Integer, BlueVal As Integer

For h = hMax To 0 Step -1

  For w = 0 To wMax - 1

' from black to dark blue

    If pixelArray(h, w) < 64 Then

      RedVal = 0

      GreenVal = 0

      BlueVal = Int(4 * pixelArray(h, w))

' from dark blue to light blue

    ElseIf pixelArray(h, w) < 192 Then

      RedVal = 0

      GreenVal = Int(2 * (pixelArray(h, w) - 64))

      BlueVal = 255

' from light blue to white

    Else

      RedVal = Int(4 * (pixelArray(h, w) - 192))

      GreenVal = 255

      BlueVal = 255

    End If

    WriteAPixel RedVal Mod 256, GreenVal Mod 256, _

      BlueVal Mod 256

  Next w

' Do not change the following, essential row padding

  w = wMax * 3

  Do While (w Mod 4) <> 0

    WriteAByte 0

    w = w + 1

  Loop

Next h

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteAWord(ByVal WordValue As Integer)

' Write a two byte short word to the binary data stream.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , WordValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteALong(ByVal LongValue As Long)

' Write a four-byte long word to the binary data stream.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , LongValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteAPixel(ByVal RedValue As Byte, _

  ByVal GreenValue As Byte, ByVal BlueValue As Byte)

' Write a three-byte pixel color value to

' the binary data stream. Note that there is

' no padding up to a word or long boundary.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , BlueValue

Put #OutputFileNumber, , GreenValue

Put #OutputFileNumber, , RedValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

Private Sub WriteAByte(ByVal ByteValue As Byte)

' Write a byte.

Dim OutputFileNumber As Integer

OutputFileNumber = 1

Put #OutputFileNumber, , ByteValue

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''^          ROOT FINDER          ^'''''''''''''

''''''''''''''^                             ^''''''''''''''

'''''''''''''''^^^^^^^^^^^^^^^^^^^^^^^^^^^^^'''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''v4, March 1, 2004

' PURPOSE:

' This macro finds the real root of a function F(x) of a

' single variable x by bisection inside a user-defined

' range. Note: because the macro looks for a sign change

' in F(x), it will also find singularities, such as with

' tan(x) for x = pi/2.

' The value of the resolution criterion FRes can be changed

' depending on the requirements of the problem. For simple

' functions, a high resolution (i.e., a small value of

' FRes, such as 1E-12) may be desired, while complicated

' expressions may require a lower resolution. For functions

' that reach values much larger than 1, a relative rather

' than an absolute resolution may be more appropriate.

' SUBROUTINES:

' This macro does not require any subroutines.

' INPUT:

' The locations of the function F(x) and the independent

' variable x must be in columns of equal length, and are

' entered into the macro via input boxes.

' The parameter x must change monotonically (i.e., either

' ascend or descend continuously) in its column, while

' the function F(x) should have one and only one zero

' crossing in the selected column or column fragment.

' The parameter x can be a value or a function of some

' other parameters.

' If the function on the spreadsheet exhibits more than one

' zero crossing, restrict the size of the block so that it

' will contain only one of them.

' OUTPUT:

' The root is displayed in a message box.

' PROCEDURE:

' Call the macro, then enter the column ranges for F(x) and

' x in the input boxes.

Sub RootFinder()

Dim i As Integer, n As Integer

Dim nc As Integer, nr As Integer

Dim FFirst As Double, FLast As Double, FNew As Double

Dim FRes As Double

Dim XFirst As Double, XLast As Double, XNew As Double

Dim FFormula As Variant, FValue As Variant

Dim XFormula As Variant, XValue As Variant

Dim FFormulaSave As Variant, XFormulaSave As Variant

Dim FRange As Range, XRange As Range

Dim FPrompt, XPrompt, FTitle, XTitle

' Input the function F(x) and its associated files

FPrompt = "The function F(x) is in column:"

FTitle = "RootFinder: Input F(x)"

Set FRange = Application.InputBox(FPrompt, FTitle, Type:=8)

FRange.Select

FFormula = Selection.Formula

FValue = Selection.Value

FFormulaSave = Selection.Formula

' Test the F(x) input

nc = Selection.Columns.Count

If nc <> 1 Then

  MsgBox "The function should be in a single column."

  End

End If

nr = Selection.Rows.Count

If nr < 3 Then

  MsgBox "The column should have at least three rows."

  End

End If

' Input the variable x and its associated files

XPrompt = "The variable x is in column:"

XTitle = "RootFinder: Input x"

Set XRange = Application.InputBox(XPrompt, XTitle, Type:=8)

XRange.Select

XFormula = Selection.Formula

XValue = Selection.Value

XFormulaSave = Selection.Formula

' Test the x input

nc = Selection.Columns.Count

If nc <> 1 Then

  MsgBox "The variable should be in a single column."

  End

End If

If nr <> Selection.Rows.Count Then

  MsgBox "The F(x) and x columns must have equal lengths."

  End

End If

' Check for one root

' Note: an odd number of roots will also pass this test

FFirst = FValue(1, 1)

FLast = FValue(nr, 1)

If FFirst * FLast >= 0 Then

  MsgBox "The array should encompass a single root."

  End

End If

' Set the resolution; FRes must be positive

FRes = 0.0000000000001

' Suppress screen updating

Application.ScreenUpdating = False

' Bisection

n = 0

Do

  XRange.Select

  XFirst = XValue(1, 1)

  XLast = XValue(nr, 1)

  XNew = (XFirst + XLast) / 2

  XValue(2, 1) = XNew

  Selection.Value = XValue

  FRange.Select

  FValue = Selection.Value

  FFirst = FValue(1, 1)

  FLast = FValue(nr, 1)

  FNew = FValue(2, 1)

  XRange.Select

  If FFirst * FNew = 0 Then GoTo A

  If FFirst * FNew > 0 Then

    For i = 1 To nr

      XValue(i, 1) = XNew + _

        (XLast - XNew) * (i - 1) / (nr - 1)

    Next i

  End If

  If FFirst * FNew < 0 Then

    For i = 1 To nr

      XValue(i, 1) = XFirst + _

        (XNew - XFirst) * (i - 1) / (nr - 1)

    Next i

  End If

  n = n + 1

  Selection.Value = XValue

Loop Until Abs(FLast - FFirst) < FRes Or n = 100

' Restore data input area and display the found root

A:

XRange.Select

Selection.Formula = XFormulaSave

FRange.Select

Selection.Formula = FFormulaSave

MsgBox "Root at x = " & XNew

End Sub

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''''''''''''''''''''''''''''''''''''''''''

'''''''''''''''''''^^^^^^^^^^^^^^^^^^^^''''''''''''''''''''

''''''''''''''''''^                    ^'''''''''''''''''''

'''''''''''''''''^     COLUMN SOLVER    ^''''''''''''''''''

''''''''''''''''''^                    ^'''''''''''''''''''

'''''''''''''''''''^^^^^^^^^^^^^^^^^^^^''''''''''''''''''''

''''''''''''''''''''''''''''''''''''''''''''(c) R. de Levie

''''''''''''''''''''''''''''''''''''''''''v.4 March 1, 2004

' PURPOSE:

'

' ColumnSolver allows the user to apply Solver sequentially

' to data located within in single row, whereupon it moves

' to the next row and repeats the procedure.

' The user specifies the location of the target cells, and

' the location of the adjustable parameter(s). Moreover,

' for ColumnSolverValue, the location of the wanted target

' values must be specified. All input information must be

' in columns and blocks of the same length, with all input

' information for any individual application Solver

' confined to a single row. When there are several

' adjustable parameters, these must be in a single

' contiguous block of the same length as the target column.

'

' ColumnSolverMax maximizes the value in the target cell,

' ColumnSolverMin minimizes the value in the target cell,

' while ColumnSolverValue makes the values of the target

' cells equal to the values specified in the WantedValue

' column. The column or block of parameter values should

' contain appropriate initial values for Solver, which

' ColumnSolver then changes into the final parameter values

' computed by Solver.

'

' INPUT:

'

' Set up Solver as if for a single operation, including all

' necessary specifications of constraints and options, if

' any. Then call ColumnSolver and answer the questions in

' its input boxes. ColumnSolver will start once the needed

' input information has been entered.

'

' OUTPUT:

'

' ColumnSolver adjusts the parameters so that the target cells

' yield the specified criterion, Min, Max, or EqualToValue.

'

' REQUIREMENTS:

'

' ColumnSolver requires that both Solver and Solver.xla have

' been installed.

'

' ONE-TIME PROCEDURE FOR INSTALLING SOLVER.XLA

'

' ColumnSolver can only call Solver as a subroutine when

' Solver.xla is properly referenced in the VBAProject.

' To see whether this is the case, go to the spreadsheet,

' select its VBA module, and look under Tools -> References

' to see whether SOLVER has been activated. It should be

' part of the contiguous, check-marked top items in the

' listed References - VBAProject.

'

' If it is not, first prepare your computer to make Solver

' callable. The procedure to do so is outlined below, and

' is needed only once until the operating system and/or

' Excel is re-installed or updated.

'

' Click on the Start button, and in the resulting menu, on

' Search, under For Files or Folders...(or Find > Files or

' Folders), search for Solver.xla. Note down where

' Solver.xla is located. If you have Microsoft Office, it

' most likely is in C:\Program Files\MicrosoftOffice\Office

' \Library\Solver.

'

' Exit the Search or Find dialog box, which will bring you

' back to the spreadsheet, select the VBA editor with

' Alt+F11, then go to Tools -> References, which will

' display the References - VBAProject dialog box. Click on

' Browse, select Filers of type: Microsoft Excel Files

' (*.xls, *.xla), and now that you know where to find it,

' navigate your way to Solver, and Open it. This will

' return you to the References - VBAProject dialog box,

' where you now use the Priority Up button to move it up,

' so that it is listed contiguously with the other, already

' activated add-ins. Click OK; from then on (i.e., until

' you reload or upgrade Excel) VBA will know how to find

' Solver when it is called from a macro.

Sub ColumnSolverMax()

Dim d As Integer

d = 1

Call ColumnSolver(d)

End Sub

Sub ColumnSolverMin()

Dim d As Integer

d = 2

Call ColumnSolver(d)

End Sub

Sub ColumnSolverValue()

Dim d As Integer

d = 3

Call ColumnSolver(d)

End Sub

Sub ColumnSolver(d)

Dim i As Integer, nc As Integer, nr As Integer

Dim PositionOf2ndDollar As Integer

Dim ParameterAddress As Variant, RowNumber As Variant

Dim TargetAddress As Variant, WantedValue As Variant

Dim myRow As Range, ParameterRange As Range

Dim TargetRange As Range, WantedValues As Range

Dim AAA, Cell, EqualTo, Prompt, Title

' Read the location of the target column

A:

Prompt = "The target cells are located in:" & Chr(13) _

  & "(Highlight a column or enter its address)"

Title = "ColumnSolver: Address of target column"

Set TargetRange = Application.InputBox(Prompt, Title, _

  Type:=8)

TargetRange.Select

nr = Selection.Rows.Count

i = 0

ReDim TargetAddress(1 To nr)

For Each Cell In TargetRange

  i = i + 1

  TargetAddress(i) = Cell.Address

Next Cell

ReDim outputArray(1 To nr)

' Verify that the target range is a single column

nc = Selection.Columns.Count

If nc <> 1 Then

  MsgBox "There can be only one single Target column"

  GoTo A

End If

' Read the column containing the target values

B:

If d = 3 Then

  Prompt = "The wanted target values are located in:" & Chr(13) _

    & "(Highlight a column or enter its address)"

  Title = "ColumnSolver: Wanted target values"

  Set WantedValues = Application.InputBox(Prompt, Title, _

    Type:=8)

  WantedValues.Select

  ReDim WantedValue(1 To nr) As Double

  i = 0

  For Each Cell In WantedValues

    i = i + 1

    WantedValue(i) = Cell.Value

  Next Cell

' Verify that the target range is a single column

  nc = Selection.Columns.Count

  If nc <> 1 Then

    MsgBox "There can be only one single" & _

      Chr(13) & " column with target values"

    GoTo B

  End If

End If

' Verify that the target range has the correct length

If nr - Selection.Rows.Count <> 0 Then

  MsgBox "The column of target values must have the" _

    & Chr(13) & "same number of rows as the target column."

  GoTo B

End If

' Read the location of the adjustable parameter column

C:

Prompt = "The location of the adjustable parameter(s) is:" & _

  Chr(13) & "(Highlight a column or block, or enter its address)"

Title = "ColumnSolver: Address of parameter range"

Set ParameterRange = Application.InputBox(Prompt, Title, _

  Type:=8)

ParameterRange.Select

nc = Selection.Columns.Count

If nc < 1 Then

  MsgBox "There must be at least one" _

    & Chr(13) & "column of adjustable parameters."

  GoTo C

End If

' Verify that the parameter range has the correct length

If nr - Selection.Rows.Count <> 0 Then

  MsgBox "The parameter block must have the same" _

    & Chr(13) & "number of rows as the target column."

  GoTo C

End If

ReDim ParameterAddress(1 To nr)

Application.ScreenUpdating = False

' For one parameter per row

If nc = 1 Then

  i = 0

  For Each Cell In ParameterRange

    i = i + 1

    ParameterAddress(i) = Cell.Address

  Next Cell

' Run Solver

  For i = 1 To nr

    If d < 3 Then EqualTo = " "

    If d = 3 Then EqualTo = WantedValue(i)

    SolverOK SetCell:=TargetAddress(i), MaxMinVal:=d, _

      valueof:=EqualTo, ByChange:=ParameterAddress(i)

    SolverSolve UserFinish:=True

  Next i

  GoTo E

' For multiple parameters per row

Else

' Find the parameter range for the proper row

' at the intersection of ParameterRange and myRow

  ReDim ParameterAddress(1 To nr)

  ReDim RowNumber(1 To nr)

  For i = 1 To nr

    PositionOf2ndDollar = InStr(2, TargetAddress(i), "$")

    RowNumber(i) = Mid(TargetAddress(i), PositionOf2ndDollar)

    AAA = "$A" & RowNumber(i) & ":" & "$IV" & RowNumber(i)

    Set myRow = Range(AAA)

    Intersect(ParameterRange, myRow).Select

    ParameterAddress = Selection.Address

' Run Solver

    If d < 3 Then EqualTo = " "

    If d = 3 Then EqualTo = WantedValue(i)

    SolverOK SetCell:=TargetAddress(i), MaxMinVal:=d, _

      valueof:=EqualTo, ByChange:=ParameterAddress

    SolverSolve UserFinish:=True

  Next i

End If

E:

End Sub

